
Changelog for v1.1
    


  
    
  
  Quick update guide


Here is a quick summary of the changes necessary to upgrade to LiveView v1.1:
	In your mix.exs, update phoenix_live_view to latest and add lazy_html as a dependency:
 {:phoenix_live_view, "~> 1.1"},
 {:lazy_html, ">= 0.0.0", only: :test},
Note you may remove floki as a dependency if you don't use it anywhere.

	Still in your mix.exs, prepend :phoenix_live_view to your list of compilers inside def project, such as:
 compilers: [:phoenix_live_view] ++ Mix.compilers(),

	(optional) In your config/dev.exs, find debug_heex_annotations, and also add debug_tags_location for improved annotations:
 config :phoenix_live_view,
   debug_heex_annotations: true,
   debug_tags_location: true,
   enable_expensive_runtime_checks: true

	(optional) To enable colocated hooks, you must update esbuild with mix deps.update esbuild and then update your config/config.exs accordingly. In particular, append --alias:@=. to the args list and pass a list of paths to the "NODE_PATH" env var, as shown below:
 your_app_name: [
   args:
     ~w(js/app.js --bundle --target=es2022 --outdir=../priv/static/assets/js --external:/fonts/* --external:/images/* --alias:@=.),
   env: %{"NODE_PATH" => [Path.expand("../deps", __DIR__), Mix.Project.build_path()]},



  
    
  
  Colocated hooks


When writing hooks for a specific component, the need to place the JavaScript code in a whole separate file often feels inconvenient. LiveView 1.1 introduces colocated hooks to allow writing the hook's JavaScript code in the same file as your regular component code.
A colocated hook is defined by placing the special :type attribute on a <script> tag:
alias Phoenix.LiveView.ColocatedHook

def input(%{type: "phone-number"} = assigns) do
  ~H"""
  <input type="text" name={@name} id={@id} value={@value} phx-hook=".PhoneNumber" />
  <script :type={ColocatedHook} name=".PhoneNumber">
    export default {
      mounted() {
        this.el.addEventListener("input", e => {
          let match = this.el.value.replace(/\D/g, "").match(/^(\d{3})(\d{3})(\d{4})$/)
          if(match) {
            this.el.value = `${match[1]}-${match[2]}-${match[3]}`
          }
        })
      }
    }
  </script>
  """
end
Important: LiveView now supports the phx-hook attribute to start with a dot (.PhoneNumber above) for namespacing. Any hook name starting with a dot is prefixed at compile time with the module name of the component. If you named your hooks with a leading dot in the past, you'll need to adjust this for your hooks to work properly on LiveView 1.1.
Colocated hooks are extracted to a phoenix-colocated folder inside your _build/$MIX_ENV directory (Mix.Project.build_path()). See the quick update section at the top of the changelog on how to adjust your esbuild configuration to handle this. With everything configured, you can import your colocated hooks inside of your app.js like this:
...
  import {LiveSocket} from "phoenix_live_view"
+ import {hooks as colocatedHooks} from "phoenix-colocated/my_app"
  import topbar from "../vendor/topbar"
...
  const liveSocket = new LiveSocket("/live", Socket, {
    longPollFallbackMs: 2500,
+   params: {_csrf_token, csrfToken},
+   hooks: {...colocatedHooks}
  })
The phoenix-colocated folder has subfolders for each application that uses colocated hooks, therefore you'll need to adjust the my_app part of the import depending on the name of your project (defined in your mix.exs). You can read more about colocated hooks in the module documentation of Phoenix.LiveView.ColocatedHook. There's also a more generalized version for colocated JavaScript, see the documentation for Phoenix.LiveView.ColocatedJS.
We're planning to make the private Phoenix.Component.MacroComponent API that we use for those features public in a future release.

  
    
  
  Keyed Comprehensions


One pitfall when rendering collections in LiveView is that they are not change tracked. If you have code like this:
<ul>
  <li :for={item <- @items}>{item.name}</li>
</ul>
When changing @items, all elements are re-sent over the wire. LiveView still optimizes the static and dynamic parts of the template, but if you have 100 items in your list and only change a single one (or append, prepend, etc.), LiveView still sends the dynamic parts of all items.
To improve this, LiveView prior to version 1.1 had two solutions:
	Use streams. Streams are not kept in memory on the server and if you stream_insert a single item, only that item is sent over the wire. But because the server does not keep any state for streams, this also means that if you update an item in a stream, all the dynamic parts of the updated item are sent again.
	Use a LiveComponent for each entry. LiveComponents perform change tracking on their own assigns. So when you update a single item, LiveView only sends a list of component IDs and the changed parts for that item.

So LiveComponents allow for more granular diffs and also a more declarative approach than streams, but require more memory on the server. Thus, when memory usage is a concern, especially for very large collections, streams should be your first choice. Another downside of LiveComponents is that they require you to write a whole separate module just to get an optimized diff.
LiveView 1.1 introduces a new :key attribute that can be used with :for:
<ul>
  <li :for={item <- @items} :key={item.id}>{item.name}</li>
</ul>
Under the hood, this has the same diff over the wire as a LiveComponent for each entry, but it allows you to define the template inline. LiveView optimizes a :keyed comprehension into a special LiveComponent under the hood. Therefore, this optimization can only be used on regular HTML tags, while :for without :key also works on components and slots. In the future, we might introduce further optimizations that would allow this to also work on components and slots using the same :key syntax.

  
    
  
  Types for public interfaces


LiveView 1.1 adds official types to the JavaScript client. This allows IntelliSense to work in editors that support it and is a massive improvement to the user experience when writing JavaScript hooks.
If you're not using TypeScript, you can also add the necessary JSDoc hints to your hook definitions, assuming your editor supports them.
Example when defining a hook object that is meant to be passed to the LiveSocket constructor:
/**
 * @type {import("phoenix_live_view").HooksOptions}
 */
let Hooks = {}
Hooks.PhoneNumber = {
  mounted() {
    this.el.addEventListener("input", e => {
      let match = this.el.value.replace(/\D/g, "").match(/^(\d{3})(\d{3})(\d{4})$/)
      if(match) {
        this.el.value = `${match[1]}-${match[2]}-${match[3]}`
      }
    })
  }
}

let liveSocket = new LiveSocket("/live", Socket, {hooks: Hooks, ...})
...
Example when defining a hook on its own:
/**
 * @type {import("phoenix_live_view").Hook}
 */
Hooks.InfiniteScroll = {
  page() { return this.el.dataset.page },
  mounted(){
    this.pending = this.page()
    window.addEventListener("scroll", e => {
      if(this.pending == this.page() && scrollAt() > 90){
        this.pending = this.page() + 1
        this.pushEvent("load-more", {})
      }
    })
  },
  updated(){ this.pending = this.page() }
}
Also, hooks can now be defined as a subclass of ViewHook, if you prefer native classes:
import { LiveSocket, ViewHook } from "phoenix_live_view"

class MyHook extends ViewHook {
  mounted() {
    ...
  }
}

let liveSocket = new LiveSocket(..., {
  hooks: {
    MyHook
  }
})
Using @types/phoenix_live_view (not maintained by the Phoenix team) is not necessary any more.

  
    
  
  Phoenix.Component.portal


When designing reusable HTML components for UI elements like tooltips or dialogs, it is sometimes necessary to render a part of a component's template outside of the regular DOM hierarchy of that component, for example to prevent clipping due to CSS rules like overflow: hidden that are not controlled by the component itself. Modern browser APIs for rendering elements in the top layer can help in many cases, but if you cannot use those for whatever reasons, LiveView previously did not have a solution to solve that problem. In LiveView 1.1, we introduce a new Phoenix.Component.portal/1 component:
<%!-- in some nested LiveView or component --%>
<.portal id="my-element" target="body">
  <%!-- any content here will be teleported into the body tag --%>
</.portal>
Any element can be teleported, even LiveComponents and nested LiveViews, and any phx-* events from inside a portal will still be handled by the correct LiveView. This is similar to <Teleport> in Vue.js or createPortal in React.
As a demo, we created an example for implementing tooltips using Phoenix.Component.portal as a single-file Elixir script. When saved as portal.exs, you can execute it as elixir portal.exs and visit http://localhost:5001 in your browser.

  
    
  
  JS.ignore_attributes


Sometimes it is useful to prevent some attributes from being patched by LiveView. One example where this frequently came up is when using a native <dialog> or <details> element that is controlled by the open attribute, which is special in that it is actually set (and removed) by the browser. Previously, LiveView would remove those attributes on update and required additional patching, now you can simply call JS.ignore_attributes in a phx-mounted binding:
<details phx-mounted={JS.ignore_attributes(["open"])}>
  <summary>...</summary>
  ...
</details>

  
    
  
  Moving from Floki to LazyHTML


LiveView 1.1 moves to LazyHTML as the HTML engine used by LiveViewTest. LazyHTML is based on lexbor and allows the use of modern CSS selector features, like :is(), :has(), etc. to target elements. Lexbor's stated goal is to create output that "should match that of modern browsers, meeting industry specifications".
This is a mostly backwards compatible change. The only way in which this affects LiveView projects is when using Floki specific selectors (fl-contains, fl-icontains), which will not work any more in selectors passed to LiveViewTest's element/3 function. In most cases, the text_filter option of element/3 should be a sufficient replacement, which has been a feature since LiveView v0.12.0.
Note that in Phoenix versions prior to v1.8, the phx.gen.auth generator used the Floki specific fl-contains selector in its generated tests in two instances, so if you used the phx.gen.auth generator to scaffold your authentication solution, those tests will need to be adjusted when updating to LiveView v1.1. In both cases, changing to use the text_filter option is enough to get you going again:
 {:ok, _login_live, login_html} =
   lv
-  |> element(~s|main a:fl-contains("Sign up")|)
+  |> element("main a", "Sign up")
   |> render_click()
   |> follow_redirect(conn, ~p"<%= schema.route_prefix %>/register")
If you're using Floki itself in your tests through its API (Floki.parse_document, Floki.find, etc.), you are not required to rewrite them when you update to LiveView v1.1.

  
    
  
  Slot and line annotations


When :debug_heex_annotations is enabled, LiveView will now annotate the beginning and end of each slot. A new :debug_tags_location has also been added, which adds the starting line of each tag. The goal is to provide more precise information to tools.
To enable this, a new callback called annotate_slot/4 was added. Custom implementations of Phoenix.LiveView.TagEngine must implement it accordingly.

  
    
  
  v1.1.0-rc.0 (2025-06-17)



  
    
  
  Enhancements


	Add type annotations to all public JavaScript APIs (#3789)
	Add Phoenix.LiveView.JS.ignore_attributes/1 to allow marking specific attributes to be ignored when LiveView patches an element (#3765)
	Add Phoenix.LiveView.Debug module with functions for inspecting LiveViews at runtime (#3776)
	Add Phoenix.LiveView.ColocatedHook and Phoenix.LiveView.ColocatedJS (#3810)
	Add :update_only option to Phoenix.LiveView.stream_insert/4 (#3573)
	Use LazyHTML instead of Floki internally for LiveViewTest
	Normalize whitespace in LiveViewTest's text filters (#3621)
	Raise by default when LiveViewTest detects duplicate DOM or LiveComponent IDs. This can be changed by passing on_error to Phoenix.LiveViewTest.live/3 / Phoenix.LiveViewTest.live_isolated/3
	Raise an exception when trying to bind a single DOM element to multiple views (this could happen when accidentally loading your app.js twice) (#3805)
	Ensure promise rejections include stack traces (#3738)
	Treat form associated custom elements as form inputs (3823)
	Add :inline_matcher option to Phoenix.LiveView.HTMLFormatter which can be configured as a list of strings and regular expressions to match against tag names to treat them as inline (#3795)


  
    
  
  v1.0


The CHANGELOG for v1.0 and earlier releases can be found in the v1.0 branch.


  

    
Welcome
    

Welcome to Phoenix LiveView documentation. Phoenix LiveView enables
rich, real-time user experiences with server-rendered HTML. A general
overview of LiveView and its benefits is available in our README.

  
    
  
  What is a LiveView?


LiveViews are processes that receive events, update their state,
and render updates to a page as diffs.
The LiveView programming model is declarative: instead of saying
"once event X happens, change Y on the page", events in LiveView
are regular messages which may cause changes to the state. Once
the state changes, the LiveView will re-render the relevant parts of
its HTML template and push it to the browser, which updates the page
in the most efficient manner.
LiveView state is nothing more than functional and immutable
Elixir data structures. The events are either internal application messages
(usually emitted by Phoenix.PubSub) or sent by the client/browser.
Every LiveView is first rendered statically as part of a regular
HTTP request, which provides quick times for "First Meaningful
Paint", in addition to helping search and indexing engines.
A persistent connection is then established between the client and
server. This allows LiveView applications to react faster to user
events as there is less work to be done and less data to be sent
compared to stateless requests that have to authenticate, decode, load,
and encode data on every request.

  
    
  
  Example


LiveView is included by default in Phoenix applications.
Therefore, to use LiveView, you must have already installed Phoenix
and created your first application. If you haven't done so,
check Phoenix' installation guide
to get started.
The behaviour of a LiveView is outlined by a module which implements
a series of functions as callbacks. Let's see an example. Write the
file below to lib/my_app_web/live/thermostat_live.ex:
defmodule MyAppWeb.ThermostatLive do
  use MyAppWeb, :live_view

  def render(assigns) do
    ~H"""
    Current temperature: {@temperature}°F
    <button phx-click="inc_temperature">+</button>
    """
  end

  def mount(_params, _session, socket) do
    temperature = 70 # Let's assume a fixed temperature for now
    {:ok, assign(socket, :temperature, temperature)}
  end

  def handle_event("inc_temperature", _params, socket) do
    {:noreply, update(socket, :temperature, &(&1 + 1))}
  end
end
The module above defines three functions (they are callbacks
required by LiveView). The first one is render/1,
which receives the socket assigns and is responsible for returning
the content to be rendered on the page. We use the ~H sigil to define
a HEEx template, which stands for HTML+EEx. They are an extension of
Elixir's builtin EEx templates, with support for HTML validation, syntax-based
components, smart change tracking, and more. You can learn more about
the template syntax in Phoenix.Component.sigil_H/2 (note
Phoenix.Component is automatically imported when you use Phoenix.LiveView).
The data used on rendering comes from the mount callback. The
mount callback is invoked when the LiveView starts. In it, you
can access the request parameters, read information stored in the
session (typically information which identifies who is the current
user), and a socket. The socket is where we keep all state, including
assigns. mount proceeds to assign a default temperature to the socket.
Because Elixir data structures are immutable, LiveView APIs often
receive the socket and return an updated socket. Then we return
{:ok, socket} to signal that we were able to mount the LiveView
successfully. After mount, LiveView will render the page with the
values from assigns and send it to the client.
If you look at the HTML rendered, you will notice there is a button
with a phx-click attribute. When the button is clicked, a
"inc_temperature" event is sent to the server, which is matched and
handled by the handle_event callback. This callback updates the socket
and returns {:noreply, socket} with the updated socket.
handle_* callbacks in LiveView (and in Elixir in general) are
invoked based on some action, in this case, the user clicking a button.
The {:noreply, socket} return means there is no additional replies
sent to the browser, only that a new version of the page is rendered.
LiveView then computes diffs and sends them to the client.
Now we are ready to render our LiveView. You can serve the LiveView
directly from your router:
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  pipeline :browser do
    ...
  end

  scope "/", MyAppWeb do
    pipe_through :browser
    ...

    live "/thermostat", ThermostatLive
  end
end
Once the LiveView is rendered, a regular HTML response is sent. In your
app.js file, you should find the following:
import {Socket} from "phoenix"
import {LiveSocket} from "phoenix_live_view"

let csrfToken = document.querySelector("meta[name='csrf-token']").getAttribute("content")
let liveSocket = new LiveSocket("/live", Socket, {params: {_csrf_token: csrfToken}})
liveSocket.connect()
Now the JavaScript client will connect over WebSockets and mount/3 will be invoked
inside a spawned LiveView process.

  
    
  
  Parameters and session


The mount callback receives three arguments: the request parameters, the session, and the socket.
The parameters can be used to read information from the URL. For example, assuming you have a Thermostat module defined somewhere that can read this information based on the house name, you could write this:
def mount(%{"house" => house}, _session, socket) do
  temperature = Thermostat.get_house_reading(house)
  {:ok, assign(socket, :temperature, temperature)}
end
And then in your router:
live "/thermostat/:house", ThermostatLive
The session retrieves information from a signed (or encrypted) cookie. This is where you can store authentication information, such as current_user_id:
def mount(_params, %{"current_user_id" => user_id}, socket) do
  temperature = Thermostat.get_user_reading(user_id)
  {:ok, assign(socket, :temperature, temperature)}
end
Phoenix comes with built-in authentication generators. See mix phx.gen.auth.

Most times, in practice, you will use both:
def mount(%{"house" => house}, %{"current_user_id" => user_id}, socket) do
  temperature = Thermostat.get_house_reading(user_id, house)
  {:ok, assign(socket, :temperature, temperature)}
end
In other words, you want to read the information about a given house, as long as the user has access to it.

  
    
  
  Bindings


Phoenix supports DOM element bindings for client-server interaction. For
example, to react to a click on a button, you would render the element:
<button phx-click="inc_temperature">+</button>
Then on the server, all LiveView bindings are handled with the handle_event/3
callback, for example:
def handle_event("inc_temperature", _value, socket) do
  {:noreply, update(socket, :temperature, &(&1 + 1))}
end
To update UI state, for example, to open and close dropdowns, switch tabs,
etc, LiveView also supports JS commands (Phoenix.LiveView.JS), which
execute directly on the client without reaching the server. To learn more,
see our bindings page for a complete list of all LiveView
bindings as well as our JavaScript interoperability guide.
LiveView has built-in support for forms, including uploads and association
management. See Phoenix.Component.form/1 as a starting point and
Phoenix.Component.inputs_for/1 for working with associations.
The Uploads and Form bindings guides provide
more information about advanced features.

  
    
  
  Navigation


LiveView provides functionality to allow page navigation using the
browser's pushState API.
With live navigation, the page is updated without a full page reload.
You can either patch the current LiveView, updating its URL, or
navigate to a new LiveView. You can learn more about them in the
Live Navigation guide.

  
    
  
  Generators


Phoenix v1.6 and later includes code generators for LiveView. If you want to see
an example of how to structure your application, from the database all the way up
to LiveViews, run the following:
$ mix phx.gen.live Blog Post posts title:string body:text

For more information, run mix help phx.gen.live.
For authentication, with built-in LiveView support, run mix phx.gen.auth Account User users.

  
    
  
  Compartmentalize state, markup, and events in LiveView


LiveView supports two extension mechanisms: function components, provided by
HEEx templates, and stateful components, known as LiveComponents.

  
    
  
  Function components to organize markup and event handling


Similar to render(assigns) in our LiveView, a function component is any
function that receives an assigns map and returns a ~H template. For example:
def weather_greeting(assigns) do
  ~H"""
  <div title="My div" class={@class}>
    <p>Hello {@name}</p>
    <MyApp.Weather.city name="Kraków"/>
  </div>
  """
end
You can learn more about function components in the Phoenix.Component
module. At the end of the day, they are a useful mechanism for code organization
and to reuse markup in your LiveViews.
Sometimes you need to share more than just markup across LiveViews. When you also
want to move events to a separate module, or use the same event handler in multiple
places, function components can be paired with
Phoenix.LiveView.attach_hook/4.

  
    
  
  Live components to encapsulate additional state


A component will occasionally need control over not only its own events,
but also its own separate state. For these cases, LiveView
provides Phoenix.LiveComponent, which are rendered using
live_component/1:
<.live_component module={UserComponent} id={user.id} user={user} />
LiveComponents have their own mount/1 and handle_event/3 callbacks, as well
as their own state with change tracking support, similar to LiveViews. They are
lightweight since they "run" in the same process as the parent LiveView, but
are more complex than function components themselves. Given they all run in the
same process, errors in components cause the whole view to fail to render.
For a complete rundown, see Phoenix.LiveComponent.
When in doubt over Functional components or live components?, default to the former.
Rely on the latter only when you need the additional state.

  
    
  
  live_render/3 to encapsulate state (with error isolation)


Finally, if you want complete isolation between parts of a LiveView, you can
always render a LiveView inside another LiveView by calling
live_render/3. This child LiveView
runs in a separate process than the parent, with its own callbacks. If a child
LiveView crashes, it won't affect the parent. If the parent crashes, all children
are terminated.
When rendering a child LiveView, the :id option is required to uniquely
identify the child. A child LiveView will only ever be rendered and mounted
a single time, provided its ID remains unchanged. To force a child to re-mount
with new session data, a new ID must be provided.
Given that it runs in its own process, a nested LiveView is an excellent tool
for creating completely isolated UI elements, but it is a slightly expensive
abstraction if all you want is to compartmentalize markup or events (or both).

  
    
  
  Summary


	use Phoenix.Component for code organization and reusing markup (optionally with attach_hook/4 for event handling reuse)
	use Phoenix.LiveComponent for sharing state, markup, and events between LiveViews
	use nested Phoenix.LiveView to compartmentalize state, markup, and events (with error isolation)


  
    
  
  Guides


This documentation is split into two categories. We have the API
reference for all LiveView modules, that's where you will learn
more about Phoenix.Component, Phoenix.LiveView, and so on.
LiveView also has many guides to help you on your journey,
split on server-side and client-side:

  
    
  
  Server-side


These guides focus on server-side functionality:
	Assigns and HEEx templates
	Deployments and recovery
	Error and exception handling
	Gettext for internationalization
	Live layouts
	Live navigation
	Security considerations
	Telemetry
	Uploads


  
    
  
  Client-side


These guides focus on LiveView bindings and client-side integration:
	Bindings
	External uploads
	Form bindings
	JavaScript interoperability
	Syncing changes and optimistic UIs



  

    
Assigns and HEEx templates
    

All of the data in a LiveView is stored in the socket, which is a server
side struct called Phoenix.LiveView.Socket. Your own data is stored
under the assigns key of said struct. The server data is never shared
with the client beyond what your template renders.
Phoenix template language is called HEEx (HTML+EEx). EEx is Embedded
Elixir, an Elixir string template engine. Those templates
are either files with the .heex extension or they are created
directly in source files via the ~H sigil. You can learn more about
the HEEx syntax by checking the docs for the ~H sigil.
The Phoenix.Component.assign/2 and Phoenix.Component.assign/3
functions help store those values. Those values can be accessed
in the LiveView as socket.assigns.name but they are accessed
inside HEEx templates as @name.
In this section, we are going to cover how LiveView minimizes
the payload over the wire by understanding the interplay between
assigns and templates.

  
    
  
  Change tracking


When you first render a .heex template, it will send all of the
static and dynamic parts of the template to the client. Imagine the
following template:
<h1>{expand_title(@title)}</h1>
It has two static parts, <h1> and </h1> and one dynamic part
made of expand_title(@title). Further rendering of this template
won't resend the static parts and it will only resend the dynamic
part if it changes.
The tracking of changes is done via assigns. If the @title assign
changes, then LiveView will execute the dynamic parts of the template,
expand_title(@title), and send the new content. If @title is the same,
nothing is executed and nothing is sent.
Change tracking also works when accessing map/struct fields.
Take this template:
<div id={"user_#{@user.id}"}>
  {@user.name}
</div>
If the @user.name changes but @user.id doesn't, then LiveView
will re-render only @user.name and it will not execute or resend @user.id
at all.
The change tracking also works when rendering other templates as
long as they are also .heex templates:
{render("child_template.html", assigns)}
Or when using function components:
<.show_name name={@user.name} />
The assign tracking feature also implies that you MUST avoid performing
direct operations in the template. For example, if you perform a database
query in your template:
<%= for user <- Repo.all(User) do %>
  {user.name}
<% end %>
Then Phoenix will never re-render the section above, even if the number of
users in the database changes. Instead, you need to store the users as
assigns in your LiveView before it renders the template:
assign(socket, :users, Repo.all(User))
Generally speaking, data loading should never happen inside the template,
regardless if you are using LiveView or not. The difference is that LiveView
enforces this best practice.

  
    
  
  Common pitfalls


There are some common pitfalls to keep in mind when using the ~H sigil
or .heex templates inside LiveViews.

  
    
  
  Variables


Due to the scope of variables, LiveView has to disable change tracking
whenever variables are used in the template, with the exception of
variables introduced by Elixir block constructs such as case,
for, if, and others. Therefore, you must avoid code like
this in your HEEx templates:
<% some_var = @x + @y %>
{some_var}
Instead, use a function:
{sum(@x, @y)}
Similarly, do not define variables at the top of your render function
for LiveViews or LiveComponents. Since LiveView cannot track sum or title,
if either value changes, both must be re-rendered by LiveView.
def render(assigns) do
  sum = assigns.x + assigns.y
  title = assigns.title

  ~H"""
  <h1>{title}</h1>

  {sum}
  """
end
Instead use the assign/2, assign/3, assign_new/3, and update/3
functions to compute it. Any assign defined or updated this way will be marked as
changed, while other assigns like @title will still be tracked by LiveView.
assign(assigns, sum: assigns.x + assigns.y)
The same functions can be used inside function components too:
attr :x, :integer, required: true
attr :y, :integer, required: true
attr :title, :string, required: true
def sum_component(assigns) do
  assigns = assign(assigns, sum: assigns.x + assigns.y)

  ~H"""
  <h1>{@title}</h1>

  {@sum}
  """
end
Generally speaking, avoid accessing variables inside HEEx templates, as code that
access variables is always executed on every render. The exception are variables
introduced by Elixir's block constructs, such as if and for comprehensions.
For example, accessing the post variable defined by the comprehension below
works as expected:
<%= for post <- @posts do %>
  ...
<% end %>

  
    
  
  The assigns variable


When talking about variables, it is also worth discussing the assigns
special variable. Every time you use the ~H sigil, you must define an
assigns variable, which is also available on every .heex template.
However, we must avoid accessing this variable directly inside templates
and instead use @ for accessing specific keys. This also applies to
function components. Let's see some examples.
Sometimes you might want to pass all assigns from one function component to
another. For example, imagine you have a complex card component with
header, content and footer section. You might refactor your component
into three smaller components internally:
def card(assigns) do
  ~H"""
  <div class="card">
    <.card_header {assigns} />
    <.card_body {assigns} />
    <.card_footer {assigns} />
  </div>
  """
end

defp card_header(assigns) do
  ...
end

defp card_body(assigns) do
  ...
end

defp card_footer(assigns) do
  ...
end
Because of the way function components handle attributes, the above code will
not perform change tracking and it will always re-render all three components
on every change.
Generally, you should avoid passing all assigns and instead be explicit about
which assigns the child components need:
def card(assigns) do
  ~H"""
  <div class="card">
    <.card_header title={@title} class={@title_class} />
    <.card_body>
      {render_slot(@inner_block)}
    </.card_body>
    <.card_footer on_close={@on_close} />
  </div>
  """
end
If you really need to pass all assigns you should instead use the regular
function call syntax. This is the only case where accessing assigns inside
templates is acceptable:
def card(assigns) do
  ~H"""
  <div class="card">
    {card_header(assigns)}
    {card_body(assigns)}
    {card_footer(assigns)}
  </div>
  """
end
This ensures that the change tracking information from the parent component
is passed to each child component, only re-rendering what is necessary.
However, generally speaking, it is best to avoid passing assigns altogether
and instead let LiveView figure out the best way to track changes.

  
    
  
  Comprehensions


HEEx supports comprehensions in templates, which is a way to traverse lists
and collections. For example:
<%= for post <- @posts do %>
  <section>
    <h1>{expand_title(post.title)}</h1>
  </section>
<% end %>
Or using the special :for attribute:
<section :for={post <- @posts}>
  <h1>{expand_title(post.title)}</h1>
</section>
Comprehensions in templates are optimized so the static parts of
a comprehension are only sent once, regardless of the number of items.
However, keep in mind LiveView does not track changes within the
collection given to the comprehension. In other words, if one entry
in @posts changes, all posts are sent again.
There are two common solutions to this problem.
The first one is to also provide a :key expression:
<section :for={post <- @posts} :key={post.id}>
  <h1>{expand_title(post.title)}</h1>
</section>
This is functionally equivalent to doing:
<section :for={post <- @posts}>
  <.live_component module={PostComponent} id={"post-#{post.id}"} post={post} />
</section>
Since LiveComponents have their own assigns, LiveComponents would allow
you to perform change tracking for each item. If the @posts variable
changes, the client will simply send a list of component IDs (which are
integers) and only the data for the posts that actually changed.
You can read more about :key in the documentation for sigil_H/2.
The second solution is to use Phoenix.LiveView.stream/4, which gives you
precise control over how elements are added, removed, and updated. Streams
are particularly useful when you don't need to keep the collection in memory,
allowing you to reduce the data sent over the wire and the server memory
usage.

  
    
  
  Summary


To sum up:
	Avoid defining local variables inside HEEx templates, except within Elixir's constructs

	Avoid passing or accessing the assigns variable inside HEEx templates




  

    
Deployments and recovery
    

One of the questions that arise from LiveView stateful model is what considerations are necessary when deploying a new version of LiveView (or when recovering from an error).
First off, whenever LiveView disconnects, it will automatically attempt to reconnect to the server using exponential back-off. This means it will try immediately, then wait 2s and try again, then 5s and so on. If you are deploying, this typically means the next reconnection will immediately succeed and your load balancer will automatically redirect to the new servers.
However, your LiveView may still have state that will be lost in this transition. How to deal with it? The good news is that there are a series of practices you can follow that will not only help with deployments but it will improve your application in general.
	Keep state in the query parameters when appropriate. For example, if your application has tabs and the user clicked a tab, instead of using phx-click and Phoenix.LiveView.handle_event/3 to manage it, you should implement it using <.link patch={...}> passing the tab name as parameter. You will then receive the new tab name Phoenix.LiveView.handle_params/3 which will set the relevant assign to choose which tab to display. You can even define specific URLs for each tab in your application router. By doing this, you will reduce the amount of server state, make tab navigation shareable via links, improving search engine indexing, and more.

	Consider storing other relevant state in the database. For example, if you are building a chat app and you want to store which messages have been read, you can store so in the database. Once the page is loaded, you retrieve the index of the last read message. This makes the application more robust, allow data to be synchronized across devices, etc.

	If your application uses forms (which is most likely the case), keep in mind that Phoenix performs automatic form recovery: in case of disconnections, Phoenix will collect the form data and resubmit it on reconnection. This mechanism works out of the box for most forms but you may want to customize it or test it for your most complex forms. See the relevant section in the "Form bindings" document to learn more.


The idea is that: if you follow the practices above, most of your state is already handled within your app and therefore deployments should not bring additional concerns. Not only that, it will bring overall benefits to your app such as indexing, link sharing, device sharing, and so on.
If you really have complex state that cannot be immediately handled, then you may need to resort to special strategies. This may be persisting "old" state to Redis/S3/Database and loading the new state on the new connections. Or you may take special care when migrating connections (for example, if you are building a game, you may want to wait for on-going sessions to finish before turning down the old server while routing new sessions to the new ones). Such cases will depend on your requirements (and they would likely exist regardless of which application stack you are using).


  

    
Error and exception handling
    

As with any other Elixir code, exceptions may happen during the LiveView
life-cycle. This page describes how LiveView handles errors at different
stages.

  
    
  
  Expected scenarios


In this section, we will talk about error cases that you expect to happen
within your application. For example, a user filling in a form with invalid
data is expected. In a LiveView, we typically handle those cases by storing
the form state in LiveView assigns and rendering any relevant error message
back to the client.
We may also use flash messages for this. For example, imagine you have a
page to manage all "Team members" in an organization. However, if there is
only one member left in the organization, they should not be allowed to
leave. You may want to handle this by using flash messages:
if MyApp.Org.leave(socket.assigns.current_org, member) do
  {:noreply, socket}
else
  {:noreply, put_flash(socket, :error, "last member cannot leave organization")}
end
However, one may argue that, if the last member of an organization cannot
leave it, it may be better to not even show the "Leave" button in the UI
when the organization has only one member.
Given the button does not appear in the UI, triggering the "leave" action when
the organization has only one member is an unexpected scenario. This means we
can rewrite the code above to:
true = MyApp.Org.leave(socket.assigns.current_org, member)
{:noreply, socket}
If leave does not return true, Elixir will raise a MatchError
exception. Or you could provide a leave! function that raises a specific
exception:
MyApp.Org.leave!(socket.assigns.current_org, member)
{:noreply, socket}
However, what will happen with a LiveView in case of exceptions?
Let's talk about unexpected scenarios.

  
    
  
  Unexpected scenarios


Elixir developers tend to write assertive code. This means that, if we
expect leave to always return true, we can explicitly match on its
result, as we did above:
true = MyApp.Org.leave(socket.assigns.current_org, member)
{:noreply, socket}
If leave fails and returns false, an exception is raised. It is common
for Elixir developers to use exceptions for unexpected scenarios in their
Phoenix applications.
For example, if you are building an application where a user may belong to
one or more organizations, when accessing the organization page, you may want to
check that the user has access to it like this:
organizations_query = Ecto.assoc(socket.assigns.current_user, :organizations)
Repo.get!(organizations_query, params["org_id"])
The code above builds a query that returns all organizations that belongs to
the current user and then validates that the given org_id belongs to the
user. If there is no such org_id or if the user has no access to it,
Repo.get! will raise an Ecto.NoResultsError exception.
During a regular controller request, this exception will be converted to a
404 exception and rendered as a custom error page, as
detailed here.
LiveView will react to exceptions in three different ways, depending on
where it is in its life-cycle.

  
    
  
  Exceptions during HTTP mount


When you first access a LiveView, a regular HTTP request is sent to the server
and processed by the LiveView. The mount callback is invoked and then a page
is rendered. Any exception here is caught, logged, and converted to an exception
page by Phoenix error views - exactly how it works with controllers too.

  
    
  
  Exceptions during connected mount


If the initial HTTP request succeeds, LiveView will connect to the server
using a stateful connection, typically a WebSocket. This spawns a long-running
lightweight Elixir process on the server, which invokes the mount callback
and renders an updated version of the page.
An exception during this stage will crash the LiveView process, which will be logged.
Once the client notices the crash, it fully reloads the page. This will cause mount
to be invoked again during a regular HTTP request (the exact scenario of the previous
subsection).
In other words, LiveView will reload the page in case of errors, making it
fail as if LiveView was not involved in the rendering in the first place.

  
    
  
  Exceptions after connected mount


Once your LiveView is mounted and connected, any error will cause the LiveView process
to crash and be logged. Once the client notices the error, it will remount the LiveView
over the stateful connection, without reloading the page (the exact scenario of the
previous subsection). If remounting succeeds, the LiveView goes back to a working
state, updating the page and showing the user the latest information.
For example, let's say two users try to leave the organization at the same time.
In this case, both of them see the "Leave" button, but our leave function call
will succeed only for one of them:
true = MyApp.Org.leave(socket.assigns.current_org, member)
{:noreply, socket}
When the exception raises, the client will remount the LiveView. Once you remount,
your code will now notice that there is only one user in the organization and
therefore no longer show the "Leave" button. In other words, by remounting,
we often update the state of the page, allowing exceptions to be automatically
handled.
Note that the choice between conditionally checking on the result of the leave
function with an if, or simply asserting it returns true, is completely
up to you. If the likelihood of everyone leaving the organization at the same
time is low, then you may as well treat it as an unexpected scenario. Although
other developers will be more comfortable by explicitly handling those cases.
In both scenarios, LiveView has you covered.
Finally, if your LiveView crashes, its current state will be lost. Luckily,
LiveView has a series of mechanisms and best practices you can follow to ensure
the user is shown the same page as before during reconnections. See the
"Deployments and recovery" guide for more information.


  

    
Gettext for internationalization
    

For internationalization with gettext,
you must call Gettext.put_locale/2 on the LiveView mount callback to instruct
the LiveView which locale should be used for rendering the page.
However, one question that has to be answered is how to retrieve the locale in
the first place. There are many approaches to solve this problem:
	The locale could be stored in the URL as a parameter
	The locale could be stored in the session
	The locale could be stored in the database

We will briefly cover these approaches to provide some direction.

  
    
  
  Locale from parameters


You can say all URLs have a locale parameter. In your router:
scope "/:locale" do
  live ...
  get ...
end
Accessing a page without a locale should automatically redirect
to a URL with locale (the best locale could be fetched from
HTTP headers, which is outside of the scope of this guide).
Then, assuming all URLs have a locale, you can set the Gettext
locale accordingly:
def mount(%{"locale" => locale}, _session, socket) do
  Gettext.put_locale(MyApp.Gettext, locale)
  {:ok, socket}
end
You can also use the on_mount hook to
automatically restore the locale for every LiveView in your application:
defmodule MyAppWeb.RestoreLocale do
  def on_mount(:default, %{"locale" => locale}, _session, socket) do
    Gettext.put_locale(MyApp.Gettext, locale)
    {:cont, socket}
  end

  # catch-all case
  def on_mount(:default, _params, _session, socket), do: {:cont, socket}
end
Then, add this hook to def live_view under MyAppWeb, to run it on all
LiveViews by default:
def live_view do
  quote do
    use Phoenix.LiveView

    on_mount MyAppWeb.RestoreLocale
    unquote(view_helpers())
  end
end
Note that, because the Gettext locale is not stored in the assigns, if you
want to change the locale, you must use <.link navigate={...} />, instead
of simply patching the page.

  
    
  
  Locale from session


You may also store the locale in the Plug session. For example, in a controller
you might do:
def put_user_session(conn, current_user) do
  Gettext.put_locale(MyApp.Gettext, current_user.locale)

  conn
  |> put_session(:user_id, current_user.id)
  |> put_session(:locale, current_user.locale)
end
and then restore the locale from session within your LiveView mount:
def mount(_params, %{"locale" => locale}, socket) do
  Gettext.put_locale(MyApp.Gettext, locale)
  {:ok, socket}
end
You can also encapsulate this in a hook, as done in the previous section.
However, if the locale is stored in the session, you can only change it
by using regular controller requests. Therefore you should always use
<.link to={...} /> to point to a controller that change the session
accordingly, reloading any LiveView.

  
    
  
  Locale from database


You may also allow users to store their locale configuration in the database.
Then, on mount/3, you can retrieve the user id from the session and load
the locale:
def mount(_params, %{"user_id" => user_id}, socket) do
  user = Users.get_user!(user_id)
  Gettext.put_locale(MyApp.Gettext, user.locale)
  {:ok, socket}
end
In practice, you may end-up mixing more than one approach listed here.
For example, reading from the database is great once the user is logged in
but, before that happens, you may need to store the locale in the session
or in the URL.
Similarly, you can keep the locale in the URL, but change the URL accordingly
to the user preferred locale once they sign in. Hopefully this guide gives
some suggestions on how to move forward and explore the best approach for your
application.


  

    
Live layouts
    

Your LiveView applications can be made of two layouts:
	the root layout - this layout typically contains the <html>
definition alongside the head and body tags. Any content defined
in the root layout will remain the same, even as you live navigate
across LiveViews. The root layout is typically declared on the
router with put_root_layout and defined as "root.html.heex"
in your layouts folder. It calls {@inner_content} to inject the
content rendered by the layout

	the app layout - this is the dynamic layout part of your application,
it often includes the menu, sidebar, flash messages, and more.
From Phoenix v1.8, this layout is explicitly rendered in your templates
by calling the <Layouts.app /> component. In Phoenix v1.7 and earlier,
the layout was typically configured as part of the lib/my_app_web.ex
file, such as use Phoenix.LiveView, layout: ...


Overall, those layouts are found in components/layouts and are
embedded within MyAppWeb.Layouts.

  
    
  
  Root layout


The "root" layout is rendered only on the initial request and
therefore it has access to the @conn assign. The root layout
is typically defined in your router:
plug :put_root_layout, html: {MyAppWeb.Layouts, :root}
The root layout can also be set via the :root_layout option
in your router via Phoenix.LiveView.Router.live_session/2.

  
    
  
  Updating document title


Because the root layout from the Plug pipeline is rendered outside of
LiveView, the contents cannot be dynamically changed. The one exception
is the <title> of the HTML document. Phoenix LiveView special cases
the @page_title assign to allow dynamically updating the title of the
page, which is useful when using live navigation, or annotating the browser
tab with a notification. For example, to update the user's notification
count in the browser's title bar, first set the page_title assign on
mount:
def mount(_params, _session, socket) do
  socket = assign(socket, page_title: "Latest Posts")
  {:ok, socket}
end
Then access @page_title in the root layout:
<title>{@page_title}</title>
You can also use the Phoenix.Component.live_title/1 component to support
adding automatic prefix and suffix to the page title when rendered and
on subsequent updates:
<Phoenix.Component.live_title default="Welcome" prefix="MyApp – ">
  {assigns[:page_title]}
</Phoenix.Component.live_title>
Although the root layout is not updated by LiveView, by simply assigning
to page_title, LiveView knows you want the title to be updated:
def handle_info({:new_messages, count}, socket) do
  {:noreply, assign(socket, page_title: "Latest Posts (#{count} new)")}
end
Note: If you find yourself needing to dynamically patch other parts of the
base layout, such as injecting new scripts or styles into the <head> during
live navigation, then a regular, non-live, page navigation should be used
instead. Assigning the @page_title updates the document.title directly,
and therefore cannot be used to update any other part of the base layout.


  

    
Live navigation
    

LiveView provides functionality to allow page navigation using the
browser's pushState API.
With live navigation, the page is updated without a full page reload.
You can trigger live navigation in two ways:
	From the client - this is done by passing either patch={url} or navigate={url}
to the Phoenix.Component.link/1 component.

	From the server - this is done by Phoenix.LiveView.push_patch/2 or Phoenix.LiveView.push_navigate/2.


For example, instead of writing the following in a template:
<.link href={~p"/pages/#{@page + 1}"}>Next</.link>
You would write:
<.link patch={~p"/pages/#{@page + 1}"}>Next</.link>
Or in a LiveView:
{:noreply, push_patch(socket, to: ~p"/pages/#{@page + 1}")}
The "patch" operations must be used when you want to navigate to the
current LiveView, simply updating the URL and the current parameters,
without mounting a new LiveView. When patch is used, the
handle_params/3 callback is
invoked and the minimal set of changes are sent to the client.
See the next section for more information.
The "navigate" operations must be used when you want to dismount the
current LiveView and mount a new one. You can only "navigate" between
LiveViews in the same session. While redirecting, a phx-loading class
is added to the LiveView, which can be used to indicate to the user a
new page is being loaded.
If you attempt to patch to another LiveView or navigate across live sessions,
a full page reload is triggered. This means your application continues to work,
in case your application structure changes and that's not reflected in the navigation.
Here is a quick breakdown:
	<.link href={...}> and redirect/2
are HTTP-based, work everywhere, and perform full page reloads

	<.link navigate={...}> and push_navigate/2
work across LiveViews in the same session. They mount a new LiveView
while keeping the current layout

	<.link patch={...}> and push_patch/2
updates the current LiveView and sends only the minimal diff while also
maintaining the scroll position



  
    
  
  handle_params/3


The handle_params/3 callback is invoked
after mount/3 and before the initial render.
It is also invoked every time <.link patch={...}>
or push_patch/2 are used.
It receives the request parameters as first argument, the url as second,
and the socket as third.
For example, imagine you have a UserTable LiveView to show all users in
the system and you define it in the router as:
live "/users", UserTable
Now to add live sorting, you could do:
<.link patch={~p"/users?sort_by=name"}>Sort by name</.link>
When clicked, since we are navigating to the current LiveView,
handle_params/3 will be invoked.
Remember you should never trust the received params, so you must use the callback to
validate the user input and change the state accordingly:
def handle_params(params, _uri, socket) do
  socket =
    case params["sort_by"] do
      sort_by when sort_by in ~w(name company) -> assign(socket, sort_by: sort_by)
      _ -> socket
    end

  {:noreply, load_users(socket)}
end
Note we returned {:noreply, socket}, where :noreply means no
additional information is sent to the client. As with other handle_*
callbacks, changes to the state inside
handle_params/3 will trigger
a new server render.
Note the parameters given to handle_params/3
are the same as the ones given to mount/3.
So how do you decide which callback to use to load data?
Generally speaking, data should always be loaded on mount/3,
since mount/3 is invoked once per LiveView life-cycle.
Only the params you expect to be changed via
<.link patch={...}> or
push_patch/2 must be loaded on
handle_params/3.
For example, imagine you have a blog. The URL for a single post is:
"/blog/posts/:post_id". In the post page, you have comments and they are paginated.
You use <.link patch={...}> to update the shown
comments every time the user paginates, updating the URL to "/blog/posts/:post_id?page=X".
In this example, you will access "post_id" on mount/3 and
the page of comments on handle_params/3.

  
    
  
  Replace page address


LiveView also allows the current browser URL to be replaced. This is useful when you
want certain events to change the URL but without polluting the browser's history.
This can be done by passing the <.link replace> option to any of the navigation helpers.

  
    
  
  Multiple LiveViews in the same page


LiveView allows you to have multiple LiveViews in the same page by calling
Phoenix.Component.live_render/3 in your templates. However, only
the LiveViews defined directly in your router can use the "Live Navigation"
functionality described here. This is important because LiveViews work
closely with your router, guaranteeing you can only navigate to known
routes.


  

    
Security considerations
    

LiveView begins its life-cycle as a regular HTTP request. Then a stateful
connection is established. Both the HTTP request and the stateful connection
receive the client data via parameters and session.
This means that any session validation must happen both in the HTTP request
(plug pipeline) and the stateful connection (LiveView mount).

  
    
  
  Authentication vs authorization


When speaking about security, there are two terms commonly used:
authentication and authorization. Authentication is about identifying
a user. Authorization is about telling if a user has access to a certain
resource or feature in the system.
In a regular web application, once a user is authenticated, for example by
entering their email and password, or by using a third-party service such as
Google, Twitter, or Facebook, a token identifying the user is stored in the
session, which is a cookie (a key-value pair) stored in the user's browser.
Every time there is a request, we read the value from the session, and, if
valid, we fetch the user stored in the session from the database. The session
is automatically validated by Phoenix and tools like mix phx.gen.auth can
generate the building blocks of an authentication system for you.
Once the user is authenticated, they may perform many actions on the page,
and some of those actions require specific permissions. This is called
authorization and the specific rules often change per application.
In a regular web application, we perform authentication and authorization
checks on every request. Given LiveViews start as a regular HTTP request,
they share the authentication logic with regular requests through plugs.
The request starts in your endpoint, which then invokes the router.
Plugs are used to ensure the user is authenticated and stores the
relevant information in the session.
Once the user is authenticated, we typically validate the sessions on
the mount callback. Authorization rules generally happen on mount
(for instance, is the user allowed to see this page?) and also on
handle_event (is the user allowed to delete this item?).

  
    
  
  live_session


The primary mechanism for grouping LiveViews is via the
Phoenix.LiveView.Router.live_session/2. LiveView will then ensure
that navigation events within the same live_session skip the regular
HTTP requests without going through the plug pipeline. Events across
live sessions will go through the router.
For example, imagine you need to authenticate two distinct types of users.
Your regular users login via email and password, and you have an admin
dashboard that uses HTTP auth. You can specify different live_sessions
for each authentication flow:
scope "/" do
  pipe_through [:authenticate_user]
  get ...

  live_session :default do
    live ...
  end
end

scope "/admin" do
  pipe_through [:http_auth_admin]
  get ...

  live_session :admin do
    live ...
  end
end
Now every time you try to navigate to an admin panel, and out of it,
a regular page navigation will happen and a brand new live connection
will be established.
It is worth remembering that LiveViews require their own security checks,
so we use pipe_through above to protect the regular routes (get, post, etc.)
and the LiveViews should run their own checks on the mount callback
(or using Phoenix.LiveView.on_mount/1 hooks).
For this purpose, you can combine live_session with on_mount, as well
as other options, such as the :root_layout. Instead of declaring on_mount
on every LiveView, you can declare it at the router level and it will enforce
it on all LiveViews under it:
scope "/" do
  pipe_through [:authenticate_user]

  live_session :default, on_mount: MyAppWeb.UserLiveAuth do
    live ...
  end
end

scope "/admin" do
  pipe_through [:authenticate_admin]

  live_session :admin, on_mount: MyAppWeb.AdminLiveAuth do
    live ...
  end
end
Each live route under the :default live_session will invoke
the MyAppWeb.UserLiveAuth hook on mount. This module was defined
earlier in this guide. We will also pipe regular web requests through
:authenticate_user, which must execute the same checks as
MyAppWeb.UserLiveAuth, but tailored to plug.
Similarly, the :admin live_session has its own authentication
flow, powered by MyAppWeb.AdminLiveAuth. It also defines a plug
equivalent named :authenticate_admin, which will be used by any
regular request. If there are no regular web requests defined under
a live session, then the pipe_through checks are not necessary.
Declaring the on_mount on live_session is exactly the same as
declaring it in each LiveView. Let's talk about which logic we typically
execute on mount.

  
    
  
  Mounting considerations


The mount/3 callback is invoked both on
the initial HTTP mount and when LiveView is connected. Therefore, any
authorization performed during mount will cover all scenarios.
Once the user is authenticated and stored in the session, the logic to fetch the user and further authorize its account needs to happen inside LiveView. For example, if you have the following plugs:
plug :ensure_user_authenticated
plug :ensure_user_confirmed
Then the mount/3 callback of your LiveView
should execute those same verifications:
def mount(_params, %{"user_id" => user_id} = _session, socket) do
  socket = assign(socket, current_user: Accounts.get_user!(user_id))

  socket =
    if socket.assigns.current_user.confirmed_at do
      socket
    else
      redirect(socket, to: "/login")
    end

  {:ok, socket}
end
The on_mount hook allows you to encapsulate this logic and execute it on every mount:
defmodule MyAppWeb.UserLiveAuth do
  import Phoenix.Component
  import Phoenix.LiveView
  alias MyAppWeb.Accounts # from `mix phx.gen.auth`

  def on_mount(:default, _params, %{"user_token" => user_token} = _session, socket) do
    socket =
      assign_new(socket, :current_user, fn ->
        Accounts.get_user_by_session_token(user_token)
      end)

    if socket.assigns.current_user.confirmed_at do
      {:cont, socket}
    else
      {:halt, redirect(socket, to: "/login")}
    end
  end
end
We use assign_new/3. This is a
convenience to avoid fetching the current_user multiple times across
parent-child LiveViews.
Now we can use the hook whenever relevant. One option is to specify
the hook in your router under live_session:
live_session :default, on_mount: MyAppWeb.UserLiveAuth do
  # Your routes
end
Alternatively, you can either specify the hook directly in the LiveView:
defmodule MyAppWeb.PageLive do
  use MyAppWeb, :live_view
  on_mount MyAppWeb.UserLiveAuth

  ...
end
If you prefer, you can add the hook to def live_view under MyAppWeb,
to run it on all LiveViews by default:
def live_view do
  quote do
    use Phoenix.LiveView

    on_mount MyAppWeb.UserLiveAuth
    unquote(html_helpers())
  end
end

  
    
  
  Events considerations


Every time the user performs an action on your system, you should verify if the user
is authorized to do so, regardless if you are using LiveViews or not. For example,
imagine a user can see all projects in a web application, but they may not have
permission to delete any of them. At the UI level, you handle this accordingly
by not showing the delete button in the projects listing, but a savvy user can
directly talk to the server and request a deletion anyway. For this reason, you
must always verify permissions on the server.
In LiveView, most actions are handled by the handle_event callback. Therefore,
you typically authorize the user within those callbacks. In the scenario just
described, one might implement this:
on_mount MyAppWeb.UserLiveAuth

def mount(_params, _session, socket) do
  {:ok, load_projects(socket)}
end

def handle_event("delete_project", %{"project_id" => project_id}, socket) do
  Project.delete!(socket.assigns.current_user, project_id)
  {:noreply, update(socket, :projects, &Enum.reject(&1, fn p -> p.id == project_id end)}
end

defp load_projects(socket) do
  projects = Project.all_projects(socket.assigns.current_user)
  assign(socket, projects: projects)
end
First, we used on_mount to authenticate the user based on the data stored in
the session. Then we load all projects based on the authenticated user. Now,
whenever there is a request to delete a project, we still pass the current user
as argument to the Project context, so it verifies if the user is allowed to
delete it or not. In case it cannot delete, it is fine to just raise an exception.
After all, users are not meant to trigger this code path anyway (unless they are
fiddling with something they are not supposed to!).

  
    
  
  Disconnecting all instances of a live user


So far, the security model between LiveView and regular web applications have
been remarkably similar. After all, we must always authenticate and authorize
every user. The main difference between them happens on logout or when revoking
access.
Because LiveView is a permanent connection between client and server, if a user
is logged out, or removed from the system, this change won't reflect on the
LiveView part unless the user reloads the page.
Luckily, it is possible to address this by setting a live_socket_id in the
session. For example, when logging in a user, you could do:
conn
|> put_session(:current_user_id, user.id)
|> put_session(:live_socket_id, "users_socket:#{user.id}")
Now all LiveView sockets will be identified and listen to the given live_socket_id.
You can then disconnect all live users identified by said ID by broadcasting on
the topic:
MyAppWeb.Endpoint.broadcast("users_socket:#{user.id}", "disconnect", %{})
Note: If you use mix phx.gen.auth to generate your authentication system,
lines to that effect are already present in the generated code. The generated
code uses a user_token instead of referring to the user_id.

Once a LiveView is disconnected, the client will attempt to reestablish
the connection and re-execute the mount/3
callback. In this case, if the user is no longer logged in or it no longer has
access to the current resource, mount/3 will fail and the user will be
redirected.
This is the same mechanism provided by Phoenix.Channels. Therefore, if
your application uses both channels and LiveViews, you can use the same
technique to disconnect any stateful connection.

  
    
  
  Summing up


The important concepts to keep in mind are:
	live_session can be used to draw boundaries between groups of
LiveViews. While you could use live_session to draw lines between
different authorization rules, doing so would lead to frequent page
reloads. For this reason, we typically use live_session to enforce
different authentication requirements or whenever you need to
change root layouts

	Your authentication logic (logging the user in) is typically part of
your regular web request pipeline and it is shared by both controllers
and LiveViews. Authentication then stores the user information in the
session. Regular web requests use plug to read the user from a session,
LiveViews read it inside an on_mount callback. This is typically a
single database lookup on both cases. Running mix phx.gen.auth sets
up all that is necessary

	Once authenticated, your authorization logic in LiveViews will happen
both during mount (such as "can the user see this page?") and during
events (like "can the user delete this item?"). Those rules are often
domain/business specific, and typically happen in your context modules.
This is also a requirement for regular requests and responses




  

    
Telemetry
    

LiveView currently exposes the following telemetry events:
	[:phoenix, :live_view, :mount, :start] - Dispatched by a Phoenix.LiveView
immediately before mount/3 is invoked.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  params: unsigned_params | :not_mounted_at_router,
  session: map,
  uri: String.t() | nil
}



	[:phoenix, :live_view, :mount, :stop] - Dispatched by a Phoenix.LiveView
when the mount/3 callback completes successfully.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  params: unsigned_params | :not_mounted_at_router,
  session: map,
  uri: String.t() | nil
}



	[:phoenix, :live_view, :mount, :exception] - Dispatched by a Phoenix.LiveView
when an exception is raised in the mount/3 callback.
	Measurement: %{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  params: unsigned_params | :not_mounted_at_router,
  session: map,
  uri: String.t() | nil
}



	[:phoenix, :live_view, :handle_params, :start] - Dispatched by a Phoenix.LiveView
immediately before handle_params/3 is invoked.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  params: unsigned_params,
  uri: String.t()
}



	[:phoenix, :live_view, :handle_params, :stop] - Dispatched by a Phoenix.LiveView
when the handle_params/3 callback completes successfully.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  params: unsigned_params,
  uri: String.t()
}



	[:phoenix, :live_view, :handle_params, :exception] - Dispatched by a Phoenix.LiveView
when an exception is raised in the handle_params/3 callback.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  params: unsigned_params,
  uri: String.t()
}



	[:phoenix, :live_view, :handle_event, :start] - Dispatched by a Phoenix.LiveView
immediately before handle_event/3 is invoked.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_view, :handle_event, :stop] - Dispatched by a Phoenix.LiveView
when the handle_event/3 callback completes successfully.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_view, :handle_event, :exception] - Dispatched by a Phoenix.LiveView
when an exception is raised in the handle_event/3 callback.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_view, :render, :start] - Dispatched by a Phoenix.LiveView
immediately before render/1 is invoked.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  force?: boolean,
  changed?: boolean
}



	[:phoenix, :live_view, :render, :stop] - Dispatched by a Phoenix.LiveView
when the render/1 callback completes successfully.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  force?: boolean,
  changed?: boolean
}



	[:phoenix, :live_view, :render, :exception] - Dispatched by a Phoenix.LiveView
when an exception is raised in the render/1 callback.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  force?: boolean,
  changed?: boolean
}



	[:phoenix, :live_component, :update, :start] - Dispatched by a Phoenix.LiveComponent
immediately before update/2 or a
update_many/1 is invoked.
In the case ofupdate/2 it might dispatch one event
for multiple calls.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  component: atom,
  assigns_sockets: [{map(), Phoenix.LiveView.Socket.t}]
}



	[:phoenix, :live_component, :update, :stop] - Dispatched by a Phoenix.LiveComponent
when the update/2 or a
update_many/1 callback completes successfully.
In the case ofupdate/2 it might dispatch one event
for multiple calls. The sockets metadata contain the updated sockets.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  component: atom,
  assigns_sockets: [{map(), Phoenix.LiveView.Socket.t}],
  sockets: [Phoenix.LiveView.Socket.t]
}



	[:phoenix, :live_component, :update, :exception] - Dispatched by a Phoenix.LiveComponent
when an exception is raised in the update/2 or a
update_many/1 callback.
In the case ofupdate/2 it might dispatch one event
for multiple calls.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  component: atom,
  assigns_sockets: [{map(), Phoenix.LiveView.Socket.t}]
}



	[:phoenix, :live_component, :handle_event, :start] - Dispatched by a Phoenix.LiveComponent
immediately before handle_event/3 is invoked.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  component: atom,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_component, :handle_event, :stop] - Dispatched by a Phoenix.LiveComponent
when the handle_event/3 callback completes successfully.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  component: atom,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_component, :handle_event, :exception] - Dispatched by a Phoenix.LiveComponent
when an exception is raised in the handle_event/3 callback.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  component: atom,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_component, :destroyed] - Dispatched by a Phoenix.LiveComponent
after it is destroyed. No measurement.
	Metadata:
  %{
socket: Phoenix.LiveView.Socket.t,
component: atom,
cid: integer(),
live_view_socket: Phoenix.LiveView.Socket.t
  }






  

    
Uploads
    

LiveView supports interactive file uploads with progress for
both direct to server uploads as well as direct-to-cloud
external uploads on the client.

  
    
  
  Built-in Features


	Accept specification - Define accepted file types, max
number of entries, max file size, etc. When the client
selects file(s), the file metadata is automatically
validated against the specification. See
Phoenix.LiveView.allow_upload/3.

	Reactive entries - Uploads are populated in an
@uploads assign in the socket. Entries automatically
respond to progress, errors, cancellation, etc.

	Drag and drop - Use the phx-drop-target attribute to
enable. See Phoenix.Component.live_file_input/1.



  
    
  
  Allow uploads


You enable an upload, typically on mount, via allow_upload/3.
For this example, we will also keep a list of uploaded files in
a new assign named uploaded_files, but you could name it
something else if you wanted.
@impl Phoenix.LiveView
def mount(_params, _session, socket) do
  {:ok,
   socket
   |> assign(:uploaded_files, [])
   |> allow_upload(:avatar, accept: ~w(.jpg .jpeg), max_entries: 2)}
end
That's it for now! We will come back to the LiveView to
implement some form- and upload-related callbacks later, but
most of the functionality around uploads takes place in the
template.

  
    
  
  Render reactive elements


Use the Phoenix.Component.live_file_input/1 component
to render a file input for the upload:
<%!-- lib/my_app_web/live/upload_live.html.heex --%>

<form id="upload-form" phx-submit="save" phx-change="validate">
  <.live_file_input upload={@uploads.avatar} />
  <button type="submit">Upload</button>
</form>
Important: You must bind phx-submit and phx-change on the form.

Note that while live_file_input/1
allows you to set additional attributes on the file input,
many attributes such as id, accept, and multiple will
be set automatically based on the allow_upload/3 spec.
Reactive updates to the template will occur as the end-user
interacts with the file input.

  
    
  
  Upload entries


Uploads are populated in an @uploads assign in the socket.
Each allowed upload contains a list of entries,
irrespective of the :max_entries value in the
allow_upload/3 spec. These entry structs contain all the
information about an upload, including progress, client file
info, errors, etc.
Let's look at an annotated example:
<%!-- lib/my_app_web/live/upload_live.html.heex --%>

<%!-- use phx-drop-target with the upload ref to enable file drag and drop --%>
<section phx-drop-target={@uploads.avatar.ref}>
  <%!-- render each avatar entry --%>
  <article :for={entry <- @uploads.avatar.entries} class="upload-entry">
    <figure>
      <.live_img_preview entry={entry} />
      <figcaption>{entry.client_name}</figcaption>
    </figure>

    <%!-- entry.progress will update automatically for in-flight entries --%>
    <progress value={entry.progress} max="100"> {entry.progress}% </progress>

    <%!-- a regular click event whose handler will invoke Phoenix.LiveView.cancel_upload/3 --%>
    <button type="button" phx-click="cancel-upload" phx-value-ref={entry.ref} aria-label="cancel">&times;</button>

    <%!-- Phoenix.Component.upload_errors/2 returns a list of error atoms --%>
    <p :for={err <- upload_errors(@uploads.avatar, entry)} class="alert alert-danger">{error_to_string(err)}</p>
  </article>

  <%!-- Phoenix.Component.upload_errors/1 returns a list of error atoms --%>
  <p :for={err <- upload_errors(@uploads.avatar)} class="alert alert-danger">
    {error_to_string(err)}
  </p>
</section>
The section element in the example acts as the
phx-drop-target for the :avatar upload. Users can interact
with the file input or they can drop files over the element
to add new entries.
Upload entries are created when a file is added to the form
input and each will exist until it has been consumed,
following a successfully completed upload.

  
    
  
  Entry validation


Validation occurs automatically based on any conditions
that were specified in allow_upload/3 however, as
mentioned previously you are required to bind phx-change
on the form in order for the validation to be performed.
Therefore you must implement at least a minimal callback:
@impl Phoenix.LiveView
def handle_event("validate", _params, socket) do
  {:noreply, socket}
end
Entries for files that do not match the allow_upload/3
spec will contain errors. Use
Phoenix.Component.upload_errors/2 and your own
helper function to render a friendly error message:
defp error_to_string(:too_large), do: "Too large"
defp error_to_string(:not_accepted), do: "You have selected an unacceptable file type"
For error messages that affect all entries, use
Phoenix.Component.upload_errors/1, and your own
helper function to render a friendly error message:
defp error_to_string(:too_many_files), do: "You have selected too many files"

  
    
  
  Cancel an entry


Upload entries may also be canceled, either programmatically
or as a result of a user action. For instance, to handle the
click event in the template above, you could do the following:
@impl Phoenix.LiveView
def handle_event("cancel-upload", %{"ref" => ref}, socket) do
  {:noreply, cancel_upload(socket, :avatar, ref)}
end

  
    
  
  Consume uploaded entries


When the end-user submits a form containing a live_file_input/1,
the JavaScript client first uploads the file(s) before
invoking the callback for the form's phx-submit event.
Within the callback for the phx-submit event, you invoke
the Phoenix.LiveView.consume_uploaded_entries/3 function
to process the completed uploads, persisting the relevant
upload data alongside the form data:
@impl Phoenix.LiveView
def handle_event("save", _params, socket) do
  uploaded_files =
    consume_uploaded_entries(socket, :avatar, fn %{path: path}, _entry ->
      dest = Path.join(Application.app_dir(:my_app, "priv/static/uploads"), Path.basename(path))
      # You will need to create `priv/static/uploads` for `File.cp!/2` to work.
      File.cp!(path, dest)
      {:ok, ~p"/uploads/#{Path.basename(dest)}"}
    end)

  {:noreply, update(socket, :uploaded_files, &(&1 ++ uploaded_files))}
end
Note: While client metadata cannot be trusted, max file size validations
are enforced as each chunk is received when performing direct to server uploads.

This example writes the file directly to disk, under the priv folder.
In order to access your upload, for example in an <img /> tag, you need
to add the uploads directory to static_paths/0.  In a vanilla Phoenix
project, this is found in lib/my_app_web.ex.
Another thing to be aware of is that in development, changes to
priv/static/uploads will be picked up by live_reload.  This means that as
soon as your upload succeeds, your app will be reloaded in the browser.  This
can be temporarily disabled by setting code_reloader: false in config/dev.exs.
Besides the above, this approach also has limitations in production. If you are
running multiple instances of your application, the uploaded file will be stored
only in one of the instances. Any request routed to the other machine will
ultimately fail.
For these reasons, it is best if uploads are stored elsewhere, such as the
database (depending on the size and contents) or a separate storage service.
For more information on implementing client-side, direct-to-cloud uploads,
see the External uploads guide for details.

  
    
  
  Appendix A: UploadLive


A complete example of the LiveView from this guide:
# lib/my_app_web/live/upload_live.ex
defmodule MyAppWeb.UploadLive do
  use MyAppWeb, :live_view

  @impl Phoenix.LiveView
  def mount(_params, _session, socket) do
    {:ok,
    socket
    |> assign(:uploaded_files, [])
    |> allow_upload(:avatar, accept: ~w(.jpg .jpeg), max_entries: 2)}
  end

  @impl Phoenix.LiveView
  def handle_event("validate", _params, socket) do
    {:noreply, socket}
  end

  @impl Phoenix.LiveView
  def handle_event("cancel-upload", %{"ref" => ref}, socket) do
    {:noreply, cancel_upload(socket, :avatar, ref)}
  end

  @impl Phoenix.LiveView
  def handle_event("save", _params, socket) do
    uploaded_files =
      consume_uploaded_entries(socket, :avatar, fn %{path: path}, _entry ->
        dest = Path.join([:code.priv_dir(:my_app), "static", "uploads", Path.basename(path)])
        # You will need to create `priv/static/uploads` for `File.cp!/2` to work.
        File.cp!(path, dest)
        {:ok, ~p"/uploads/#{Path.basename(dest)}"}
      end)

    {:noreply, update(socket, :uploaded_files, &(&1 ++ uploaded_files))}
  end

  defp error_to_string(:too_large), do: "Too large"
  defp error_to_string(:too_many_files), do: "You have selected too many files"
  defp error_to_string(:not_accepted), do: "You have selected an unacceptable file type"
end
To access your uploads via your app, make sure to add uploads to
MyAppWeb.static_paths/0.


  

    
Bindings
    

Phoenix supports DOM element bindings for client-server interaction. For
example, to react to a click on a button, you would render the element:
<button phx-click="inc_temperature">+</button>
Then on the server, all LiveView bindings are handled with the handle_event
callback, for example:
def handle_event("inc_temperature", _value, socket) do
  {:ok, new_temp} = Thermostat.inc_temperature(socket.assigns.id)
  {:noreply, assign(socket, :temperature, new_temp)}
end
	Binding	Attributes
	Params	phx-value-*
	Click Events	phx-click, phx-click-away
	Form Events	phx-change, phx-submit, phx-disable-with, phx-trigger-action, phx-auto-recover
	Focus Events	phx-blur, phx-focus, phx-window-blur, phx-window-focus
	Key Events	phx-keydown, phx-keyup, phx-window-keydown, phx-window-keyup, phx-key
	Scroll Events	phx-viewport-top, phx-viewport-bottom
	DOM Patching	phx-update, phx-mounted, phx-remove
	JS Interop	phx-hook
	Lifecycle Events	phx-connected, phx-disconnected
	Rate Limiting	phx-debounce, phx-throttle
	Static tracking	phx-track-static

If you need to trigger commands actions via JavaScript, see JavaScript interoperability.

  
    
  
  Click Events


The phx-click binding is used to send click events to the server.
When any client event, such as a phx-click click is pushed, the value
sent to the server will be chosen with the following priority:
	The :value specified in Phoenix.LiveView.JS.push/3, such as:
<div phx-click={JS.push("inc", value: %{myvar1: @val1})}>

	Any number of optional phx-value- prefixed attributes, such as:
<div phx-click="inc" phx-value-myvar1="val1" phx-value-myvar2="val2">
will send the following map of params to the server:
def handle_event("inc", %{"myvar1" => "val1", "myvar2" => "val2"}, socket) do
If the phx-value- prefix is used, the server payload will also contain a "value"
if the element's value attribute exists.

	The payload will also include any additional user defined metadata of the client event.
For example, the following LiveSocket client option would send the coordinates and
altKey information for all clicks:
let liveSocket = new LiveSocket("/live", Socket, {
  params: {_csrf_token: csrfToken},
  metadata: {
    click: (e, el) => {
      return {
        altKey: e.altKey,
        clientX: e.clientX,
        clientY: e.clientY
      }
    }
  }
})


The phx-click-away event is fired when a click event happens outside of the element.
This is useful for hiding toggled containers like drop-downs.

  
    
  
  Focus and Blur Events


Focus and blur events may be bound to DOM elements that emit
such events, using the phx-blur, and phx-focus bindings, for example:
<input name="email" phx-focus="myfocus" phx-blur="myblur"/>
To detect when the page itself has received focus or blur,
phx-window-focus and phx-window-blur may be specified. These window
level events may also be necessary if the element in consideration
(most often a div with no tabindex) cannot receive focus. Like other
bindings, phx-value-* can be provided on the bound element, and those
values will be sent as part of the payload. For example:
<div class="container"
    phx-window-focus="page-active"
    phx-window-blur="page-inactive"
    phx-value-page="123">
  ...
</div>

  
    
  
  Key Events


The onkeydown, and onkeyup events are supported via the phx-keydown,
and phx-keyup bindings. Each binding supports a phx-key attribute, which triggers
the event for the specific key press. If no phx-key is provided, the event is triggered
for any key press. When pushed, the value sent to the server will contain the "key"
that was pressed, plus any user-defined metadata. For example, pressing the
Escape key looks like this:
%{"key" => "Escape"}
To capture additional user-defined metadata, the metadata option for keydown events
may be provided to the LiveSocket constructor. For example:
let liveSocket = new LiveSocket("/live", Socket, {
  params: {_csrf_token: csrfToken},
  metadata: {
    keydown: (e, el) => {
      return {
        key: e.key,
        metaKey: e.metaKey,
        repeat: e.repeat
      }
    }
  }
})
To determine which key has been pressed you should use key value. The
available options can be found on
MDN
or via the Key Event Viewer.
Note: phx-keyup and phx-keydown are not supported on inputs.
Instead use form bindings, such as phx-change, phx-submit, etc.
Note: it is possible for certain browser features like autofill to trigger key events
with no "key" field present in the value map sent to the server. For this reason, we
recommend always having a fallback catch-all event handler for LiveView key bindings.
By default, the bound element will be the event listener, but a
window-level binding may be provided via phx-window-keydown or phx-window-keyup,
for example:
def render(assigns) do
  ~H"""
  <div id="thermostat" phx-window-keyup="update_temp">
    Current temperature: {@temperature}
  </div>
  """
end

def handle_event("update_temp", %{"key" => "ArrowUp"}, socket) do
  {:ok, new_temp} = Thermostat.inc_temperature(socket.assigns.id)
  {:noreply, assign(socket, :temperature, new_temp)}
end

def handle_event("update_temp", %{"key" => "ArrowDown"}, socket) do
  {:ok, new_temp} = Thermostat.dec_temperature(socket.assigns.id)
  {:noreply, assign(socket, :temperature, new_temp)}
end

def handle_event("update_temp", _, socket) do
  {:noreply, socket}
end

  
    
  
  Rate limiting events with Debounce and Throttle


All events can be rate-limited on the client by using the
phx-debounce and phx-throttle bindings, with the exception of the phx-blur
binding, which is fired immediately.
Rate limited and debounced events have the following behavior:
	phx-debounce - Accepts either an integer timeout value (in milliseconds),
or "blur". When an integer is provided, emitting the event is delayed by
the specified milliseconds. When "blur" is provided, emitting the event is
delayed until the field is blurred by the user. When the value is omitted
a default of 300ms is used. Debouncing is typically used for input elements.

	phx-throttle - Accepts an integer timeout value to throttle the event in milliseconds.
Unlike debounce, throttle will immediately emit the event, then rate limit it at once
per provided timeout. When the value is omitted a default of 300ms is used.
Throttling is typically used to rate limit clicks, mouse and keyboard actions.


For example, to avoid validating an email until the field is blurred, while validating
the username at most every 2 seconds after a user changes the field:
<form phx-change="validate" phx-submit="save">
  <input type="text" name="user[email]" phx-debounce="blur"/>
  <input type="text" name="user[username]" phx-debounce="2000"/>
</form>
And to rate limit a volume up click to once every second:
<button phx-click="volume_up" phx-throttle="1000">+</button>
Likewise, you may throttle held-down keydown:
<div phx-window-keydown="keydown" phx-throttle="500">
  ...
</div>
Unless held-down keys are required, a better approach is generally to use
phx-keyup bindings which only trigger on key up, thereby being self-limiting.
However, phx-keydown is useful for games and other use cases where a constant
press on a key is desired. In such cases, throttle should always be used.

  
    
  
  Debounce and Throttle special behavior


The following specialized behavior is performed for forms and keydown bindings:
	When a phx-submit, or a phx-change for a different input is triggered,
any current debounce or throttle timers are reset for existing inputs.

	A phx-keydown binding is only throttled for key repeats. Unique keypresses
back-to-back will dispatch the pressed key events.



  
    
  
  JS commands


LiveView bindings support a JavaScript command interface via the Phoenix.LiveView.JS module, which allows you to specify utility operations that execute on the client when firing phx- binding events, such as phx-click, phx-change, etc. Commands compose together to allow you to push events, add classes to elements, transition elements in and out, and more.
See the Phoenix.LiveView.JS documentation for full usage.
For a small example of what's possible, imagine you want to show and hide a modal on the page without needing to make the round trip to the server to render the content:
<div id="modal" class="modal">
  My Modal
</div>

<button phx-click={JS.show(to: "#modal", transition: "fade-in")}>
  show modal
</button>

<button phx-click={JS.hide(to: "#modal", transition: "fade-out")}>
  hide modal
</button>

<button phx-click={JS.toggle(to: "#modal", in: "fade-in", out: "fade-out")}>
  toggle modal
</button>
Or if your UI library relies on classes to perform the showing or hiding:
<div id="modal" class="modal">
  My Modal
</div>

<button phx-click={JS.add_class("show", to: "#modal", transition: "fade-in")}>
  show modal
</button>

<button phx-click={JS.remove_class("show", to: "#modal", transition: "fade-out")}>
  hide modal
</button>
Commands compose together. For example, you can push an event to the server and
immediately hide the modal on the client:
<div id="modal" class="modal">
  My Modal
</div>

<button phx-click={JS.push("modal-closed") |> JS.remove_class("show", to: "#modal", transition: "fade-out")}>
  hide modal
</button>
It is also useful to extract commands into their own functions:
alias Phoenix.LiveView.JS

def hide_modal(js \\ %JS{}, selector) do
  js
  |> JS.push("modal-closed")
  |> JS.remove_class("show", to: selector, transition: "fade-out")
end
<button phx-click={hide_modal("#modal")}>hide modal</button>
The Phoenix.LiveView.JS.push/3 command is particularly powerful in allowing you to customize the event being pushed to the server. For example, imagine you start with a familiar phx-click which pushes a message to the server when clicked:
<button phx-click="clicked">click</button>
Now imagine you want to customize what happens when the "clicked" event is pushed, such as which component should be targeted, which element should receive CSS loading state classes, etc. This can be accomplished with options on the JS push command. For example:
<button phx-click={JS.push("clicked", target: @myself, loading: ".container")}>click</button>
See Phoenix.LiveView.JS.push/3 for all supported options.

  
    
  
  DOM patching


A container can be marked with phx-update to configure how the DOM
is updated. The following values are supported:
	replace - the default operation. Replaces the element with the contents

	stream - supports stream operations. Streams are used to manage large
collections in the UI without having to store the collection on the server

	ignore - ignores updates to the DOM regardless of new content changes.
This is useful for client-side interop with existing libraries that do
their own DOM operations


When using phx-update, a unique DOM ID must always be set in the
container. If using "stream", a DOM ID must also be set
for each child. When inserting stream elements containing an
ID already present in the container, LiveView will replace the existing
element with the new content. See Phoenix.LiveView.stream/3 for more
information.
The "ignore" behaviour is frequently used when you need to integrate
with another JS library. Updates from the server to the element's content
and attributes are ignored, except for data attributes. Changes, additions,
and removals from the server to data attributes are merged with the ignored
element which can be used to pass data to the JS handler.
To react to elements being mounted to the DOM, the phx-mounted binding
can be used. For example, to animate an element on mount:
<div phx-mounted={JS.transition("animate-ping", time: 500)}>
If phx-mounted is used on the initial page render, it will be invoked only
after the initial WebSocket connection is established.
To react to elements being removed from the DOM, the phx-remove binding
may be specified, which can contain a Phoenix.LiveView.JS command to execute.
The phx-remove command is only executed for the removed parent element.
It does not cascade to children.
To react to elements being updated in the DOM, you'll need to use a
hook, which gives you full access
to the element life-cycle.

  
    
  
  Lifecycle events


LiveView supports the phx-connected and phx-disconnected bindings to react
to connection lifecycle events with JS commands. For example, to show an element
when the LiveView has lost its connection and hide it when the connection
recovers:
<div id="status" class="hidden" phx-disconnected={JS.show()} phx-connected={JS.hide()}>
  Attempting to reconnect...
</div>
phx-connected and phx-disconnected are only executed when operating
inside a LiveView container. For static templates, they will have no effect.

  
    
  
  LiveView events prefix


The lv: event prefix supports LiveView specific features that are handled
by LiveView without calling the user's handle_event/3 callbacks. Today,
the following events are supported:
	lv:clear-flash – clears the flash when sent to the server. If a
phx-value-key is provided, the specific key will be removed from the flash.

For example:
<p class="alert" phx-click="lv:clear-flash" phx-value-key="info">
  {Phoenix.Flash.get(@flash, :info)}
</p>

  
    
  
  Scroll events and infinite pagination


The phx-viewport-top and phx-viewport-bottom bindings allow you to detect when a container's
first child reaches the top of the viewport, or the last child reaches the bottom of the viewport.
This is useful for infinite scrolling where you want to send paging events for the next results set or previous results set as the user is scrolling up and down and reaches the top or bottom of the viewport.
Generally, applications will add padding above and below a container when performing infinite scrolling to allow smooth scrolling as results are loaded. Combined with Phoenix.LiveView.stream/3, the phx-viewport-top and phx-viewport-bottom allow for infinite virtualized list that only keeps a small set of actual elements in the DOM. For example:
def mount(_, _, socket) do
  {:ok,
    socket
    |> assign(page: 1, per_page: 20)
    |> paginate_posts(1)}
end

defp paginate_posts(socket, new_page) when new_page >= 1 do
  %{per_page: per_page, page: cur_page} = socket.assigns
  posts = Blog.list_posts(offset: (new_page - 1) * per_page, limit: per_page)

  {posts, at, limit} =
    if new_page >= cur_page do
      {posts, -1, per_page * 3 * -1}
    else
      {Enum.reverse(posts), 0, per_page * 3}
    end

  case posts do
    [] ->
      assign(socket, end_of_timeline?: at == -1)

    [_ | _] = posts ->
      socket
      |> assign(end_of_timeline?: false)
      |> assign(:page, new_page)
      |> stream(:posts, posts, at: at, limit: limit)
  end
end
Our paginate_posts function fetches a page of posts, and determines if the user is paging to a previous page or next page. Based on the direction of paging, the stream is either prepended to, or appended to with at of 0 or -1 respectively. We also set the limit of the stream to three times the per_page to allow enough posts in the UI to appear as an infinite list, but small enough to maintain UI performance. We also set an @end_of_timeline? assign to track whether the user is at the end of results or not. Finally, we update the @page assign and posts stream. We can then wire up our container to support the viewport events:
<ul
  id="posts"
  phx-update="stream"
  phx-viewport-top={@page > 1 && JS.push("prev-page", page_loading: true)}
  phx-viewport-bottom={!@end_of_timeline? && JS.push("next-page", page_loading: true)}
  class={[
    if(@end_of_timeline?, do: "pb-10", else: "pb-[calc(200vh)]"),
    if(@page == 1, do: "pt-10", else: "pt-[calc(200vh)]")
  ]}
>
  <li :for={{id, post} <- @streams.posts} id={id}>
    <.post_card post={post} />
  </li>
</ul>
<div :if={@end_of_timeline?} class="mt-5 text-[50px] text-center">
  🎉 You made it to the beginning of time 🎉
</div>
There's not much here, but that's the point! This little snippet of UI is driving a fully virtualized list with bidirectional infinite scrolling. We use the phx-viewport-top binding to send the "prev-page" event to the LiveView, but only if the user is beyond the first page. It doesn't make sense to load negative page results, so we remove the binding entirely in those cases. Next, we wire up phx-viewport-bottom to send the "next-page" event, but only if we've yet to reach the end of the timeline. Finally, we conditionally apply some CSS classes which sets a large top and bottom padding to twice the viewport height based on the current pagination for smooth scrolling.
To complete our solution, we only need to handle the "prev-page" and "next-page" events in the LiveView:
def handle_event("next-page", _, socket) do
  {:noreply, paginate_posts(socket, socket.assigns.page + 1)}
end

def handle_event("prev-page", %{"_overran" => true}, socket) do
  {:noreply, paginate_posts(socket, 1)}
end

def handle_event("prev-page", _, socket) do
  if socket.assigns.page > 1 do
    {:noreply, paginate_posts(socket, socket.assigns.page - 1)}
  else
    {:noreply, socket}
  end
end
This code simply calls the paginate_posts function we defined as our first step, using the current or next page to drive the results. Notice that we match on a special "_overran" => true parameter in our "prev-page" event. The viewport events send this parameter when the user has "overran" the viewport top or bottom. Imagine the case where the user is scrolling back up through many pages of results, but grabs the scrollbar and returns immediately to the top of the page. This means our <ul id="posts"> container was overrun by the top of the viewport, and we need to reset the the UI to page the first page.
When testing, you can use Phoenix.LiveViewTest.render_hook/3 to test the viewport events:
view
|> element("#posts")
|> render_hook("next-page")


  

    
External uploads
    

This guide continues from the configuration started in the
server Uploads guide.

Uploads to external cloud providers, such as Amazon S3,
Google Cloud, etc., can be achieved by using the
:external option in allow_upload/3.
You provide a 2-arity function to allow the server to
generate metadata for each upload entry, which is passed to
a user-specified JavaScript function on the client.
Typically when your function is invoked, you will generate a
pre-signed URL, specific to your cloud storage provider, that
will provide temporary access for the end-user to upload data
directly to your cloud storage.

  
    
  
  Chunked HTTP Uploads


For any service that supports large file
uploads via chunked HTTP requests with Content-Range
headers, you can use the UpChunk JS library by Mux to do all
the hard work of uploading the file. For small file uploads
or to get started quickly, consider uploading directly to S3
instead.
You only need to wire the UpChunk instance to the LiveView
UploadEntry callbacks, and LiveView will take care of the rest.
Install UpChunk by
saving its contents
to assets/vendor/upchunk.js or by installing it with npm:
$ npm install --prefix assets --save @mux/upchunk

Configure your uploader on Phoenix.LiveView.mount/3:
def mount(_params, _session, socket) do
  {:ok,
   socket
   |> assign(:uploaded_files, [])
   |> allow_upload(:avatar, accept: :any, max_entries: 3, external: &presign_upload/2)}
end
Supply the :external option to
Phoenix.LiveView.allow_upload/3. It requires a 2-arity
function that generates a signed URL where the client will
push the bytes for the upload entry. This function must
return either {:ok, meta, socket} or {:error, meta, socket},
where meta must be a map.
For example, if you were using a context that provided a
start_session
function, you might write something like this:
defp presign_upload(entry, socket) do
  {:ok, %{"Location" => link}} =
    SomeTube.start_session(%{
      "uploadType" => "resumable",
      "x-upload-content-length" => entry.client_size
    })

  {:ok, %{uploader: "UpChunk", entrypoint: link}, socket}
end
Finally, on the client-side, we use UpChunk to create an
upload from the temporary URL generated on the server and
attach listeners for its events to the entry's callbacks:
import * as UpChunk from "@mux/upchunk"

let Uploaders = {}

Uploaders.UpChunk = function(entries, onViewError){
  entries.forEach(entry => {
    // create the upload session with UpChunk
    let { file, meta: { entrypoint } } = entry
    let upload = UpChunk.createUpload({ endpoint: entrypoint, file })

    // stop uploading in the event of a view error
    onViewError(() => upload.pause())

    // upload error triggers LiveView error
    upload.on("error", (e) => entry.error(e.detail.message))

    // notify progress events to LiveView
    upload.on("progress", (e) => {
      if(e.detail < 100){ entry.progress(e.detail) }
    })

    // success completes the UploadEntry
    upload.on("success", () => entry.progress(100))
  })
}

// Don't forget to assign Uploaders to the liveSocket
let liveSocket = new LiveSocket("/live", Socket, {
  uploaders: Uploaders,
  params: {_csrf_token: csrfToken}
})

  
    
  
  Direct to S3


The largest object that can be uploaded to S3 in a single PUT is 5 GB
according to S3 FAQ. For larger file
uploads, consider using chunking as shown above.
This guide assumes an existing S3 bucket is set up with the correct CORS configuration
which allows uploading directly to the bucket.
An example CORS config is:
[
    {
        "AllowedHeaders": [ "*" ],
        "AllowedMethods": [ "PUT", "POST" ],
        "AllowedOrigins": [ "*" ],
        "ExposeHeaders": []
    }
]
You may put your domain in the "allowedOrigins" instead. More information on configuring CORS for
S3 buckets is available on AWS.
In order to enforce all of your file constraints when uploading to S3,
it is necessary to perform a multipart form POST with your file data.
You should have the following S3 information ready before proceeding:
	aws_access_key_id
	aws_secret_access_key
	bucket_name
	region

We will first implement the LiveView portion:
def mount(_params, _session, socket) do
  {:ok,
    socket
    |> assign(:uploaded_files, [])
    |> allow_upload(:avatar, accept: :any, max_entries: 3, external: &presign_upload/2)}
end

defp presign_upload(entry, socket) do
  uploads = socket.assigns.uploads
  bucket = "phx-upload-example"
  key = "public/#{entry.client_name}"

  config = %{
    region: "us-east-1",
    access_key_id: System.fetch_env!("AWS_ACCESS_KEY_ID"),
    secret_access_key: System.fetch_env!("AWS_SECRET_ACCESS_KEY")
  }

  {:ok, fields} =
    SimpleS3Upload.sign_form_upload(config, bucket,
      key: key,
      content_type: entry.client_type,
      max_file_size: uploads[entry.upload_config].max_file_size,
      expires_in: :timer.hours(1)
    )

  meta = %{uploader: "S3", key: key, url: "http://#{bucket}.s3-#{config.region}.amazonaws.com", fields: fields}
  {:ok, meta, socket}
end
Here, we implemented a presign_upload/2 function, which we passed as a
captured anonymous function to :external. It generates a pre-signed URL
for the upload and returns our :ok result, with a payload of metadata
for the client, along with our unchanged socket. 
Next, we add a missing module SimpleS3Upload to generate pre-signed URLs
for S3. Create a file called simple_s3_upload.ex. Get the file's content
from this zero-dependency module called SimpleS3Upload
written by Chris McCord.
Tip: if you encounter errors with the :crypto module or with S3 blocking ACLs, 
please read the comments in the gist above for solutions.

Next, we add our JavaScript client-side uploader. The metadata must contain the
:uploader key, specifying the name of the JavaScript client-side uploader.
In this case, it's "S3", as shown above.
Add a new file uploaders.js in the following directory assets/js/ next to app.js.
The content for this S3 client uploader:
let Uploaders = {}

Uploaders.S3 = function(entries, onViewError){
  entries.forEach(entry => {
    let formData = new FormData()
    let {url, fields} = entry.meta
    Object.entries(fields).forEach(([key, val]) => formData.append(key, val))
    formData.append("file", entry.file)
    let xhr = new XMLHttpRequest()
    onViewError(() => xhr.abort())
    xhr.onload = () => xhr.status === 204 ? entry.progress(100) : entry.error()
    xhr.onerror = () => entry.error()
    xhr.upload.addEventListener("progress", (event) => {
      if(event.lengthComputable){
        let percent = Math.round((event.loaded / event.total) * 100)
        if(percent < 100){ entry.progress(percent) }
      }
    })

    xhr.open("POST", url, true)
    xhr.send(formData)
  })
}

export default Uploaders;
We define an Uploaders.S3 function, which receives our entries. It then
performs an AJAX request for each entry, using the entry.progress() and
entry.error() functions to report upload events back to the LiveView.
The name of the uploader must match the one we return on the :uploader
metadata in LiveView.
Finally, head over to app.js and add the uploaders: Uploaders key to
the LiveSocket constructor to tell phoenix where to find the uploaders returned 
within the external metadata.
// for uploading to S3
import Uploaders from "./uploaders"

let liveSocket = new LiveSocket("/live",
   Socket, {
     params: {_csrf_token: csrfToken},
     uploaders: Uploaders
  }
)
Now "S3" returned from the server will match the one in the client.
To debug client-side JavaScript when trying to upload, you can inspect your
browser and look at the console or networks tab to view the error logs.

  
    
  
  Direct to S3-Compatible


This section assumes that you installed and configured ExAws
and ExAws.S3 correctly in your project and can execute
the examples in the page without errors.

Most S3 compatible platforms like Cloudflare R2 don't support POST when
uploading files so we need to use PUT with a signed URL instead of the
signed POSTand send the file straight to the service, to do so we need to
change the presign_upload/2 function and the Uploaders.S3 that does the upload.
The new presign_upload/2:
def presign_upload(entry, socket) do
  config = ExAws.Config.new(:s3)
  bucket = "bucket"
  key = "public/#{entry.client_name}"

  {:ok, url} =
    ExAws.S3.presigned_url(config, :put, bucket, key,
      expires_in: 3600,
      query_params: [{"Content-Type", entry.client_type}]
    )
   {:ok, %{uploader: "S3", key: key, url: url}, socket}
end
The new Uploaders.S3:
Uploaders.S3 = function (entries, onViewError) {
  entries.forEach(entry => {
    let xhr = new XMLHttpRequest()
    onViewError(() => xhr.abort())
    xhr.onload = () => xhr.status === 200 ? entry.progress(100) : entry.error()
    xhr.onerror = () => entry.error()

    xhr.upload.addEventListener("progress", (event) => {
      if(event.lengthComputable){
        let percent = Math.round((event.loaded / event.total) * 100)
        if(percent < 100){ entry.progress(percent) }
      }
    })

    let url = entry.meta.url
    xhr.open("PUT", url, true)
    xhr.send(entry.file)
  })
}


  

    
Form bindings
    


  
    
  
  Form events


To handle form changes and submissions, use the phx-change and phx-submit
events. In general, it is preferred to handle input changes at the form level,
where all form fields are passed to the LiveView's callback given any
single input change. For example, to handle real-time form validation and
saving, your form would use both phx-change and phx-submit bindings.
Let's get started with an example:
<.form for={@form} phx-change="validate" phx-submit="save">
  <.input type="text" field={@form[:username]} />
  <.input type="email" field={@form[:email]} />
  <button>Save</button>
</.form>
.form is the function component defined in Phoenix.Component.form/1,
we recommend reading its documentation for more details on how it works
and all supported options. .form expects a @form assign, which can
be created from a changeset or user parameters via Phoenix.Component.to_form/1.
input/1 is a function component for rendering inputs, most often
defined in your own application, often encapsulating labelling,
error handling, and more. Here is a simple version to get started with:
attr :field, Phoenix.HTML.FormField
attr :rest, :global, include: ~w(type)
def input(assigns) do
  ~H"""
  <input id={@field.id} name={@field.name} value={@field.value} {@rest} />
  """
end

  
    
  
  The CoreComponents module


If your application was generated with Phoenix v1.7, then mix phx.new
automatically imports many ready-to-use function components, such as
.input component with built-in features and styles.
With the form rendered, your LiveView picks up the events in handle_event
callbacks, to validate and attempt to save the parameter accordingly:
def render(assigns) ...

def mount(_params, _session, socket) do
  {:ok, assign(socket, form: to_form(Accounts.change_user(%User{})))}
end

def handle_event("validate", %{"user" => params}, socket) do
  form =
    %User{}
    |> Accounts.change_user(params)
    |> to_form(action: :validate)

  {:noreply, assign(socket, form: form)}
end

def handle_event("save", %{"user" => user_params}, socket) do
  case Accounts.create_user(user_params) do
    {:ok, user} ->
      {:noreply,
       socket
       |> put_flash(:info, "user created")
       |> redirect(to: ~p"/users/#{user}")}

    {:error, %Ecto.Changeset{} = changeset} ->
      {:noreply, assign(socket, form: to_form(changeset))}
  end
end
The validate callback simply updates the changeset based on all form input
values, then convert the changeset to a form and assign it to the socket.
If the form changes, such as generating new errors, render/1
is invoked and the form is re-rendered.
Likewise for phx-submit bindings, the same callback is invoked and
persistence is attempted. On success, a :noreply tuple is returned and the
socket is annotated for redirect with Phoenix.LiveView.redirect/2 to
the new user page, otherwise the socket assigns are updated with the errored
changeset to be re-rendered for the client.
You may wish for an individual input to use its own change event or to target
a different component. This can be accomplished by annotating the input itself
with phx-change, for example:
<.form for={@form} phx-change="validate" phx-submit="save">
  ...
  <.input field={@form[:email]}  phx-change="email_changed" phx-target={@myself} />
</.form>
Then your LiveView or LiveComponent would handle the event:
def handle_event("email_changed", %{"user" => %{"email" => email}}, socket) do
  ...
end
Note
	Only the individual input is sent as params for an input marked with phx-change.
	While it is possible to use phx-change on individual inputs, those inputs
must still be within a form.


  
    
  
  Error feedback


For proper error feedback on form updates, LiveView sends special parameters on form events
starting with _unused_ to indicate that the input for the specific field has not been interacted with yet.
When creating a form from these parameters through Phoenix.Component.to_form/2 or Phoenix.Component.form/1,
Phoenix.Component.used_input?/1 can be used to filter error messages.
For example, your MyAppWeb.CoreComponents may use this function:
def input(%{field: %Phoenix.HTML.FormField{} = field} = assigns) do
  errors = if Phoenix.Component.used_input?(field), do: field.errors, else: []

  assigns
  |> assign(field: nil, id: assigns.id || field.id)
  |> assign(:errors, Enum.map(errors, &translate_error(&1)))
Now only errors for fields that were interacted with are shown.

  
    
  
  Number inputs


Number inputs are a special case in LiveView forms. On programmatic updates,
some browsers will clear invalid inputs. So LiveView will not send change events
from the client when an input is invalid, instead allowing the browser's native
validation UI to drive user interaction. Once the input becomes valid, change and
submit events will be sent normally.
<input type="number">
This is known to have a plethora of problems including accessibility, large numbers
are converted to exponential notation, and scrolling can accidentally increase or
decrease the number.
One alternative is the inputmode attribute, which may serve your application's needs
and users much better. According to Can I Use?,
the following is supported by 94% of the global market (as of Nov 2024):
<input type="text" inputmode="numeric" pattern="[0-9]*">

  
    
  
  Password inputs


Password inputs are also special cased in Phoenix.HTML. For security reasons,
password field values are not reused when rendering a password input tag. This
requires explicitly setting the :value in your markup, for example:
<.input field={f[:password]} value={input_value(f[:password].value)} />
<.input field={f[:password_confirmation]} value={input_value(f[:password_confirmation].value)} />

  
    
  
  Nested inputs


Nested inputs are handled using .inputs_for function component. By default
it will add the necessary hidden input fields for tracking ids of Ecto associations.
<.inputs_for :let={fp} field={f[:friends]}>
  <.input field={fp[:name]} type="text" />
</.inputs_for>

  
    
  
  File inputs


LiveView forms support reactive file inputs,
including drag and drop support via the phx-drop-target
attribute:
<div class="container" phx-drop-target={@uploads.avatar.ref}>
  ...
  <.live_file_input upload={@uploads.avatar} />
</div>
See Phoenix.Component.live_file_input/1 for more.

  
    
  
  Submitting the form action over HTTP


The phx-trigger-action attribute can be added to a form to trigger a standard
form submit on DOM patch to the URL specified in the form's standard action
attribute. This is useful to perform pre-final validation of a LiveView form
submit before posting to a controller route for operations that require
Plug session mutation. For example, in your LiveView template you can
annotate the phx-trigger-action with a boolean assign:
<.form :let={f} for={@changeset}
  action={~p"/users/reset_password"}
  phx-submit="save"
  phx-trigger-action={@trigger_submit}>
Then in your LiveView, you can toggle the assign to trigger the form with the current
fields on next render:
def handle_event("save", params, socket) do
  case validate_change_password(socket.assigns.user, params) do
    {:ok, changeset} ->
      {:noreply, assign(socket, changeset: changeset, trigger_submit: true)}

    {:error, changeset} ->
      {:noreply, assign(socket, changeset: changeset)}
  end
end
Once phx-trigger-action is true, LiveView disconnects and then submits the form.

  
    
  
  Recovery following crashes or disconnects


By default, all forms marked with phx-change and having id
attribute will recover input values automatically after the user has
reconnected or the LiveView has remounted after a crash. This is
achieved by the client triggering the same phx-change to the server
as soon as the mount has been completed.
Note: if you want to see form recovery working in development, please
make sure to disable live reloading in development by commenting out the
LiveReload plug in your endpoint.ex file or by setting code_reloader: false
in your config/dev.exs. Otherwise live reloading may cause the current page
to be reloaded whenever you restart the server, which will discard all form
state.
For most use cases, this is all you need and form recovery will happen
without consideration. In some cases, where forms are built step-by-step in a
stateful fashion, it may require extra recovery handling on the server outside
of your existing phx-change callback code. To enable specialized recovery,
provide a phx-auto-recover binding on the form to specify a different event
to trigger for recovery, which will receive the form params as usual. For example,
imagine a LiveView wizard form where the form is stateful and built based on what
step the user is on and by prior selections:
<form id="wizard" phx-change="validate_wizard_step" phx-auto-recover="recover_wizard">
On the server, the "validate_wizard_step" event is only concerned with the
current client form data, but the server maintains the entire state of the wizard.
To recover in this scenario, you can specify a recovery event, such as "recover_wizard"
above, which would wire up to the following server callbacks in your LiveView:
def handle_event("validate_wizard_step", params, socket) do
  # regular validations for current step
  {:noreply, socket}
end

def handle_event("recover_wizard", params, socket) do
  # rebuild state based on client input data up to the current step
  {:noreply, socket}
end
To forgo automatic form recovery, set phx-auto-recover="ignore".

  
    
  
  Resetting forms


To reset a LiveView form, you can use the standard type="reset" on a
form button or input. When clicked, the form inputs will be reset to their
original values.
After the form is reset, a phx-change event is emitted with the _target param
containing the reset name. For example, the following element:
<form phx-change="changed">
  ...
  <button type="reset" name="reset">Reset</button>
</form>
Can be handled on the server differently from your regular change function:
def handle_event("changed", %{"_target" => ["reset"]} = params, socket) do
  # handle form reset
end

def handle_event("changed", params, socket) do
  # handle regular form change
end

  
    
  
  JavaScript client specifics


The JavaScript client is always the source of truth for current input values.
For any given input with focus, LiveView will never overwrite the input's current
value, even if it deviates from the server's rendered updates. This works well
for updates where major side effects are not expected, such as form validation
errors, or additive UX around the user's input values as they fill out a form.
For these use cases, the phx-change input does not concern itself with disabling
input editing while an event to the server is in flight. When a phx-change event
is sent to the server, the input tag and parent form tag receive the
phx-change-loading CSS class, then the payload is pushed to the server with a
"_target" param in the root payload containing the keyspace of the input name
which triggered the change event.
For example, if the following input triggered a change event:
<input name="user[username]"/>
The server's handle_event/3 would receive a payload:
%{"_target" => ["user", "username"], "user" => %{"username" => "Name"}}
The phx-submit event is used for form submissions where major side effects
typically happen, such as rendering new containers, calling an external
service, or redirecting to a new page.
On submission of a form bound with a phx-submit event:
	The form's inputs are set to readonly
	Any submit button on the form is disabled
	The form receives the "phx-submit-loading" class

On completion of server processing of the phx-submit event:
	The submitted form is reactivated and loses the "phx-submit-loading" class
	The last input with focus is restored (unless another input has received focus)
	Updates are patched to the DOM as usual

To handle latent events, the <button> tag of a form can be annotated with
phx-disable-with, which swaps the element's innerText with the provided
value during event submission. For example, the following code would change
the "Save" button to "Saving...", and restore it to "Save" on acknowledgment:
<button type="submit" phx-disable-with="Saving...">Save</button>
A note on disabled buttons
By default, LiveView only disables submit buttons and inputs within forms
while waiting for a server acknowledgement. If you want a button outside of
a form to be disabled without changing its text, you can add phx-disable-with
without a value:
 <button type="button" phx-disable-with>...</button>
Note also that LiveView ignores clicks on elements that are currently awaiting
an acknowledgement from the server. This means that although a regular button
without phx-disable-with is not semantically disabled while waiting for a
server response, it will not trigger duplicate events.
Finally, phx-disable-with works with an element‘s innerText,
therefore nested DOM elements, like svg or icons, won't be preserved.
See "CSS loading states" for alternative approaches to this.
You may also take advantage of LiveView's CSS loading state classes to
swap out your form content while the form is submitting. For example,
with the following rules in your app.css:
.while-submitting { display: none; }
.inputs { display: block; }

.phx-submit-loading .while-submitting { display: block; }
.phx-submit-loading .inputs { display: none; }
You can show and hide content with the following markup:
<form phx-change="update">
  <div class="while-submitting">Please wait while we save our content...</div>
  <div class="inputs">
    <input type="text" name="text" value={@text}>
  </div>
</form>
Additionally, we strongly recommend including a unique HTML "id" attribute on the form.
When DOM siblings change, elements without an ID will be replaced rather than moved,
which can cause issues such as form fields losing focus.

  
    
  
  Triggering phx- form events with JavaScript


Often it is desirable to trigger an event on a DOM element without explicit
user interaction on the element. For example, a custom form element such as a
date picker or custom select input which utilizes a hidden input element to
store the selected state.
In these cases, the event functions on the DOM API can be used, for example
to trigger a phx-change event:
document.getElementById("my-select").dispatchEvent(
  new Event("input", {bubbles: true})
)
When using a client hook, this.el can be used to determine the element as
outlined in the "Client hooks" documentation.
It is also possible to trigger a phx-submit using a "submit" event:
document.getElementById("my-form").dispatchEvent(
  new Event("submit", {bubbles: true, cancelable: true})
)

  
    
  
  Preventing form submission with JavaScript


In some cases, you may want to conditionally prevent form submission based on client-side validation or other business logic before allowing a phx-submit to be processed by the server.
JavaScript can be used to prevent the default form submission behavior, for example with a hook:
/**
 * @type {import("phoenix_live_view").HooksOptions}
 */
let Hooks = {}
Hooks.CustomFormSubmission = {
  mounted() {
    this.el.addEventListener("submit", (event) => {
      if (!this.shouldSubmit()) {
        // prevent the event from bubbling to the default LiveView handler
        event.stopPropagation()
        // prevent the default browser behavior (submitting the form over HTTP)
        event.preventDefault()
      }
    })
  },
  shouldSubmit() {
    // Check if we should submit the form
    ...
  }
}
This hook can be set on your form as such:
<form phx-hook="CustomFormSubmission">
  <input type="text" name="text" value={@text}>
</form>


  

    
JavaScript interoperability
    

To enable LiveView client/server interaction, we instantiate a LiveSocket. For example:
import {Socket} from "phoenix"
import {LiveSocket} from "phoenix_live_view"

let csrfToken = document.querySelector("meta[name='csrf-token']").getAttribute("content")
let liveSocket = new LiveSocket("/live", Socket, {params: {_csrf_token: csrfToken}})
liveSocket.connect()
All options are passed directly to the Phoenix.Socket constructor,
except for the following LiveView specific options:
	bindingPrefix - the prefix to use for phoenix bindings. Defaults "phx-"
	params - the connect_params to pass to the view's mount callback. May be
a literal object or closure returning an object. When a closure is provided,
the function receives the view's element.
	hooks - a reference to a user-defined hooks namespace, containing client
callbacks for server/client interop. See the Client hooks
section below for details.
	uploaders - a reference to a user-defined uploaders namespace, containing
client callbacks for client-side direct-to-cloud uploads. See the
External uploads guide for details.
	metadata - additional user-defined metadata that is sent along events to the server.
See the Key events section in the bindings guide
for an example.

The liveSocket instance exposes the following methods:
	connect() - call this once after creation to connect to the server
	enableDebug() -  turns on debug logging, see Debugging client events
	disableDebug() -  turns off debug logging
	enableLatencySim(milliseconds) - turns on latency simulation, see Simulating latency
	disableLatencySim() - turns off latency simulation
	execJS(el, encodedJS) - executes encoded JavaScript in the context of the element
	js() - returns an object with methods to manipluate the DOM and execute JavaScript. The applied changes integrate with server DOM patching. See JS commands.


  
    
  
  Debugging client events


To aid debugging on the client when troubleshooting issues, the enableDebug()
and disableDebug() functions are exposed on the LiveSocket JavaScript instance.
Calling enableDebug() turns on debug logging which includes LiveView life-cycle and
payload events as they come and go from client to server. In practice, you can expose
your instance on window for quick access in the browser's web console, for example:
// app.js
let liveSocket = new LiveSocket(...)
liveSocket.connect()
window.liveSocket = liveSocket

// in the browser's web console
>> liveSocket.enableDebug()
The debug state uses the browser's built-in sessionStorage, so it will remain in effect
for as long as your browser session lasts.

  
    
  
  Simulating Latency


Proper handling of latency is critical for good UX. LiveView's CSS loading states allow
the client to provide user feedback while awaiting a server response. In development,
near zero latency on localhost does not allow latency to be easily represented or tested,
so LiveView includes a latency simulator with the JavaScript client to ensure your
application provides a pleasant experience. Like the enableDebug() function above,
the LiveSocket instance includes enableLatencySim(milliseconds) and disableLatencySim()
functions which apply throughout the current browser session. The enableLatencySim function
accepts an integer in milliseconds for the one-way latency to and from the server. For example:
// app.js
let liveSocket = new LiveSocket(...)
liveSocket.connect()
window.liveSocket = liveSocket

// in the browser's web console
>> liveSocket.enableLatencySim(1000)
[Log] latency simulator enabled for the duration of this browser session.
      Call disableLatencySim() to disable

  
    
  
  Handling server-pushed events


When the server uses Phoenix.LiveView.push_event/3, the event name
will be dispatched in the browser with the phx: prefix. For example,
imagine the following template where you want to highlight an existing
element from the server to draw the user's attention:
<div id={"item-#{item.id}"} class="item">
  {item.title}
</div>
Next, the server can issue a highlight using the standard push_event:
def handle_info({:item_updated, item}, socket) do
  {:noreply, push_event(socket, "highlight", %{id: "item-#{item.id}"})}
end
Finally, a window event listener can listen for the event and conditionally
execute the highlight command if the element matches:
let liveSocket = new LiveSocket(...)
window.addEventListener("phx:highlight", (e) => {
  let el = document.getElementById(e.detail.id)
  if(el) {
    // logic for highlighting
  }
})
If you desire, you can also integrate this functionality with Phoenix'
JS commands, executing JS commands for the given element whenever highlight
is triggered. First, update the element to embed the JS command into a data
attribute:
<div id={"item-#{item.id}"} class="item" data-highlight={JS.transition("highlight")}>
  {item.title}
</div>
Now, in the event listener, use LiveSocket.execJS to trigger all JS
commands in the new attribute:
let liveSocket = new LiveSocket(...)
window.addEventListener("phx:highlight", (e) => {
  document.querySelectorAll(`[data-highlight]`).forEach(el => {
    if(el.id == e.detail.id){
      liveSocket.execJS(el, el.getAttribute("data-highlight"))
    }
  })
})

  
    
  
  Client hooks via phx-hook


To handle custom client-side JavaScript when an element is added, updated,
or removed by the server, a hook object may be provided via phx-hook.
phx-hook must point to an object with the following life-cycle callbacks:
	mounted - the element has been added to the DOM and its server
LiveView has finished mounting
	beforeUpdate - the element is about to be updated in the DOM.
Note: any call here must be synchronous as the operation cannot
be deferred or cancelled.
	updated - the element has been updated in the DOM by the server
	destroyed - the element has been removed from the page, either
by a parent update, or by the parent being removed entirely
	disconnected - the element's parent LiveView has disconnected from the server
	reconnected - the element's parent LiveView has reconnected to the server

Note: When using hooks outside the context of a LiveView, mounted is the only
callback invoked, and only those elements on the page at DOM ready will be tracked.
For dynamic tracking of the DOM as elements are added, removed, and updated, a LiveView
should be used.
The above life-cycle callbacks have in-scope access to the following attributes:
	el - attribute referencing the bound DOM node
	liveSocket - the reference to the underlying LiveSocket instance
	pushEvent(event, payload, (reply, ref) => ...) - method to push an event from the client to the LiveView server.
If no callback function is passed, a promise that resolves to the reply is returned.
	pushEventTo(selectorOrTarget, event, payload, (reply, ref) => ...) - method to push targeted events from the client
to LiveViews and LiveComponents. It sends the event to the LiveComponent or LiveView the selectorOrTarget is
defined in, where its value can be either a query selector or an actual DOM element. If the query selector returns
more than one element it will send the event to all of them, even if all the elements are in the same LiveComponent
or LiveView. pushEventTo supports passing the node element e.g. this.el instead of selector e.g. "#" + this.el.id
as the first parameter for target.
As there can be multiple targets, if no callback is passed, a promise is returned that matches the return value of
Promise.allSettled(). Individual fulfilled values are of the format { reply, ref }.
	handleEvent(event, (payload) => ...) - method to handle an event pushed from the server. Returns a value that can be passed to removeHandleEvent to remove the event handler.
	removeHandleEvent(ref) - method to remove an event handler added via handleEvent
	upload(name, files) - method to inject a list of file-like objects into an uploader.
	uploadTo(selectorOrTarget, name, files) - method to inject a list of file-like objects into an uploader.
The hook will send the files to the uploader with name defined by allow_upload/3
on the server-side. Dispatching new uploads triggers an input change event which will be sent to the
LiveComponent or LiveView the selectorOrTarget is defined in, where its value can be either a query selector or an
actual DOM element. If the query selector returns more than one live file input, an error will be logged.
	js() - returns an object with methods to manipluate the DOM and execute JavaScript. The applied changes integrate with server DOM patching. See JS commands.

For example, the markup for a controlled input for phone-number formatting could be written
like this:
<input type="text" name="user[phone_number]" id="user-phone-number" phx-hook="PhoneNumber" />
Then a hook callback object could be defined and passed to the socket:
/**
 * @type {import("phoenix_live_view").HooksOptions}
 */
let Hooks = {}
Hooks.PhoneNumber = {
  mounted() {
    this.el.addEventListener("input", e => {
      let match = this.el.value.replace(/\D/g, "").match(/^(\d{3})(\d{3})(\d{4})$/)
      if(match) {
        this.el.value = `${match[1]}-${match[2]}-${match[3]}`
      }
    })
  }
}

let liveSocket = new LiveSocket("/live", Socket, {hooks: Hooks, ...})
...
Note: when using phx-hook, a unique DOM ID must always be set.
For integration with client-side libraries which require a broader access to full
DOM management, the LiveSocket constructor accepts a dom option with an
onBeforeElUpdated callback. The fromEl and toEl DOM nodes are passed to the
function just before the DOM patch operations occurs in LiveView. This allows external
libraries to (re)initialize DOM elements or copy attributes as necessary as LiveView
performs its own patch operations. The update operation cannot be cancelled or deferred,
and the return value is ignored.
For example, the following option could be used to guarantee that some attributes set on the client-side are kept intact:
...
let liveSocket = new LiveSocket("/live", Socket, {
  params: {_csrf_token: csrfToken},
  hooks: Hooks,
  dom: {
    onBeforeElUpdated(from, to) {
      for (const attr of from.attributes) {
        if (attr.name.startsWith("data-js-")) {
          to.setAttribute(attr.name, attr.value);
        }
      }
    }
  }
})
In the example above, all attributes starting with data-js- won't be replaced when the DOM is patched by LiveView.
A hook can also be defined as a subclass of ViewHook:
import { ViewHook } from "phoenix_live_view"

class MyHook extends ViewHook {
  mounted() {
    ...
  }
}

let liveSocket = new LiveSocket(..., {
  hooks: {
    MyHook
  }
})

  
    
  
  Colocated Hooks / Colocated JavaScript


When writing components that require some more control over the DOM, it often feels inconvenient to
have to write a hook in a separate file. Instead, one wants to have the hook logic right next to the component
code. For such cases, HEEx supports Phoenix.LiveView.ColocatedHook and Phoenix.LiveView.ColocatedJS.
Let's see an example:
def phone_number_input(assigns) do
  ~H"""
  <input type="text" name="user[phone_number]" id="user-phone-number" phx-hook=".PhoneNumber" />
  <script :type={Phoenix.LiveView.ColocatedHook} name=".PhoneNumber">
    export default {
      mounted() {
        this.el.addEventListener("input", e => {
          let match = this.el.value.replace(/\D/g, "").match(/^(\d{3})(\d{3})(\d{4})$/)
          if(match) {
            this.el.value = `${match[1]}-${match[2]}-${match[3]}`
          }
        })
      }
    }
  </script>
  """
end
When LiveView finds a <script> element with :type={ColocatedHook}, it will extract the
hook code at compile time and write it into a special folder inside the _build/ directory.
To use the hooks, all that needs to be done is to import the manifest into your JS bundle,
which is automatically done in the app.js file generated by mix phx.new for new Phoenix 1.8 apps:
...
  import {Socket} from "phoenix"
  import {LiveSocket} from "phoenix_live_view"
  import topbar from "../vendor/topbar"
+ import {hooks as colocatedHooks} from "phoenix-colocated/my_app"

  let csrfToken = document.querySelector("meta[name='csrf-token']").getAttribute("content")
  let liveSocket = new LiveSocket("/live", Socket, {
    longPollFallbackMs: 2500,
    params: {_csrf_token: csrfToken},
+   hooks: {...colocatedHooks}
 })
The "phoenix-colocated" package is a folder inside the Mix.Project.build_path(),
which is included by default in the esbuild configuration of new
Phoenix projects (requires {:esbuild, "~> 0.10"} or later):
config :esbuild,
  ...
  my_app: [
    args:
      ~w(js/app.js --bundle --target=es2022 --outdir=../priv/static/assets/js --external:/fonts/* --external:/images/* --alias:@=.),
    cd: Path.expand("../assets", __DIR__),
    env: %{
      "NODE_PATH" => [Path.expand("../deps", __DIR__), Mix.Project.build_path()]
    }
  ]
When rendering a component that includes a colocated hook, the <script> tag is omitted
from the rendered output. Furthermore, to prevent conflicts with other components, colocated hooks
require you to use the special dot syntax when naming the hook, as well as in the phx-hook attribute.
LiveView will prefix the hook name by the current module name at compile time. This also means
that in cases where a hook is meant to be used in multiple components across a project, the hook
should be defined as a regular, non-colocated hook instead.
You can read more about colocated hooks in the module documentation for ColocatedHook.
LiveView also supports colocating other JavaScript code, for more information, see Phoenix.LiveView.ColocatedJS.

  
    
  
  Client-server communication


A hook can push events to the LiveView by using the pushEvent function and receive a
reply from the server via a {:reply, map, socket} return value. The reply payload will be
passed to the optional pushEvent response callback.
Communication with the hook from the server can be done by reading data attributes on the
hook element or by using Phoenix.LiveView.push_event/3 on the server and handleEvent on the client.
An example of responding with :reply might look like this.
<div phx-hook="ClickMeHook" id="click-me">
  Click me for a message!
</div>
Hooks.ClickMeHook = {
  mounted() {
    this.el.addEventListener("click", () => {
      // Push event to LiveView with callback for reply
      this.pushEvent("get_message", {}, (reply) => {
        console.debug(reply.message);
      });
    });
  }
}
Then in your callback you respond with {:reply, map, socket}
def handle_event("get_message", _params, socket) do
  # Use :reply to respond to the pushEvent
  {:reply, %{message: "Hello from LiveView!"}, socket}
end
Another example, to implement infinite scrolling, one can pass the current page using data attributes:
<div id="infinite-scroll" phx-hook="InfiniteScroll" data-page={@page}>
And then in the client:
/**
 * @type {import("phoenix_live_view").Hook}
 */
Hooks.InfiniteScroll = {
  page() { return this.el.dataset.page },
  mounted(){
    this.pending = this.page()
    window.addEventListener("scroll", e => {
      if(this.pending == this.page() && scrollAt() > 90){
        this.pending = this.page() + 1
        this.pushEvent("load-more", {})
      }
    })
  },
  updated(){ this.pending = this.page() }
}
However, the data attribute approach is not a good approach if you need to frequently push data to the client. To push out-of-band events to the client, for example to render charting points, one could do:
<div id="chart" phx-hook="Chart">
And then on the client:
/**
 * @type {import("phoenix_live_view").Hook}
 */
Hooks.Chart = {
  mounted(){
    this.handleEvent("points", ({points}) => MyChartLib.addPoints(points))
  }
}
And then you can push events as:
{:noreply, push_event(socket, "points", %{points: new_points})}
Events pushed from the server via push_event are global and will be dispatched
to all active hooks on the client who are handling that event. If you need to scope events
(for example when pushing from a live component that has siblings on the current live view),
then this must be done by namespacing them:
def update(%{id: id, points: points} = assigns, socket) do
  socket =
    socket
    |> assign(assigns)
    |> push_event("points-#{id}", points)

  {:ok, socket}
end
And then on the client:
Hooks.Chart = {
  mounted(){
    this.handleEvent(`points-${this.el.id}`, (points) => MyChartLib.addPoints(points));
  }
}
Note: In case a LiveView pushes events and renders content, handleEvent callbacks are invoked after the page is updated. Therefore, if the LiveView redirects at the same time it pushes events, callbacks won't be invoked on the old page's elements. Callbacks would be invoked on the redirected page's newly mounted hook elements.

  
    
  
  JS commands


Note: If possible, construct commands via Elixir using Phoenix.LiveView.JS and trigger them via Phoenix DOM Bindings.
While Phoenix.LiveView.JS allows you to construct a declarative representation of a command, it may not cover all use cases.
In addition, you can execute commands that integrate with server DOM patching via JavaScript using:
	Client hooks: this.js() or the
	LiveSocket instance: liveSocket.js().

The command interface returned by js() above offers the following functions:
	show(el, opts = {}) - shows an element. Options: display, transition, time, blocking. For more details, see Phoenix.LiveView.JS.show/1.
	hide(el, opts = {}) - hides an element. Options: transition, time, blocking. For more details, see Phoenix.LiveView.JS.hide/1.
	toggle(el, opts = {}) - toggles the visibility of an element. Options: display, in, out, time, blocking. For more details, see Phoenix.LiveView.JS.toggle/1.
	addClass(el, names, opts = {}) - adds CSS class(es) to an element. Options: transition, time, blocking. For more details, see Phoenix.LiveView.JS.add_class/1.
	removeClass(el, names, opts = {}) - removes CSS class(es) to an element. Options: transition, time, blocking. For more details, see Phoenix.LiveView.JS.remove_class/1.
	toggleClass(el, names, opts = {}) - toggles CSS class(es) to an element. Options: transition, time, blocking. For more details, see Phoenix.LiveView.JS.toggle_class/1.
	transition(el, transition, opts = {}) - applies a CSS transition to an element. Options: time, blocking. For more details, see Phoenix.LiveView.JS.transition/1.
	setAttribute(el, attr, val) - sets an attribute on an element
	removeAttribute(el, attr) - removes an attribute from an element
	toggleAttribute(el, attr, val1, val2) - toggles an attribute on an element between two values
	push(el, type, opts = {}) - pushes an event to the server. To target a LiveComponent by its ID, pass a separate target in the options. Options: target, loading, page_loading, value. For more details, see Phoenix.LiveView.JS.push/1.
	navigate(href, opts = {}) - sends a navigation event to the server and updates the browser's pushState history. Options: replace. For more details, see Phoenix.LiveView.JS.navigate/1.
	patch(href, opts = {}) - sends a patch event to the server and updates the browser's pushState history. Options: replace. For more details, see Phoenix.LiveView.JS.patch/1.
	exec(encodedJS) - only via Client hook this.js(): executes encoded JavaScript in the context of the hook's root node. The encoded JS command should be constructed via Phoenix.LiveView.JS and is usually stored as an HTML attribute. Example: this.js().exec(this.el.getAttribute('phx-remove')).
	exec(el, encodedJS) - only via liveSocket.js(): executes encoded JavaScript in the context of any element.



  

    
Syncing changes and optimistic UIs
    

When using LiveView, whenever you change the state in your LiveView process, changes are automatically sent and applied in the client.
However, in many occasions, the client may have its own state: inputs, buttons, focused UI elements, and more. In order to avoid server updates from destroying state on the client, LiveView provides several features and out-of-the-box conveniences.
Let's start by discussing which problems may arise from client-server integration, which may apply to any web application, and explore how LiveView solves it automatically. If you want to focus on the more practical aspects, you can jump to later sections or watch the video below:

  
    
  
  The problem in a nutshell


Imagine your web application has a form. The form has a single email input and a button. We have to validate that the email is unique in our database and render a tiny “✗” or “✓“ accordingly close to the input. Because we are using server-side rendering, we are debouncing/throttling form changes to the server. And, to avoid double-submissions, we want to disable the button as soon as it is clicked.
Here is what could happen. The user has typed “hello@example.” and debounce kicks in, causing the client to send an event to the server. Here is how the client looks like at this moment:
[ hello@example.    ]

    ------------
       SUBMIT
    ------------
While the server is processing this information, the user finishes typing the email and presses submit. The client sends the submit event to the server, then proceeds to disable the button, and change its value to “SUBMITTING”:
[ hello@example.com ]

    ------------
     SUBMITTING
    ------------
Immediately after pressing submit, the client receives an update from the server, but this is an update from the debounce event! If the client were to simply render this server update, the client would effectively roll back the form to the previous state shown below, which would be a disaster:
[ hello@example.    ] ✓

    ------------
       SUBMIT
    ------------
This is a simple example of how client and server state can evolve and differ for periods of times, due to the latency (distance) between them, in any web application, not only LiveView.
LiveView solves this in two ways:
	The JavaScript client is always the source of truth for current input values

	LiveView tracks how many events are currently in flight in a given input/button/form. The changes to the form are applied behind the scenes as they arrive, but LiveView only shows them once all in-flight events have been resolved


In other words, for the most common cases, LiveView will automatically sync client and server state for you. This is a huge benefit of LiveView, as many other stacks would require developers to tackle these problems themselves. For complete detail in how LiveView handles forms, see the JavaScript client specifics in the Form Bindings page.

  
    
  
  Optimistic UIs via loading classes


Whenever an HTML element pushes an event to the server, LiveView will attach a -loading class to it. For example the following markup:
<button phx-click="clicked" phx-window-keydown="key">...</button>
On click, would receive the phx-click-loading class, and on keydown would receive the phx-keydown-loading class. The CSS loading classes are maintained until an acknowledgement is received on the client for the pushed event. If the element is triggered several times, the loading state is removed only when all events are resolved.
This means the most trivial optimistic UI enhancements can be done in LiveView by simply adding a CSS rule. For example, imagine you want to fade the text of an element when it is clicked, while it waits for a response:
.phx-click-loading.opaque-on-click {
  opacity: 50%;
}
Now, by adding the class opaque-on-click to any element, the elements give an immediate feedback on click.
The following events receive CSS loading classes:
	phx-click - phx-click-loading
	phx-change - phx-change-loading
	phx-submit - phx-submit-loading
	phx-focus - phx-focus-loading
	phx-blur - phx-blur-loading
	phx-window-keydown - phx-keydown-loading
	phx-window-keyup - phx-keyup-loading

Events that happen inside a form have their state applied to both the element and the form. When an input changes, phx-change-loading applies to both input and form. On submit, both button and form get the phx-submit-loading classes. Buttons, in particular, also support a phx-disabled-with attribute, which allows you to customize the text of the button on click:
<button phx-disable-with="Submitting...">Submit</button>

  
    
  
  Tailwind integration


If you are using Tailwind, you may want to use the addVariant plugin to make it even easier to customize your elements loading state.
plugins: [
  plugin(({ addVariant }) => {
    addVariant("phx-click-loading", [".phx-click-loading&", ".phx-click-loading &",]);
    addVariant("phx-submit-loading", [".phx-submit-loading&", ".phx-submit-loading &",]);
    addVariant("phx-change-loading", [".phx-change-loading&", ".phx-change-loading &",]);
  }),
],
Now to fade one element on click, you simply need to add:
<button phx-click="clicked" class="phx-click-loading:opacity-50">...</button>

  
    
  
  Optimistic UIs via JS commands


While loading classes are extremely handy, they only apply to the element currently clicked. Sometimes, you may to click a "Delete" button but mark the whole row that holds the button as loading (for example, to fade it out).
By using JS commands, you can tell LiveView which elements get the loading state:
<button phx-click={JS.push("delete", loading: "#post-row-13")}>Delete</button>
Besides custom loading elements, you can use JS commands for a huge variety of operations, such as adding/removing classes, toggling attributes, hiding elements, transitions, and more.
For example, imagine that you want to immediately remove an element from the page on click, you can do this:
<button phx-click={JS.push("delete") |> JS.hide()}>Delete</button>
If the element you want to delete is not the clicked button, but its parent (or other element), you can pass a selector to hide:
<button phx-click={JS.push("delete") |> JS.hide("#post-row-13")}>Delete</button>
Or if you'd rather add a class instead:
<button phx-click={JS.push("delete") |> JS.add_class("opacity-50")}>Delete</button>
One key property of JS commands, such as hide and add_class, is that they are DOM-patch aware, so operations applied by the JS APIs will stick to elements across patches from the server.
JS commands also include a dispatch function, which dispatches an event to the DOM element to trigger client-specific functionality. For example, to trigger copying to a clipboard, you may implement this event listener:
window.addEventListener("app:clipcopy", (event) => {
  if ("clipboard" in navigator) {
    if (event.target.tagName === "INPUT") {
      navigator.clipboard.writeText(event.target.value);
    } else {
      navigator.clipboard.writeText(event.target.textContent);
    }
  } else {
    alert(
      "Sorry, your browser does not support clipboard copy.\nThis generally requires a secure origin — either HTTPS or localhost.",
    );
  }
});
And then trigger it as follows:
<button phx-click={JS.dispatch("app:clipcopy", to: "#printed-output")}>Copy</button>
Transitions are also only a few characters away:
<div id="item">My Item</div>
<button phx-click={JS.transition("shake", to: "#item")}>Shake!</button>
See Phoenix.LiveView.JS for more examples and documentation.

  
    
  
  Optimistic UIs via JS hooks


On the most complex cases, you can assume control of a DOM element, and control exactly how and when server updates apply to the element on the page. See the Client hooks via phx-hook section in the JavaScript interoperability page to learn more.

  
    
  
  Live navigation


LiveView also provides mechanisms to customize and interact with navigation events.

  
    
  
  Navigation classes


The following classes are applied to the LiveView's parent container:
	"phx-connected" - applied when the view has connected to the server
	"phx-loading" - applied when the view is not connected to the server
	"phx-error" - applied when an error occurs on the server. Note, this
class will be applied in conjunction with "phx-loading" if connection
to the server is lost.


  
    
  
  Navigation events


For live page navigation via <.link navigate={...}> and <.link patch={...}>, their server-side equivalents push_navigate and push_patch, as well as form submits via phx-submit, the JavaScript events "phx:page-loading-start" and "phx:page-loading-stop" are dispatched on window. This is useful for showing main page loading status, for example:
// app.js
import topbar from "topbar"
window.addEventListener("phx:page-loading-start", info => topbar.delayedShow(500))
window.addEventListener("phx:page-loading-stop", info => topbar.hide())
Within the callback, info.detail will be an object that contains a kind
key, with a value that depends on the triggering event:
	"redirect" - the event was triggered by a redirect
	"patch" - the event was triggered by a patch
	"initial" - the event was triggered by initial page load
	"element" - the event was triggered by a phx- bound element, such as phx-click
	"error" - the event was triggered by an error, such as a view crash or socket disconnection

Additionally, Phoenix.LiveView.JS.push/3 may dispatch page loading events by passing page_loading: true option.
For all kinds of page loading events, all but "element" will receive an additional to key in the info metadata pointing to the href associated with the page load. In the case of an "element" page loading event, the info will contain a "target" key containing the DOM element which triggered the page loading state.
A lower level phx:navigate event is also triggered any time the browser's URL bar is programmatically changed by Phoenix or the user navigation forward or back. The info.detail will contain the following information:
	"href" - the location the URL bar was navigated to.
	"patch" - the boolean flag indicating this was a patch navigation.
	"pop" - the boolean flag indication this was a navigation via popstate
from a user navigation forward or back in history.



  

    
<code class="inline">phx-</code> HTML attributes
    

A summary of special HTML attributes used in Phoenix LiveView templates.
Each attribute is linked to its documentation for more details.

  
    
  
  Event Handlers


Attribute values can be:
	An event name for the handle_event server callback
	JS commands to be executed directly on the client

Use phx-value-* attributes to pass params to the server.

Use phx-debounce and phx-throttle to control the frequency of events.


  
    
  
  Click


	Attributes
	phx-click phx-click-away


  
    
  
  Focus


	Attributes
	phx-blur phx-focus
	phx-window-blur phx-window-focus


  
    
  
  Keyboard


	Attributes
	phx-keydown phx-keyup
	phx-window-keydown phx-window-keyup

Use the phx-key attribute to listen to specific keys.


  
    
  
  Scroll


	Attributes
	phx-viewport-top phx-viewport-bottom


  
    
  
  Example


lib/hello_web/live/hello_live.html.heex
<button type="button" phx-click="click" phx-value-user={@current_user.id}>Click Me</button>
<button type="button" phx-click={JS.toggle(to: "#example")}>Toggle</button>

  
    
  
  Form Event Handlers



  
    
  
  On <form> elements


	Attribute	Value
	phx-change	Event name or JS commands
	phx-submit	Event name or JS commands
	phx-auto-recover	Event name, JS commands or "ignore"
	phx-trigger-action	true or false


  
    
  
  On <button> elements


	Attribute	Value
	phx-disable-with	Text to show during event submission


  
    
  
  Form Example


lib/hello_web/live/hello_live.html.heex
<form phx-change="validate" phx-submit="save">
  <input type="text" name="name" phx-debounce="500" phx-throttle="500" />
  <button type="submit" phx-disable-with="Saving...">Save</button>
</form>

  
    
  
  Socket Connection Lifecycle


	Attribute	Value
	phx-connected	JS commands executed after the LiveSocket connects
	phx-disconnected	JS commands executed after the LiveSocket disconnects

lib/hello_web/live/hello_live.html.heex
<div id="status" class="hidden" phx-disconnected={JS.show()} phx-connected={JS.hide()}>
  Attempting to reconnect...
</div>

  
    
  
  DOM Element Lifecycle


	Attribute	Value
	phx-mounted	JS commands executed after the element is mounted
	phx-remove	JS commands executed during the element removal
	phx-update	"replace" (default), "stream" or "ignore", configures DOM patching behavior

lib/hello_web/live/hello_live.html.heex
<div
  id="iframe-container"
  phx-mounted={JS.transition("animate-bounce", time: 2000)}
  phx-remove={JS.hide(transition: {"transition-all transform ease-in duration-200", "opacity-100", "opacity-0"})}
>
  <button type="button" phx-click={JS.exec("phx-remove", to: "#iframe-container")}>Hide</button>
  <iframe id="iframe" src="https://example.com" phx-update="ignore"></iframe>
</div>

  
    
  
  Client Hooks


	Attribute	Value
	phx-hook	The name of a previously defined JavaScript hook in the LiveSocket

Client hooks provide bidirectional communication between client and server using
this.pushEvent and this.handleEvent to send and receive events.
lib/hello_web/live/hello_live.html.heex
<div id="example" phx-hook="Example">
  <h1>Events</h1>
  <ul id="example-events"></ul>
</div>
assets/js/app.js
let Hooks = {}
Hooks.Example = {
  // Callbacks
  mounted()      { this.appendEvent("Mounted") },
  beforeUpdate() { this.appendEvent("Before Update") },
  updated()      { this.appendEvent("Updated") },
  destroyed()    { this.appendEvent("Destroyed") },
  disconnected() { this.appendEvent("Disconnected") },
  reconnected()  { this.appendEvent("Reconnected") },

  // Custom Helper
  appendEvent(name) {
    console.log(name)
    let li = document.createElement("li")
    li.innerText = name
    this.el.querySelector("#example-events").appendChild(li)
  }
}

let liveSocket = new LiveSocket("/live", Socket, {hooks: Hooks})

  
    
  
  Tracking Static Assets


	Attribute	Value
	phx-track-static	None, used to annotate static files

lib/hello_web/components/layouts/root.html.heex
<link phx-track-static rel="stylesheet" href={~p"/assets/app.css"} />
<script defer phx-track-static type="text/javascript" src={~p"/assets/app.js"}></script>


  

    
Phoenix.Component 
    



      
Define reusable function components with HEEx templates.
A function component is any function that receives an assigns
map as an argument and returns a rendered struct built with
the ~H sigil:
defmodule MyComponent do
  # In Phoenix apps, the line is typically: use MyAppWeb, :html
  use Phoenix.Component

  def greet(assigns) do
    ~H"""
    <p>Hello, {@name}!</p>
    """
  end
end
This function uses the ~H sigil to return a rendered template.
~H stands for HEEx (HTML + EEx). HEEx is a template language for
writing HTML mixed with Elixir interpolation. We can write Elixir
code inside {...} for HTML-aware interpolation inside tag attributes
and the body. We can also interpolate arbitrary HEEx blocks using <%= ... %>
We use @name to access the key name defined inside assigns.
When invoked within a ~H sigil or HEEx template file:
<MyComponent.greet name="Jane" />
The following HTML is rendered:
<p>Hello, Jane!</p>
If the function component is defined locally, or its module is imported,
then the caller can invoke the function directly without specifying the module:
<.greet name="Jane" />
For dynamic values, you can interpolate Elixir expressions into a function component:
<.greet name={@user.name} />
Function components can also accept blocks of HEEx content (more on this later):
<.card>
  <p>This is the body of my card!</p>
</.card>
In this module we will learn how to build rich and composable components to
use in our applications.

  
    
  
  Attributes


Phoenix.Component provides the attr/3 macro to declare what attributes the proceeding function
component expects to receive when invoked:
attr :name, :string, required: true

def greet(assigns) do
  ~H"""
  <p>Hello, {@name}!</p>
  """
end
By calling attr/3, it is now clear that greet/1 requires a string attribute called name
present in its assigns map to properly render. Failing to do so will result in a compilation
warning:
<MyComponent.greet />
  <!-- warning: missing required attribute "name" for component MyAppWeb.MyComponent.greet/1
           lib/app_web/my_component.ex:15 -->
Attributes can provide default values that are automatically merged into the assigns map:
attr :name, :string, default: "Bob"
Now you can invoke the function component without providing a value for name:
<.greet />
Rendering the following HTML:
<p>Hello, Bob!</p>
Accessing an attribute which is required and does not have a default value will fail.
You must explicitly declare default: nil or assign a value programmatically with the
assign_new/3 function.
Multiple attributes can be declared for the same function component:
attr :name, :string, required: true
attr :age, :integer, required: true

def celebrate(assigns) do
  ~H"""
  <p>
    Happy birthday {@name}!
    You are {@age} years old.
  </p>
  """
end
Allowing the caller to pass multiple values:
<.celebrate name={"Genevieve"} age={34} />
Rendering the following HTML:
<p>
  Happy birthday Genevieve!
  You are 34 years old.
</p>
Multiple function components can be defined in the same module, with different attributes. In the
following example, <Components.greet/> requires a name, but does not require a title, and
<Components.heading> requires a title, but does not require a name.
defmodule Components do
  # In Phoenix apps, the line is typically: use MyAppWeb, :html
  use Phoenix.Component

  attr :title, :string, required: true

  def heading(assigns) do
    ~H"""
    <h1>{@title}</h1>
    """
  end

  attr :name, :string, required: true

  def greet(assigns) do
    ~H"""
    <p>Hello {@name}</p>
    """
  end
end
With the attr/3 macro you have the core ingredients to create reusable function components.
But what if you need your function components to support dynamic attributes, such as common HTML
attributes to mix into a component's container?

  
    
  
  Global attributes


Global attributes are a set of attributes that a function component can accept when it
declares an attribute of type :global. By default, the set of attributes accepted are those
attributes common to all standard HTML tags.
See Global attributes
for a complete list of attributes.
Once a global attribute is declared, any number of attributes in the set can be passed by
the caller without having to modify the function component itself.
Below is an example of a function component that accepts a dynamic number of global attributes:
attr :message, :string, required: true
attr :rest, :global

def notification(assigns) do
  ~H"""
  <span {@rest}>{@message}</span>
  """
end
The caller can pass multiple global attributes (such as phx-* bindings or the class attribute):
<.notification message="You've got mail!" class="bg-green-200" phx-click="close" />
Rendering the following HTML:
<span class="bg-green-200" phx-click="close">You've got mail!</span>
Note that the function component did not have to explicitly declare a class or phx-click
attribute in order to render.
Global attributes can define defaults which are merged with attributes provided by the caller.
For example, you may declare a default class if the caller does not provide one:
attr :rest, :global, default: %{class: "bg-blue-200"}
Now you can call the function component without a class attribute:
<.notification message="You've got mail!" phx-click="close" />
Rendering the following HTML:
<span class="bg-blue-200" phx-click="close">You've got mail!</span>
Note that the global attribute cannot be provided directly and doing so will emit
a warning. In other words, this is invalid:
<.notification message="You've got mail!" rest={%{"phx-click" => "close"}} />

  
    
  
  Included globals


You may also specify which attributes are included in addition to the known globals
with the :include option. For example to support the form attribute on a button
component:
# <.button form="my-form"/>
attr :rest, :global, include: ~w(form)
slot :inner_block
def button(assigns) do
  ~H"""
  <button {@rest}>{render_slot(@inner_block)}</button>
  """
end
The :include option is useful to apply global additions on a case-by-case basis,
but sometimes you want to extend existing components with new global attributes,
such as Alpine.js' x- prefixes, which we'll outline next.

  
    
  
  Custom global attribute prefixes


You can extend the set of global attributes by providing a list of attribute prefixes to
use Phoenix.Component. Like the default attributes common to all HTML elements,
any number of attributes that start with a global prefix will be accepted by function
components invoked by the current module. By default, the following prefixes are supported:
phx-, aria-, and data-. For example, to support the x- prefix used by
Alpine.js, you can pass the :global_prefixes option to
use Phoenix.Component:
use Phoenix.Component, global_prefixes: ~w(x-)
In your Phoenix application, this is typically done in your
lib/my_app_web.ex file, inside the def html definition:
def html do
  quote do
    use Phoenix.Component, global_prefixes: ~w(x-)
    # ...
  end
end
Now all function components invoked by this module will accept any number of attributes
prefixed with x-, in addition to the default global prefixes.
You can learn more about attributes by reading the documentation for attr/3.

  
    
  
  Slots


In addition to attributes, function components can accept blocks of HEEx content, referred to
as slots. Slots enable further customization of the rendered HTML, as the caller can pass the
function component HEEx content they want the component to render. Phoenix.Component provides
the slot/3 macro used to declare slots for function components:
slot :inner_block, required: true

def button(assigns) do
  ~H"""
  <button>
    {render_slot(@inner_block)}
  </button>
  """
end
The expression render_slot(@inner_block) renders the HEEx content. You can invoke this function
component like so:
<.button>
  This renders <strong>inside</strong> the button!
</.button>
Which renders the following HTML:
<button>
  This renders <strong>inside</strong> the button!
</button>
Like the attr/3 macro, using the slot/3 macro will provide compile-time validations.
For example, invoking button/1 without a slot of HEEx content will result in a compilation
warning being emitted:
<.button />
  <!-- warning: missing required slot "inner_block" for component MyAppWeb.MyComponent.button/1
           lib/app_web/my_component.ex:15 -->

  
    
  
  The default slot


The example above uses the default slot, accessible as an assign named @inner_block, to render
HEEx content via the render_slot/1 function.
If the values rendered in the slot need to be dynamic, you can pass a second value back to the
HEEx content by calling render_slot/2:
slot :inner_block, required: true

attr :entries, :list, default: []

def unordered_list(assigns) do
  ~H"""
  <ul>
    <li :for={entry <- @entries}>{render_slot(@inner_block, entry)}</li>
  </ul>
  """
end
When invoking the function component, you can use the special attribute :let to take the value
that the function component passes back and bind it to a variable:
<.unordered_list :let={fruit} entries={~w(apples bananas cherries)}>
  I like <b>{fruit}</b>!
</.unordered_list>
Rendering the following HTML:
<ul>
  <li>I like <b>apples</b>!</li>
  <li>I like <b>bananas</b>!</li>
  <li>I like <b>cherries</b>!</li>
</ul>
Now the separation of concerns is maintained: the caller can specify multiple values in a list
attribute without having to specify the HEEx content that surrounds and separates them.

  
    
  
  Named slots


In addition to the default slot, function components can accept multiple, named slots of HEEx
content. For example, imagine you want to create a modal that has a header, body, and footer:
slot :header
slot :inner_block, required: true
slot :footer, required: true

def modal(assigns) do
  ~H"""
  <div class="modal">
    <div class="modal-header">
      {render_slot(@header) || "Modal"}
    </div>
    <div class="modal-body">
      {render_slot(@inner_block)}
    </div>
    <div class="modal-footer">
      {render_slot(@footer)}
    </div>
  </div>
  """
end
You can invoke this function component using the named slot HEEx syntax:
<.modal>
  This is the body, everything not in a named slot is rendered in the default slot.
  <:footer>
    This is the bottom of the modal.
  </:footer>
</.modal>
Rendering the following HTML:
<div class="modal">
  <div class="modal-header">
    Modal.
  </div>
  <div class="modal-body">
    This is the body, everything not in a named slot is rendered in the default slot.
  </div>
  <div class="modal-footer">
    This is the bottom of the modal.
  </div>
</div>
As shown in the example above, render_slot/1 returns nil when an optional slot
is declared and none is given. This can be used to attach default behaviour.

  
    
  
  Slot attributes


Unlike the default slot, it is possible to pass a named slot multiple pieces of HEEx content.
Named slots can also accept attributes, defined by passing a block to the slot/3 macro.
If multiple pieces of content are passed, render_slot/2 will merge and render all the values.
Below is a table component illustrating multiple named slots with attributes:
slot :column, doc: "Columns with column labels" do
  attr :label, :string, required: true, doc: "Column label"
end

attr :rows, :list, default: []

def table(assigns) do
  ~H"""
  <table>
    <tr>
      <th :for={col <- @column}>{col.label}</th>
    </tr>
    <tr :for={row <- @rows}>
      <td :for={col <- @column}>{render_slot(col, row)}</td>
    </tr>
  </table>
  """
end
You can invoke this function component like so:
<.table rows={[%{name: "Jane", age: "34"}, %{name: "Bob", age: "51"}]}>
  <:column :let={user} label="Name">
    {user.name}
  </:column>
  <:column :let={user} label="Age">
    {user.age}
  </:column>
</.table>
Rendering the following HTML:
<table>
  <tr>
    <th>Name</th>
    <th>Age</th>
  </tr>
  <tr>
    <td>Jane</td>
    <td>34</td>
  </tr>
  <tr>
    <td>Bob</td>
    <td>51</td>
  </tr>
</table>
You can learn more about slots and the slot/3 macro in its documentation.

  
    
  
  Embedding external template files


The embed_templates/1 macro can be used to embed .html.heex files
as function components. The directory path is based on the current
module (__DIR__), and a wildcard pattern may be used to select all
files within a directory tree. For example, imagine a directory listing:
├── components.ex
├── cards
│   ├── pricing_card.html.heex
│   └── features_card.html.heex
Then you can embed the page templates in your components.ex module
and call them like any other function component:
defmodule MyAppWeb.Components do
  use Phoenix.Component

  embed_templates "cards/*"

  def landing_hero(assigns) do
    ~H"""
    <.pricing_card />
    <.features_card />
    """
  end
end
See embed_templates/1 for more information, including declarative
assigns support for embedded templates.

  
    
  
  Debug information


HEEx templates support adding annotations and locations to the rendered
page, which are special HTML comments and attributes that help you identify
where markup in your HTML document is rendered within your function component
tree.
For example, imagine the following HEEx template:
<.header>
  <.button>Click</.button>
</.header>
By turning on debug_heex_annotations, the HTML document would receive the
following comments when debug annotations are enabled:
<!-- @caller lib/app_web/home_live.ex:20 -->
<!-- <AppWeb.CoreComponents.header> lib/app_web/core_components.ex:123 -->
<header class="p-5">
  <!-- @caller lib/app_web/home_live.ex:48 -->
  <!-- <AppWeb.CoreComponents.button> lib/app_web/core_components.ex:456 -->
  <button class="px-2 bg-indigo-500 text-white">Click</button>
  <!-- </AppWeb.CoreComponents.button> -->
</header>
<!-- </AppWeb.CoreComponents.header> -->
Similarly, you can also turn on :debug_tags_location, which adds a
data-phx-loc attribute with the line of where each HTML tag is defined:
<header data-phx-loc="125" class="p-5">
  <button data-phx-loc="458" class="px-2 bg-indigo-500 text-white">Click</button>
</header>
These features work on any ~H or .html.heex template. They can be enabled
globally with the following configuration in your config/dev.exs file:
config :phoenix_live_view,
  debug_heex_annotations: true,
  debug_tags_location: true
Changing this configuration will require mix clean and a full recompile.

  
    
  
  Dynamic Component Rendering


Sometimes you might need to decide at runtime which component to render.
Because function components are just regular functions, we can leverage
Elixir's apply/3 function to dynamically call a module and/or function passed
in as an assign.
For example, using the following function component definition:
attr :module, :atom, required: true
attr :function, :atom, required: true
# any shared attributes
attr :shared, :string, required: true

# any shared slots
slot :named_slot, required: true
slot :inner_block, required: true

def dynamic_component(assigns) do
  {mod, assigns} = Map.pop(assigns, :module)
  {func, assigns} = Map.pop(assigns, :function)

  apply(mod, func, [assigns])
end
Then you can use the dynamic_component function like so:
<.dynamic_component
  module={MyAppWeb.MyModule}
  function={:my_function}
  shared="Yay Elixir!"
>
  <p>Howdy from the inner block!</p>
  <:named_slot>
    <p>Howdy from the named slot!</p>
  </:named_slot>
</.dynamic_component>
This will call the MyAppWeb.MyModule.my_function/1 function passing in the remaining assigns.
defmodule MyAppWeb.MyModule do
  attr :shared, :string, required: true

  slot :named_slot, required: true
  slot :inner_block, required: true

  def my_function(assigns) do
    ~H"""
    <p>Dynamic component with shared assigns: {@shared}</p>
    {render_slot(@inner_block)}
    {render_slot(@named_slot)}
    """
  end
end
Resulting in the following HTML:
<p>Dynamic component with shared assigns: Yay Elixir!</p>
<p>Howdy from the inner block!</p>
<p>Howdy from the named slot!</p>
Note that to get the most out of Phoenix.Component's compile-time validations, it is beneficial to
define such a dynamic_component for a specific set of components sharing the same API, instead of
defining it for the general case.
In this example, we defined our dynamic_component to expect an assign called shared, as well as
two slots that all components we want to use with it must implement.
The called my_function component's attribute and slot definitions cannot be validated through the apply call.

      


      
        Summary


  
    Components
  


    
      
        async_result(assigns)

      


        Renders a Phoenix.LiveView.AsyncResult struct (e.g. from Phoenix.LiveView.assign_async/4)
with slots for the different loading states.
The result state takes precedence over subsequent loading and failed
states.



    


    
      
        dynamic_tag(assigns)

      


        Generates a dynamically named HTML tag.



    


    
      
        focus_wrap(assigns)

      


        Wraps tab focus around a container for accessibility.



    


    
      
        form(assigns)

      


        Renders a form.



    


    
      
        inputs_for(assigns)

      


        Renders nested form inputs for associations or embeds.



    


    
      
        intersperse(assigns)

      


        Intersperses separator slot between an enumerable.



    


    
      
        link(assigns)

      


        Generates a link to a given route.



    


    
      
        live_component(assigns)

      


        A function component for rendering Phoenix.LiveComponent within a parent LiveView.



    


    
      
        live_file_input(assigns)

      


        Builds a file input tag for a LiveView upload.



    


    
      
        live_img_preview(assigns)

      


        Generates an image preview on the client for a selected file.



    


    
      
        live_title(assigns)

      


        Renders a title with automatic prefix/suffix on @page_title updates.



    





  
    Macros
  


    
      
        attr(name, type, opts \\ [])

      


        Declares attributes for a HEEx function components.



    


    
      
        embed_templates(pattern, opts \\ [])

      


        Embeds external template files into the module as function components.



    


    
      
        sigil_H(arg, modifiers)

      


        The ~H sigil for writing HEEx templates inside source files.



    


    
      
        slot(name, opts \\ [])

      


        Declares a slot. See slot/3 for more information.



    


    
      
        slot(name, opts, block)

      


        Declares a function component slot.



    





  
    Functions
  


    
      
        assign(socket_or_assigns, keyword_or_map)

      


        Adds key-value pairs to assigns.



    


    
      
        assign(socket_or_assigns, key, value)

      


        Adds a key-value pair to socket_or_assigns.



    


    
      
        assign_new(socket_or_assigns, key, fun)

      


        Assigns the given key with value from fun into socket_or_assigns if one does not yet exist.



    


    
      
        assigns_to_attributes(assigns, exclude \\ [])

      


        Filters the assigns as a list of keywords for use in dynamic tag attributes.



    


    
      
        changed?(socket_or_assigns, key)

      


        Checks if the given key changed in socket_or_assigns.



    


    
      
        live_flash(other, key)

          deprecated

      


        Returns the flash message from the LiveView flash assign.



    


    
      
        live_render(conn_or_socket, view, opts \\ [])

      


        Renders a LiveView within a template.



    


    
      
        portal(assigns)

      


        Renders a portal.



    


    
      
        render_slot(slot, argument \\ nil)

      


        Renders a slot entry with the given optional argument.



    


    
      
        to_form(data_or_params, options \\ [])

      


        Converts a given data structure to a Phoenix.HTML.Form.



    


    
      
        update(socket_or_assigns, key, fun)

      


        Updates an existing key with fun in the given socket_or_assigns.



    


    
      
        upload_errors(conf)

      


        Returns errors for the upload as a whole.



    


    
      
        upload_errors(conf, entry)

      


        Returns errors for the upload entry.



    


    
      
        used_input?(form_field)

      


        Checks if the input field was used by the client.



    





      


      
        Components

        


  
    
      
    
    
      async_result(assigns)



        
          
        

    

  


  

Renders a Phoenix.LiveView.AsyncResult struct (e.g. from Phoenix.LiveView.assign_async/4)
with slots for the different loading states.
The result state takes precedence over subsequent loading and failed
states.
Note
The inner block receives the result of the async assign as a :let.
The let is only accessible to the inner block and is not in scope to the
other slots.

  
    
  
  Examples


def mount(%{"slug" => slug}, _, socket) do
  {:ok,
    socket
    |> assign_async(:org, fn -> {:ok, %{org: fetch_org!(slug)}} end)}
end
<.async_result :let={org} assign={@org}>
  <:loading>Loading organization...</:loading>
  <:failed :let={_failure}>there was an error loading the organization</:failed>
  <%= if org do %>
    {org.name}
  <% else %>
    You don't have an organization yet.
  <% end %>
</.async_result>
See Async Operations for more information.
To display loading and failed states again on subsequent assign_async calls,
reset the assign to a result-free %AsyncResult{}:
{:noreply,
  socket
  |> assign_async(:page, :data, &reload_data/0)
  |> assign(:page, AsyncResult.loading())}

  
    
  
  Attributes


	assign (Phoenix.LiveView.AsyncResult) (required)


  
    
  
  Slots


	loading - rendered while the assign is loading for the first time.
	failed - rendered when an error or exit is caught or assign_async returns {:error, reason} for the first time. Receives the error as a :let.
	inner_block - rendered when the assign is loaded successfully via AsyncResult.ok/2. Receives the result as a :let.


  



  
    
      
    
    
      dynamic_tag(assigns)



        
          
        

    

  


  

Generates a dynamically named HTML tag.
Raises an ArgumentError if the tag name is found to be unsafe HTML.

  
    
  
  Attributes


	tag_name (:string) (required) - The name of the tag, such as div.
	name (:string) - Deprecated: use tag_name instead. If tag_name is used, passed to the tag. Otherwise the name of the tag, such as div.
	Global attributes are accepted. Additional HTML attributes to add to the tag, ensuring proper escaping.


  
    
  
  Slots


	inner_block


  
    
  
  Examples


<.dynamic_tag tag_name="input" name="my-input" type="text"/>
<input name="my-input" type="text"/>
<.dynamic_tag tag_name="p">content</.dynamic_tag>
<p>content</p>

  



  
    
      
    
    
      focus_wrap(assigns)



        
          
        

    

  


  

Wraps tab focus around a container for accessibility.
This is an essential accessibility feature for interfaces such as modals, dialogs, and menus.

  
    
  
  Attributes


	id (:string) (required) - The DOM identifier of the container tag.
	Global attributes are accepted. Additional HTML attributes to add to the container tag.


  
    
  
  Slots


	inner_block (required) - The content rendered inside of the container tag.


  
    
  
  Examples


Simply render your inner content within this component and focus will be wrapped around the
container as the user tabs through the containers content:
<.focus_wrap id="my-modal" class="bg-white">
  <div id="modal-content">
    Are you sure?
    <button phx-click="cancel">Cancel</button>
    <button phx-click="confirm">OK</button>
  </div>
</.focus_wrap>

  



  
    
      
    
    
      form(assigns)



        
          
        

    

  


  

Renders a form.
This function receives a Phoenix.HTML.Form struct, generally created with
to_form/2, and generates the relevant form tags. It can be used either
inside LiveView or outside.
To see how forms work in practice, you can run
mix phx.gen.live Blog Post posts title body:text inside your Phoenix
application, which will setup the necessary database tables and LiveViews
to manage your data.


  
    
  
  Examples: inside LiveView


Inside LiveViews, this function component is typically called with
as for={@form}, where @form is the result of the to_form/1 function.
to_form/1 expects either a map or an Ecto.Changeset
as the source of data and normalizes it into Phoenix.HTML.Form structure.
For example, you may use the parameters received in a
Phoenix.LiveView.handle_event/3 callback to create an Ecto changeset
and then use to_form/1 to convert it to a form. Then, in your templates,
you pass the @form as argument to :for:
<.form
  for={@form}
  phx-change="change_name"
>
  <.input field={@form[:email]} />
</.form>
The .input component is generally defined as part of your own application
and adds all styling necessary:
def input(assigns) do
  ~H"""
  <input type="text" name={@field.name} id={@field.id} value={@field.value} class="..." />
  """
end
A form accepts multiple options. For example, if you are doing file uploads
and you want to capture submissions, you might write instead:
<.form
  for={@form}
  multipart
  phx-change="change_user"
  phx-submit="save_user"
>
  ...
  <input type="submit" value="Save" />
</.form>
Notice how both examples use phx-change. The LiveView must implement the
phx-change event and store the input values as they arrive on change.
This is important because, if an unrelated change happens on the page,
LiveView should re-render the inputs with their updated values. Without phx-change,
the inputs would otherwise be cleared. Alternatively, you can use phx-update="ignore"
on the form to discard any updates.

  
    
  
  Using the for attribute


The for attribute can also be a map or an Ecto.Changeset. In such cases,
a form will be created on the fly, and you can capture it using :let:
<.form
  :let={form}
  for={@changeset}
  phx-change="change_user"
>
However, such approach is discouraged in LiveView for two reasons:
	LiveView can better optimize your code if you access the form fields
using @form[:field] rather than through the let-variable form

	Ecto changesets are meant to be single use. By never storing the changeset
in the assign, you will be less tempted to use it across operations



  
    
  
  A note on :errors


Even if changeset.errors is non-empty, errors will not be displayed in a
form if the changeset
:action
is nil or :ignore.
This is useful for things like validation hints on form fields, e.g. an empty
changeset for a new form. That changeset isn't valid, but we don't want to
show errors until an actual user action has been performed.
For example, if the user submits and a Repo.insert/1 is called and fails on
changeset validation, the action will be set to :insert to show that an
insert was attempted, and the presence of that action will cause errors to be
displayed. The same is true for Repo.update/delete.
Error visibility is handled by providing the action to to_form/2, which will
set the underlying changeset action. You can also set the action manually by
directly updating on the Ecto.Changeset struct field, or by using
Ecto.Changeset.apply_action/2. Since the action can be arbitrary, you can
set it to :validate or anything else to avoid giving the impression that a
database operation has actually been attempted.

  
    
  
  Displaying errors on used and unused input fields


Used inputs are only those inputs that have been focused, interacted with, or
submitted by the client. In most cases, a user shouldn't receive error feedback
for forms they haven't yet interacted with, until they submit the form. Filtering
the errors based on used input fields can be done with used_input?/1.

  
    
  
  Example: outside LiveView (regular HTTP requests)


The form component can still be used to submit forms outside of LiveView.
In such cases, the standard HTML action attribute MUST be given.
Without said attribute, the form method and csrf token are discarded.
<.form :let={f} for={@changeset} action={~p"/comments/#{@comment}"}>
  <.input field={f[:body]} />
</.form>
In the example above, we passed a changeset to for and captured
the value using :let={f}. This approach is ok outside of LiveViews,
as there are no change tracking optimizations to consider.

  
    
  
  CSRF protection


CSRF protection is a mechanism to ensure that the user who rendered
the form is the one actually submitting it. This module generates a
CSRF token by default. Your application should check this token on
the server to avoid attackers from making requests on your server on
behalf of other users. Phoenix by default checks this token.
When posting a form with a host in its address, such as "//host.com/path"
instead of only "/path", Phoenix will include the host signature in the
token and validate the token only if the accessed host is the same as
the host in the token. This is to avoid tokens from leaking to third
party applications. If this behaviour is problematic, you can generate
a non-host specific token with Plug.CSRFProtection.get_csrf_token/0 and
pass it to the form generator via the :csrf_token option.

  
    
  
  Attributes


	for (:any) (required) - An existing form or the form source data.

	action (:string) - The action to submit the form on.
This attribute must be given if you intend to submit the form to a URL without LiveView.

	as (:atom) - The prefix to be used in names and IDs generated by the form.
For example, setting as: :user_params means the parameters
will be nested "user_params" in your handle_event or
conn.params["user_params"] for regular HTTP requests.
If you set this option, you must capture the form with :let.

	csrf_token (:any) - A token to authenticate the validity of requests.
One is automatically generated when an action is given and the method is not get.
When set to false, no token is generated.

	errors (:list) - Use this to manually pass a keyword list of errors to the form.
This option is useful when a regular map is given as the form
source and it will make the errors available under f.errors.
If you set this option, you must capture the form with :let.

	method (:string) - The HTTP method.
It is only used if an :action is given. If the method is not get nor post,
an input tag with name _method is generated alongside the form tag.
If an :action is given with no method, the method will default to the return value
of Phoenix.HTML.FormData.to_form/2 (usually post).

	multipart (:boolean) - Sets enctype to multipart/form-data.
Required when uploading files.
Defaults to false.

	Global attributes are accepted. Additional HTML attributes to add to the form tag. Supports all globals plus: ["autocomplete", "name", "rel", "enctype", "novalidate", "target"].



  
    
  
  Slots


	inner_block (required) - The content rendered inside of the form tag.


  



  
    
      
    
    
      inputs_for(assigns)



        
          
        

    

  


  

Renders nested form inputs for associations or embeds.

  
    
  
  Attributes


	field (Phoenix.HTML.FormField) (required) - A %Phoenix.HTML.Form{}/field name tuple, for example: {@form[:email]}.

	id (:string) - The id base to be used in the form inputs. Defaults to the parent form id. The computed
id will be the concatenation of the base id with the field name, along with a book keeping
index for each input in the list.

	as (:atom) - The name to be used in the form, defaults to the concatenation of the given
field to the parent form name.

	default (:any) - The value to use if none is available.

	prepend (:list) - The values to prepend when rendering. This only applies if the field value
is a list and no parameters were sent through the form.

	append (:list) - The values to append when rendering. This only applies if the field value
is a list and no parameters were sent through the form.

	skip_hidden (:boolean) - Skip the automatic rendering of hidden fields to allow for more tight control
over the generated markup.
Defaults to false.

	skip_persistent_id (:boolean) - Skip the automatic rendering of hidden _persistent_id fields used for reordering
inputs.
Defaults to false.

	options (:list) - Any additional options for the Phoenix.HTML.FormData protocol
implementation.
Defaults to [].



  
    
  
  Slots


	inner_block (required) - The content rendered for each nested form.


  
    
  
  Examples


<.form
  for={@form}
  phx-change="change_name"
>
  <.inputs_for :let={f_nested} field={@form[:nested]}>
    <.input type="text" field={f_nested[:name]} />
  </.inputs_for>
</.form>

  
    
  
  Dynamically adding and removing inputs


Dynamically adding and removing inputs is supported by rendering named buttons for
inserts and removals. Like inputs, buttons with name/value pairs are serialized with
form data on change and submit events. Libraries such as Ecto, or custom param
filtering can then inspect the parameters and handle the added or removed fields.
This can be combined with Ecto.Changeset.cast_assoc/3's :sort_param and :drop_param
options. For example, imagine a parent with an :emails has_many or embeds_many
association. To cast the user input from a nested form, one simply needs to configure
the options:
schema "mailing_lists" do
  field :title, :string

  embeds_many :emails, EmailNotification, on_replace: :delete do
    field :email, :string
    field :name, :string
  end
end

def changeset(list, attrs) do
  list
  |> cast(attrs, [:title])
  |> cast_embed(:emails,
    with: &email_changeset/2,
    sort_param: :emails_sort,
    drop_param: :emails_drop
  )
end
Here we see the :sort_param and :drop_param options in action.
Note: on_replace: :delete on the has_many and embeds_many is required
when using these options.

When Ecto sees the specified sort or drop parameter from the form, it will sort
the children based on the order they appear in the form, add new children it hasn't
seen, or drop children if the parameter instructs it to do so.
The markup for such a schema and association would look like this:
<.inputs_for :let={ef} field={@form[:emails]}>
  <input type="hidden" name="mailing_list[emails_sort][]" value={ef.index} />
  <.input type="text" field={ef[:email]} placeholder="email" />
  <.input type="text" field={ef[:name]} placeholder="name" />
  <button
    type="button"
    name="mailing_list[emails_drop][]"
    value={ef.index}
    phx-click={JS.dispatch("change")}
  >
    <.icon name="hero-x-mark" class="w-6 h-6 relative top-2" />
  </button>
</.inputs_for>

<input type="hidden" name="mailing_list[emails_drop][]" />

<button type="button" name="mailing_list[emails_sort][]" value="new" phx-click={JS.dispatch("change")}>
  add more
</button>
We used inputs_for to render inputs for the :emails association, which
contains an email address and name input for each child. Within the nested inputs,
we render a hidden mailing_list[emails_sort][] input, which is set to the index of the
given child. This tells Ecto's cast operation how to sort existing children, or
where to insert new children. Next, we render the email and name inputs as usual.
Then we render a button containing the "delete" text with the name mailing_list[emails_drop][],
containing the index of the child as its value.
Like before, this tells Ecto to delete the child at this index when the button is
clicked. We use phx-click={JS.dispatch("change")} on the button to tell LiveView
to treat this button click as a change event, rather than a submit event on the form,
which invokes our form's phx-change binding.
Outside the inputs_for, we render an empty mailing_list[emails_drop][] input,
to ensure that all children are deleted when saving a form where the user
dropped all entries. This hidden input is required whenever dropping associations.
Finally, we also render another button with the sort param name mailing_list[emails_sort][]
and value="new" name with accompanied "add more" text. Please note that this button must
have type="button" to prevent it from submitting the form.
Ecto will treat unknown sort params as new children and build a new child.
This button is optional and only necessary if you want to dynamically add entries.
You can optionally add a similar button before the <.inputs_for>, in the case you want
to prepend entries.

  
    
  
  A note on accessing a field's value


You may be tempted to access form[:field].value or attempt to manipulate
the form metadata in your templates. However, bear in mind that the form[:field]
value reflects the most recent changes. For example, an :integer field may
either contain integer values, but it may also hold a string, if the form has
been submitted.
This is particularly noticeable when using inputs_for. Accessing the .value
of a nested field may either return a struct, a changeset, or raw parameters
sent by the client (when using drop_param). This makes the form[:field].value
impractical for deriving or computing other properties.
The correct way to approach this problem is by computing any property either in
your LiveViews, by traversing the relevant changesets and data structures, or by
moving the logic to the Ecto.Changeset itself.
As an example, imagine you are building a time tracking application where:
	users enter the total work time for a day
	individual activities are tracked as embeds
	the sum of all activities should match the total time
	the form should display the remaining time

Instead of trying to calculate the remaining time in your template by
doing something like calculate_remaining(@form) and accessing
form[:activities].value, calculate the remaining time based
on the changeset in your handle_event instead:
def handle_event("validate", %{"tracked_day" => params}, socket) do
  changeset = TrackedDay.changeset(socket.assigns.tracked_day, params)
  remaining = calculate_remaining(changeset)
  {:noreply, assign(socket, form: to_form(changeset, action: :validate), remaining: remaining)}
end

# Helper function to calculate remaining time
defp calculate_remaining(changeset) do
  total = Ecto.Changeset.get_field(changeset, :total)
  activities = Ecto.Changeset.get_embed(changeset, :activities)

  Enum.reduce(activities, total, fn activity, acc ->
    duration =
      case activity do
        %{valid?: true} = changeset -> Ecto.Changeset.get_field(changeset, :duration)
        # if the activity is invalid, we don't include its duration in the calculation
        _ -> 0
      end

    acc - length
  end)
end
This logic might also be implemented directly in your schema module and, if you
often need the :remaining value, you could also add it as a :virtual field to
your schema and run the calculation when validating the changeset:
def changeset(tracked_day, attrs) do
  tracked_day
  |> cast(attrs, [:total_duration])
  |> cast_embed(:activities)
  |> validate_required([:total_duration])
  |> validate_number(:total_duration, greater_than: 0)
  |> validate_and_put_remaining_time()
end

defp validate_and_put_remaining_time(changeset) do
  remaining = calculate_remaining(changeset)
  put_change(changeset, :remaining, remaining)
end
By using this approach, you can safely render the remaining time in your template
using @form[:remaining].value, avoiding the pitfalls of directly accessing complex field values.

  



  
    
      
    
    
      intersperse(assigns)



        
          
        

    

  


  

Intersperses separator slot between an enumerable.
Useful when you need to add a separator between items such as when
rendering breadcrumbs for navigation. Provides each item to the
inner block.

  
    
  
  Examples


<.intersperse :let={item} enum={["home", "profile", "settings"]}>
  <:separator>
    <span class="sep">|</span>
  </:separator>
  {item}
</.intersperse>
Renders the following markup:
home <span class="sep">|</span> profile <span class="sep">|</span> settings

  
    
  
  Attributes


	enum (:any) (required) - the enumerable to intersperse with separators.


  
    
  
  Slots


	inner_block (required) - the inner_block to render for each item.
	separator (required) - the slot for the separator.


  



  
    
      
    
    
      link(assigns)



        
          
        

    

  


  

Generates a link to a given route.
It is typically used with one of the three attributes:
	patch - on click, it patches the current LiveView with the given path
	navigate - on click, it navigates to a new LiveView at the given path
	href - on click, it performs traditional browser navigation (as any <a> tag)


  
    
  
  Attributes


	navigate (:string) - Navigates to a LiveView.
When redirecting across LiveViews, the browser page is kept, but a new LiveView process
is mounted and its contents is loaded on the page. It is only possible to navigate
between LiveViews declared under the same router
live_session.
When used outside of a LiveView or across live sessions, it behaves like a regular
browser redirect.

	patch (:string) - Patches the current LiveView.
The handle_params callback of the current LiveView will be invoked and the minimum content
will be sent over the wire, as any other LiveView diff.

	href (:any) - Uses traditional browser navigation to the new location.
This means the whole page is reloaded on the browser.

	replace (:boolean) - When using :patch or :navigate,
should the browser's history be replaced with pushState?
Defaults to false.

	method (:string) - The HTTP method to use with the link. This is intended for usage outside of LiveView
and therefore only works with the href={...} attribute. It has no effect on patch
and navigate instructions.
In case the method is not get, the link is generated inside the form which sets the proper
information. In order to submit the form, JavaScript must be enabled in the browser.
Defaults to "get".

	csrf_token (:any) - A boolean or custom token to use for links with an HTTP method other than get. Defaults to true.

	Global attributes are accepted. Additional HTML attributes added to the a tag. Supports all globals plus: ["download", "hreflang", "referrerpolicy", "rel", "target", "type"].



  
    
  
  Slots


	inner_block (required) - The content rendered inside of the a tag.


  
    
  
  Examples


<.link href="/">Regular anchor link</.link>
<.link navigate={~p"/"} class="underline">home</.link>
<.link navigate={~p"/?sort=asc"} replace={false}>
  Sort By Price
</.link>
<.link patch={~p"/details"}>view details</.link>
<.link href={URI.parse("https://elixir-lang.org")}>hello</.link>
<.link href="/the_world" method="delete" data-confirm="Really?">delete</.link>

  
    
  
  JavaScript dependency


In order to support links where :method is not "get" or use the above data attributes,
Phoenix.HTML relies on JavaScript. You can load priv/static/phoenix_html.js into your
build tool.

  
    
  
  Data attributes


Data attributes are added as a keyword list passed to the data key. The following data
attributes are supported:
	data-confirm - shows a confirmation prompt before generating and submitting the form when
:method is not "get".


  
    
  
  Overriding the default confirm behaviour


phoenix_html.js does trigger a custom event phoenix.link.click on the clicked DOM element
when a click happened. This allows you to intercept the event on its way bubbling up
to window and do your own custom logic to enhance or replace how the data-confirm
attribute is handled. You could for example replace the browsers confirm() behavior with
a custom javascript implementation:
// Compared to a javascript window.confirm, the custom dialog does not block
// javascript execution. Therefore to make this work as expected we store
// the successful confirmation as an attribute and re-trigger the click event.
// On the second click, the `data-confirm-resolved` attribute is set and we proceed.
const RESOLVED_ATTRIBUTE = "data-confirm-resolved";
// listen on document.body, so it's executed before the default of
// phoenix_html, which is listening on the window object
document.body.addEventListener('phoenix.link.click', function (e) {
  // Prevent default implementation
  e.stopPropagation();
  // Introduce alternative implementation
  var message = e.target.getAttribute("data-confirm");
  if(!message){ return; }

  // Confirm is resolved execute the click event
  if (e.target?.hasAttribute(RESOLVED_ATTRIBUTE)) {
    e.target.removeAttribute(RESOLVED_ATTRIBUTE);
    return;
  }

  // Confirm is needed, preventDefault and show your modal
  e.preventDefault();
  e.target?.setAttribute(RESOLVED_ATTRIBUTE, "");

  vex.dialog.confirm({
    message: message,
    callback: function (value) {
      if (value == true) {
        // Customer confirmed, re-trigger the click event.
        e.target?.click();
      } else {
        // Customer canceled
        e.target?.removeAttribute(RESOLVED_ATTRIBUTE);
      }
    }
  })
}, false);
Or you could attach your own custom behavior.
window.addEventListener('phoenix.link.click', function (e) {
  // Introduce custom behaviour
  var message = e.target.getAttribute("data-prompt");
  var answer = e.target.getAttribute("data-prompt-answer");
  if(message && answer && (answer != window.prompt(message))) {
    e.preventDefault();
  }
}, false);
The latter could also be bound to any click event, but this way you can be sure your custom
code is only executed when the code of phoenix_html.js is run.

  
    
  
  CSRF Protection


By default, CSRF tokens are generated through Plug.CSRFProtection.

  



  
    
      
    
    
      live_component(assigns)



        
          
        

    

  


  

A function component for rendering Phoenix.LiveComponent within a parent LiveView.
While LiveViews can be nested, each LiveView starts its own process. A LiveComponent provides
similar functionality to LiveView, except they run in the same process as the LiveView,
with its own encapsulated state. That's why they are called stateful components.

  
    
  
  Attributes


	id (:string) (required) - A unique identifier for the LiveComponent. Note the id won't
necessarily be used as the DOM id. That is up to the component to decide.

	module (:atom) (required) - The LiveComponent module to render.


Any additional attributes provided will be passed to the LiveComponent as a map of assigns.
See Phoenix.LiveComponent for more information.

  
    
  
  Examples


<.live_component module={MyApp.WeatherComponent} id="thermostat" city="Kraków" />

  



  
    
      
    
    
      live_file_input(assigns)



        
          
        

    

  


  

Builds a file input tag for a LiveView upload.

  
    
  
  Attributes


	upload (Phoenix.LiveView.UploadConfig) (required) - The Phoenix.LiveView.UploadConfig struct.
	accept (:string) - the optional override for the accept attribute. Defaults to :accept specified by allow_upload.
	Global attributes are accepted. Supports all globals plus: ["webkitdirectory", "required", "disabled", "capture", "form"].


  
    
  
  Customizing the Label


The id attribute cannot be overwritten, but you can create a label with a for attribute
pointing to the UploadConfig ref:
<label for={@uploads.avatar.ref}>
  <.live_file_input upload={@uploads.avatar} />
</label>

  
    
  
  Drag and Drop


Drag and drop is supported by annotating the droppable container with a phx-drop-target
attribute pointing to the UploadConfig ref, so the following markup is all that is required
for drag and drop support:
<label for={@uploads.avatar.ref} phx-drop-target={@uploads.avatar.ref}>
  <.live_file_input upload={@uploads.avatar} />
</label>

  
    
  
  Examples


Rendering a file input:
<.live_file_input upload={@uploads.avatar} />
Rendering a file input with a label:
<label for={@uploads.avatar.ref}>Avatar</label>
<.live_file_input upload={@uploads.avatar} />

  



  
    
      
    
    
      live_img_preview(assigns)



        
          
        

    

  


  

Generates an image preview on the client for a selected file.

  
    
  
  Attributes


	entry (Phoenix.LiveView.UploadEntry) (required) - The Phoenix.LiveView.UploadEntry struct.
	id (:string) - the id of the img tag. Derived by default from the entry ref, but can be overridden as needed if you need to render a preview of the same entry multiple times on the same page. Defaults to nil.
	Global attributes are accepted.


  
    
  
  Examples


<.live_img_preview :for={entry <- @uploads.avatar.entries} entry={entry} width="75" />
When you need to use it multiple times, make sure that they have distinct ids
<.live_img_preview :for={entry <- @uploads.avatar.entries} entry={entry} width="75" />

<.live_img_preview :for={entry <- @uploads.avatar.entries} id={"modal-#{entry.ref}"} entry={entry} width="500" />

  



  
    
      
    
    
      live_title(assigns)



        
          
        

    

  


  

Renders a title with automatic prefix/suffix on @page_title updates.

  
    
  
  Attributes


	prefix (:string) - A prefix added before the content of inner_block. Defaults to nil.
	default (:string) - The default title to use if the inner block is empty on regular or connected mounts. Note: empty titles, such as nil or an empty string, fall back to the default value. Defaults to nil.
	suffix (:string) - A suffix added after the content of inner_block. Defaults to nil.


  
    
  
  Slots


	inner_block (required) - Content rendered inside the title tag.


  
    
  
  Examples


<.live_title default="Welcome" prefix="MyApp · ">
  {assigns[:page_title]}
</.live_title>
<.live_title default="Welcome" suffix=" · MyApp">
  {assigns[:page_title]}
</.live_title>

  


        

      

      
        Macros

        


    

  
    
      
    
    
      attr(name, type, opts \\ [])


        (macro)


        
          
        

    

  


  

Declares attributes for a HEEx function components.

  
    
  
  Arguments


	name - an atom defining the name of the attribute. Note that attributes cannot define the
same name as any other attributes or slots declared for the same component.

	type - an atom defining the type of the attribute.

	opts - a keyword list of options. Defaults to [].



  
    
  
  Types


An attribute is declared by its name, type, and options. The following types are supported:
	Name	Description
	:any	any term (including nil)
	:string	any binary string
	:atom	any atom (including true, false, and nil)
	:boolean	any boolean
	:integer	any integer
	:float	any float
	:list	any list of any arbitrary types
	:map	any map of any arbitrary types
	:fun	any function
	{:fun, arity}	any function of arity
	:global	any common HTML attributes, plus those defined by :global_prefixes
	A struct module	any module that defines a struct with defstruct/1

Note only :any and :atom expect the value to be set to nil.

  
    
  
  Options


	:required - marks an attribute as required. If a caller does not pass the given attribute,
a compile warning is issued.

	:default - the default value for the attribute if not provided. If this option is
not set and the attribute is not given, accessing the attribute will fail unless a
value is explicitly set with assign_new/3.

	:examples - a non-exhaustive list of values accepted by the attribute, used for documentation
purposes.

	:values - an exhaustive list of values accepted by the attributes. If a caller passes a literal
not contained in this list, a compile warning is issued.

	:doc - documentation for the attribute.



  
    
  
  Compile-Time Validations


LiveView performs some validation of attributes via the :phoenix_live_view compiler.
When attributes are defined, LiveView will warn at compilation time on the caller if:
	A required attribute of a component is missing.

	An unknown attribute is given.

	You specify a literal attribute (such as value="string" or value, but not value={expr})
and the type does not match. The following types currently support literal validation:
:string, :atom, :boolean, :integer, :float, :map and :list.

	You specify a literal attribute and it is not a member of the :values list.


LiveView does not perform any validation at runtime. This means the type information is mostly
used for documentation and reflection purposes.
On the side of the LiveView component itself, defining attributes provides the following quality
of life improvements:
	The default value of all attributes will be added to the assigns map upfront.

	Attribute documentation is generated for the component.

	Required struct types are annotated and emit compilation warnings. For example, if you specify
attr :user, User, required: true and then you write @user.non_valid_field in your template,
a warning will be emitted.

	Calls made to the component are tracked for reflection and validation purposes.



  
    
  
  Documentation Generation


Public function components that define attributes will have their attribute
types and docs injected into the function's documentation, depending on the
value of the @doc module attribute:
	if @doc is a string, the attribute docs are injected into that string. The optional
placeholder [INSERT LVATTRDOCS] can be used to specify where in the string the docs are
injected. Otherwise, the docs are appended to the end of the @doc string.

	if @doc is unspecified, the attribute docs are used as the default @doc string.

	if @doc is false, the attribute docs are omitted entirely.


The injected attribute docs are formatted as a markdown list:
	name (:type) (required) - attr docs. Defaults to :default.

By default, all attributes will have their types and docs injected into the function @doc
string. To hide a specific attribute, you can set the value of :doc to false.

  
    
  
  Example


attr :name, :string, required: true
attr :age, :integer, required: true

def celebrate(assigns) do
  ~H"""
  <p>
    Happy birthday {@name}!
    You are {@age} years old.
  </p>
  """
end

  



    

  
    
      
    
    
      embed_templates(pattern, opts \\ [])


        (macro)


        
          
        

    

  


  

Embeds external template files into the module as function components.

  
    
  
  Options


	:root - The root directory to embed files. Defaults to the current
module's directory (__DIR__)
	:suffix - A string value to append to embedded function names. By
default, function names will be the name of the template file excluding
the format and engine.

A wildcard pattern may be used to select all files within a directory tree.
For example, imagine a directory listing:
├── components.ex
├── pages
│   ├── about_page.html.heex
│   └── welcome_page.html.heex
Then to embed the page templates in your components.ex module:
defmodule MyAppWeb.Components do
  use Phoenix.Component

  embed_templates "pages/*"
end
Now, your module will have an about_page/1 and welcome_page/1 function
component defined. Embedded templates also support declarative assigns
via bodyless function definitions, for example:
defmodule MyAppWeb.Components do
  use Phoenix.Component

  embed_templates "pages/*"

  attr :name, :string, required: true
  def welcome_page(assigns)

  slot :header
  def about_page(assigns)
end
Multiple invocations of embed_templates is also supported, which can be
useful if you have more than one template format. For example:
defmodule MyAppWeb.Emails do
  use Phoenix.Component

  embed_templates "emails/*.html", suffix: "_html"
  embed_templates "emails/*.text", suffix: "_text"
end
Note: this function is the same as Phoenix.Template.embed_templates/2.
It is also provided here for convenience and documentation purposes.
Therefore, if you want to embed templates for other formats, which are
not related to Phoenix.Component, prefer to
import Phoenix.Template, only: [embed_templates: 1] than this module.

  



  
    
      
    
    
      sigil_H(arg, modifiers)


        (macro)


        
          
        

    

  


  

The ~H sigil for writing HEEx templates inside source files.
HEEx is a HTML-aware and component-friendly extension of Elixir Embedded
language (EEx) that provides:
	Built-in handling of HTML attributes

	An HTML-like notation for injecting function components

	Compile-time validation of the structure of the template

	The ability to minimize the amount of data sent over the wire

	Out-of-the-box code formatting via mix format



  
    
  
  Example


~H"""
<div title="My div" class={@class}>
  <p>Hello {@name}</p>
  <MyApp.Weather.city name="Kraków"/>
</div>
"""

  
    
  
  Syntax


HEEx is built on top of Embedded Elixir (EEx). In this section, we are going to
cover the basic constructs in HEEx templates as well as its syntax extensions.

  
    
  
  Interpolation


HEEx allows using {...} for HTML-aware interpolation, inside tag attributes
as well as the body:
<p>Hello, {@name}</p>
If you want to interpolate an attribute, you write:
<div class={@class}>
  ...
</div>
You can put any Elixir expression between { ... }. For example, if you want
to set classes, where some are static and others are dynamic, you can using
string interpolation:
<div class={"btn btn-#{@type}"}>
  ...
</div>
The following attribute values have special meaning on HTML tags:
	true - if a value is true, the attribute is rendered with no value at all.
For example, <input required={true}> is the same as <input required>;

	false or nil - if a value is false or nil, the attribute is omitted.
Note the class and style attributes will be rendered as empty strings,
instead of ommitted, which has the same effect as not rendering them, but
allows for rendering optimizations.

	list (only for the class attribute) - each element of the list is processed
as a different class. nil and false elements are discarded.


For multiple dynamic attributes, you can use the same notation but without
assigning the expression to any specific attribute:
<div {@dynamic_attrs}>
  ...
</div>
In this case, the expression inside {...} must be either a keyword list or
a map containing the key-value pairs representing the dynamic attributes.
If using a map, ensure your keys are atoms.

  
    
  
  Interpolating blocks


The curly braces syntax is the default mechanism for interpolating code.
However, it cannot be used in all scenarios, in particular:
	Curly braces cannot be used inside <script> and <style> tags,
as that would make writing JS and CSS quite tedious. You can also
fully disable curly braces interpolation in a given tag and
its children by adding the phx-no-curly-interpolation attribute

	it does not support multiline block constructs, such as if,
case, and similar


For example, if you need to interpolate a string inside a script tag,
you could do:
<script>
  window.URL = "<%= @my_url %>"
</script>
Similarly, for block constructs in Elixir, you can write:
<%= if @show_greeting? do %>
  <p>Hello, {@name}</p>
<% end %>
However, for conditionals and for-comprehensions, there are built-in constructs
in HEEx too, which we will explore next.
Curly braces in text within tag bodies
If you have text in your tag bodies, which includes curly braces you can use
&lbrace; or <%= "{" %> to prevent them from being considered the start of
interpolation.

  
    
  
  Special attributes


Apart from normal HTML attributes, HEEx also supports some special attributes
such as :let and :for.
:let
This is used by components and slots that want to yield a value back to the
caller. For an example, see how form/1 works:
<.form :let={f} for={@form} phx-change="validate" phx-submit="save">
  <.input field={f[:username]} type="text" />
  ...
</.form>
Notice how the variable f, defined by .form is used by your input component.
The Phoenix.Component module has detailed documentation on how to use and
implement such functionality.
:if and :for
It is a syntax sugar for <%= if .. do %> and <%= for .. do %> that can be
used in regular HTML, function components, and slots.
For example in an HTML tag:
<table id="admin-table" :if={@admin?}>
  <tr :for={user <- @users}>
    <td>{user.name}</td>
  </tr>
<table>
The snippet above will only render the table if @admin? is true,
and generate a tr per user as you would expect from the collection.
:for can be used similarly in function components:
<.error :for={msg <- @errors} message={msg}/>
Which is equivalent to writing:
<%= for msg <- @errors do %>
  <.error message={msg} />
<% end %>
And :for in slots behaves the same way:
<.table id="my-table" rows={@users}>
  <:col :for={header <- @headers} :let={user}>
    <td>{user[header]}</td>
  </:col>
<table>
You can also combine :for and :if for tags, components, and slot to act as a filter:
<.error :for={msg <- @errors} :if={msg != nil} message={msg} />
Note that unlike Elixir's regular for, HEEx' :for does not support multiple
generators in one expression. In such cases, you must use EEx's blocks.
:keyed comprehensions
When using :for, you can optionally provide a :key expression to also perform
change-tracking inside the comprehension:
<ul>
  <li :for={%{id: id, name: name} <- @items} :key={id}>
    Count: <span>{@count}</span>,
    item: {name}
  </li>
</ul>
Internally, this works by turning the tag where the comprehension is defined on
into the template of a Phoenix.LiveComponent. Because of this, the following limitations
apply to comprehensions defined with :key:
	A :key can only be defined on regular HTML tags, not on components or slots.
	The diff over the wire is optimized to only send changes for each item,
but it will always include a list of component IDs (integers) specifying
the overall order of items.
	Removing an entry involves separate round-trips with the client to confirm
the component removal.

We recommend to use :keyed comprehensions only if you already determined that you need
to optimize the diff over the write and streams
are not an option.

  
    
  
  Function components


Function components are stateless components implemented as pure functions
with the help of the Phoenix.Component module. They can be either local
(same module) or remote (external module).
HEEx allows invoking these function components directly in the template
using an HTML-like notation. For example, a remote function:
<MyApp.Weather.city name="Kraków"/>
A local function can be invoked with a leading dot:
<.city name="Kraków"/>
where the component could be defined as follows:
defmodule MyApp.Weather do
  use Phoenix.Component

  def city(assigns) do
    ~H"""
    The chosen city is: {@name}.
    """
  end

  def country(assigns) do
    ~H"""
    The chosen country is: {@name}.
    """
  end
end
It is typically best to group related functions into a single module, as
opposed to having many modules with a single render/1 function. Function
components support other important features, such as slots. You can learn
more about components in Phoenix.Component.

  
    
  
  Code formatting


You can automatically format HEEx templates (.heex) and ~H sigils
using Phoenix.LiveView.HTMLFormatter. Please check that module
for more information.

  



    

  
    
      
    
    
      slot(name, opts \\ [])


        (macro)


        
          
        

    

  


  

Declares a slot. See slot/3 for more information.

  



  
    
      
    
    
      slot(name, opts, block)


        (macro)


        
          
        

    

  


  

Declares a function component slot.

  
    
  
  Arguments


	name - an atom defining the name of the slot. Note that slots cannot define the same name
as any other slots or attributes declared for the same component.

	opts - a keyword list of options. Defaults to [].

	block - a code block containing calls to attr/3. Defaults to nil.



  
    
  
  Options


	:required - marks a slot as required. If a caller does not pass a value for a required slot,
a compilation warning is emitted. Otherwise, an omitted slot will default to [].

	:validate_attrs - when set to false, no warning is emitted when a caller passes attributes
to a slot defined without a do block. If not set, defaults to true.

	:doc - documentation for the slot. Any slot attributes declared
will have their documentation listed alongside the slot.



  
    
  
  Slot Attributes


A named slot may declare attributes by passing a block with calls to attr/3.
Unlike attributes, slot attributes cannot accept the :default option. Passing one
will result in a compile warning being issued.

  
    
  
  The Default Slot


The default slot can be declared by passing :inner_block as the name of the slot.
Note that the :inner_block slot declaration cannot accept a block. Passing one will
result in a compilation error.

  
    
  
  Compile-Time Validations


LiveView performs some validation of slots via the :phoenix_live_view compiler.
When slots are defined, LiveView will warn at compilation time on the caller if:
	A required slot of a component is missing.

	An unknown slot is given.

	An unknown slot attribute is given.


On the side of the function component itself, defining attributes provides the following
quality of life improvements:
	Slot documentation is generated for the component.

	Calls made to the component are tracked for reflection and validation purposes.



  
    
  
  Documentation Generation


Public function components that define slots will have their docs injected into the function's
documentation, depending on the value of the @doc module attribute:
	if @doc is a string, the slot docs are injected into that string. The optional placeholder
[INSERT LVATTRDOCS] can be used to specify where in the string the docs are injected.
Otherwise, the docs are appended to the end of the @doc string.

	if @doc is unspecified, the slot docs are used as the default @doc string.

	if @doc is false, the slot docs are omitted entirely.


The injected slot docs are formatted as a markdown list:
	name (required) - slot docs. Accepts attributes:	name (:type) (required) - attr docs. Defaults to :default.



By default, all slots will have their docs injected into the function @doc string.
To hide a specific slot, you can set the value of :doc to false.

  
    
  
  Example


slot :header
slot :inner_block, required: true
slot :footer

def modal(assigns) do
  ~H"""
  <div class="modal">
    <div class="modal-header">
      {render_slot(@header) || "Modal"}
    </div>
    <div class="modal-body">
      {render_slot(@inner_block)}
    </div>
    <div class="modal-footer">
      {render_slot(@footer) || submit_button()}
    </div>
  </div>
  """
end
As shown in the example above, render_slot/1 returns nil when an optional slot is declared
and none is given. This can be used to attach default behaviour.

      assign(socket_or_assigns, keyword_or_map)



        
          
        

    

  


  

Adds key-value pairs to assigns.
The first argument is either a LiveView socket or an assigns map from function components.
A keyword list or a map of assigns must be given as argument to be merged into existing assigns.

  
    
  
  Examples


iex> assign(socket, name: "Elixir", logo: "💧")
iex> assign(socket, %{name: "Elixir"})

  



  
    
      
    
    
      assign(socket_or_assigns, key, value)



        
          
        

    

  


  

Adds a key-value pair to socket_or_assigns.
The first argument is either a LiveView socket or an assigns map from function components.

  
    
  
  Examples


iex> assign(socket, :name, "Elixir")

  



  
    
      
    
    
      assign_new(socket_or_assigns, key, fun)



        
          
        

    

  


  

Assigns the given key with value from fun into socket_or_assigns if one does not yet exist.
The first argument is either a LiveView socket or an assigns map from function components.
This function is useful for lazily assigning values and sharing assigns.
We will cover both use cases next.

  
    
  
  Lazy assigns


Imagine you have a function component that accepts a color:
<.my_component bg_color="red" />
The color is also optional, so you can skip it:
<.my_component />
In such cases, the implementation can use assign_new to lazily
assign a color if none is given. Let's make it so it picks a random one
when none is given:
def my_component(assigns) do
  assigns = assign_new(assigns, :bg_color, fn -> Enum.random(~w(bg-red-200 bg-green-200 bg-blue-200)) end)

  ~H"""
  <div class={@bg_color}>
    Example
  </div>
  """
end

  
    
  
  Sharing assigns


It is possible to share assigns between the Plug pipeline and LiveView on disconnected render
and between parent-child LiveViews when connected.

  
    
  
  When disconnected


When a user first accesses an application using LiveView, the LiveView is first rendered in its
disconnected state, as part of a regular HTML response. By using assign_new in the mount
callback of your LiveView, you can instruct LiveView to re-use any assigns already set in conn
during disconnected state.
Imagine you have a Plug that does:
# A plug
def authenticate(conn, _opts) do
  if user_id = get_session(conn, :user_id) do
    assign(conn, :current_user, Accounts.get_user!(user_id))
  else
    send_resp(conn, :forbidden)
  end
end
You can re-use the :current_user assign in your LiveView during the initial render:
def mount(_params, %{"user_id" => user_id}, socket) do
  {:ok, assign_new(socket, :current_user, fn -> Accounts.get_user!(user_id) end)}
end
In such case conn.assigns.current_user will be used if present. If there is no such
:current_user assign or the LiveView was mounted as part of the live navigation, where no Plug
pipelines are invoked, then the anonymous function is invoked to execute the query instead.

  
    
  
  When connected


LiveView is also able to share assigns via assign_new with children LiveViews,
as long as the child LiveView is also mounted when the parent LiveView is mounted
and the child LiveView is not rendered with sticky: true. Let's see an example.
If the parent LiveView defines a :current_user assign and the child LiveView also
uses assign_new/3 to fetch the :current_user in its mount/3 callback, as in
the previous subsection, the assign will be fetched from the parent LiveView, once
again avoiding additional database queries.
Note that fun also provides access to the previously assigned values:
assigns =
  assigns
  |> assign_new(:foo, fn -> "foo" end)
  |> assign_new(:bar, fn %{foo: foo} -> foo <> "bar" end)
Assigns sharing is performed when possible but not guaranteed. Therefore, you must
ensure the result of the function given to assign_new/3 is the same as if the value
was fetched from the parent. Otherwise consider passing values to the child LiveView
as part of its session.

  



    

  
    
      
    
    
      assigns_to_attributes(assigns, exclude \\ [])



        
          
        

    

  


  

Filters the assigns as a list of keywords for use in dynamic tag attributes.
One should prefer to use declarative assigns and :global attributes
over this function.

  
    
  
  Examples


Imagine the following my_link component which allows a caller
to pass a new_window assign, along with any other attributes they
would like to add to the element, such as class, data attributes, etc:
<.my_link to="/" id={@id} new_window={true} class="my-class">Home</.my_link>
We could support the dynamic attributes with the following component:
def my_link(assigns) do
  target = if assigns[:new_window], do: "_blank", else: false
  extra = assigns_to_attributes(assigns, [:new_window, :to])

  assigns =
    assigns
    |> assign(:target, target)
    |> assign(:extra, extra)

  ~H"""
  <a href={@to} target={@target} {@extra}>
    {render_slot(@inner_block)}
  </a>
  """
end
The above would result in the following rendered HTML:
<a href="/" target="_blank" id="1" class="my-class">Home</a>
The second argument (optional) to assigns_to_attributes is a list of keys to
exclude. It typically includes reserved keys by the component itself, which either
do not belong in the markup, or are already handled explicitly by the component.

  



  
    
      
    
    
      changed?(socket_or_assigns, key)



        
          
        

    

  


  

Checks if the given key changed in socket_or_assigns.
The first argument is either a LiveView socket or an assigns map from function components.

  
    
  
  Examples


iex> changed?(socket, :count)

  



  
    
      
    
    
      live_flash(other, key)



        
          
        

    

  


    
      This function is deprecated. Use Phoenix.Flash.get/2 in Phoenix v1.7+.
    


  

Returns the flash message from the LiveView flash assign.

  
    
  
  Examples


<p class="alert alert-info">{live_flash(@flash, :info)}</p>
<p class="alert alert-danger">{live_flash(@flash, :error)}</p>

  



    

  
    
      
    
    
      live_render(conn_or_socket, view, opts \\ [])



        
          
        

    

  


  

Renders a LiveView within a template.
This is useful in two situations:
	When rendering a child LiveView inside a LiveView.

	When rendering a LiveView inside a regular (non-live) controller/view.


Most other cases for shared functionality, including state management and user interactions, can be
achieved with function components or LiveComponents

  
    
  
  Options


	:session - a map of binary keys with extra session data to be serialized and sent
to the client. All session data currently in the connection is automatically available
in LiveViews. You can use this option to provide extra data. Remember all session data is
serialized and sent to the client, so you should always keep the data in the session
to a minimum. For example, instead of storing a User struct, you should store the "user_id"
and load the User when the LiveView mounts.

	:container - an optional tuple for the HTML tag and DOM attributes to be used for the
LiveView container. For example: {:li, style: "color: blue;"}. By default it uses the module
definition container. See the "Containers" section below for more information.

	:id - both the DOM ID and the ID to uniquely identify a LiveView. An :id is
automatically generated when rendering root LiveViews but it is a required option when
rendering a child LiveView.

	:sticky - an optional flag to maintain the LiveView across live redirects, even if it is
nested within another LiveView. Note that this only works for LiveViews that are in the same
live_session.
If you are rendering the sticky view within another LiveView, make sure that the sticky view
itself does not use the same layout. You can do so by returning {:ok, socket, layout: false}
from mount.



  
    
  
  Examples


When rendering from a controller/view, you can call:
{live_render(@conn, MyApp.ThermostatLive)}
Or:
{live_render(@conn, MyApp.ThermostatLive, session: %{"home_id" => @home.id})}
Within another LiveView, you must pass the :id option:
{live_render(@socket, MyApp.ThermostatLive, id: "thermostat")}

  
    
  
  Containers


When a LiveView is rendered, its contents are wrapped in a container. By default,
the container is a div tag with a handful of LiveView-specific attributes.
The container can be customized in different ways:
	You can change the default container on use Phoenix.LiveView:
use Phoenix.LiveView, container: {:tr, id: "foo-bar"}

	You can override the container tag and pass extra attributes when calling live_render
(as well as on your live call in your router):
live_render socket, MyLiveView, container: {:tr, class: "highlight"}


If you don't want the container to affect layout, you can use the CSS property
display: contents or a class that applies it, like Tailwind's .contents.
Beware if you set this to :body, as any content injected inside the body
(such as Phoenix.LiveReload features) will be discarded once the LiveView
connects

  
    
  
  Testing


Note that render_click/1 and other testing functions will send events to the root LiveView, and you will want to
find_live_child/2 to interact with nested LiveViews in your live tests.

  



  
    
      
    
    
      portal(assigns)



        
          
        

    

  


  

Renders a portal.
A portal is a component that teleports its content to another place in the DOM.
It is useful in cases where you need to render some content in another place, for
example due to overflow or stacking context.
A portal consists of two parts:
	The portal source: the component that should be teleported.
	The portal target: the DOM element that will render the content of the portal source.

Any element can be a portal target. In most cases, the target would be rendered inside
the layout of your application. Portal sources must be defined with the .portal component.

  
    
  
  Examples


<.portal id="modal" target="body">
  ...
</.portal>

  
    
  
  Attributes


	id (:string) (required)
	target (:string) (required) - A CSS selector that identifies the target. The target must be unique.
	class (:string) - The class to apply to the portal wrapper. Defaults to nil.
	container (:string) - The HTML tag to use as the portal wrapper. Defaults to "div".


  
    
  
  Slots


	inner_block (required)


  



    

  
    
      
    
    
      render_slot(slot, argument \\ nil)


        (macro)


        
          
        

    

  


  

Renders a slot entry with the given optional argument.
{render_slot(@inner_block, @form)}
If the slot has no entries, nil is returned.
If multiple slot entries are defined for the same slot,render_slot/2 will automatically render
all entries, merging their contents. In case you want to use the entries' attributes, you need
to iterate over the list to access each slot individually.
For example, imagine a table component:
<.table rows={@users}>
  <:col :let={user} label="Name">
    {user.name}
  </:col>

  <:col :let={user} label="Address">
    {user.address}
  </:col>
</.table>
At the top level, we pass the rows as an assign and we define a :col slot for each column we
want in the table. Each column also has a label, which we are going to use in the table header.
Inside the component, you can render the table with headers, rows, and columns:
def table(assigns) do
  ~H"""
  <table>
    <tr>
      <th :for={col <- @col}>{col.label}</th>
    </tr>
    <tr :for={row <- @rows}>
      <td :for={col <- @col}>{render_slot(col, row)}</td>
    </tr>
  </table>
  """
end

  



    

  
    
      
    
    
      to_form(data_or_params, options \\ [])



        
          
        

    

  


  

Converts a given data structure to a Phoenix.HTML.Form.
This is commonly used to convert a map or an Ecto changeset
into a form to be given to the form/1 component.

  
    
  
  Creating a form from params


If you want to create a form based on handle_event parameters,
you could do:
def handle_event("submitted", params, socket) do
  {:noreply, assign(socket, form: to_form(params))}
end
When you pass a map to to_form/1, it assumes said map contains
the form parameters, which are expected to have string keys.
You can also specify a name to nest the parameters:
def handle_event("submitted", %{"user" => user_params}, socket) do
  {:noreply, assign(socket, form: to_form(user_params, as: :user))}
end

  
    
  
  Creating a form from changesets


When using changesets, the underlying data, form parameters, and
errors are retrieved from it. The :as option is automatically
computed too. For example, if you have a user schema:
defmodule MyApp.Users.User do
  use Ecto.Schema

  schema "..." do
    ...
  end
end
And then you create a changeset that you pass to to_form:
%MyApp.Users.User{}
|> Ecto.Changeset.change()
|> to_form()
In this case, once the form is submitted, the parameters will
be available under %{"user" => user_params}.

  
    
  
  Options


	:as - the name prefix to be used in form inputs
	:id - the id prefix to be used in form inputs
	:errors - keyword list of errors (used by maps exclusively)
	:action - The action that was taken against the form. This value can be
used to distinguish between different operations such as the user typing
into a form for validation, or submitting a form for a database insert.
For example: to_form(changeset, action: :validate),
or to_form(changeset, action: :save). The provided action is passed
to the underlying Phoenix.HTML.FormData implementation options.

The underlying data may accept additional options when
converted to forms. For example, a map accepts :errors
to list errors, but such option is not accepted by
changesets. :errors is a keyword of tuples in the shape
of {error_message, options_list}. Here is an example:
to_form(%{"search" => nil}, errors: [search: {"Can't be blank", []}])
If an existing Phoenix.HTML.Form struct is given, the
options above will override its existing values if given.
Then the remaining options are merged with the existing
form options.
Errors in a form are only displayed if the changeset's action
field is set (and it is not set to :ignore) and can be filtered
by whether the fields have been used on the client or not. Refer to
a note on :errors for more information.

  



  
    
      
    
    
      update(socket_or_assigns, key, fun)



        
          
        

    

  


  

Updates an existing key with fun in the given socket_or_assigns.
The first argument is either a LiveView socket or an assigns map from function components.
The update function receives the current key's value and returns the updated value.
Raises if the key does not exist.
The update function may also be of arity 2, in which case it receives the current key's value
as the first argument and the current assigns as the second argument.
Raises if the key does not exist.

  
    
  
  Examples


iex> update(socket, :count, fn count -> count + 1 end)
iex> update(socket, :count, &(&1 + 1))
iex> update(socket, :max_users_this_session, fn current_max, %{users: users} ->
...>   max(current_max, length(users))
...> end)

  



  
    
      
    
    
      upload_errors(conf)



        
          
        

    

  


  

Returns errors for the upload as a whole.
For errors that apply to a specific upload entry, use upload_errors/2.
The output is a list. The following error may be returned:
	:too_many_files - The number of selected files exceeds the :max_entries constraint


  
    
  
  Examples


def upload_error_to_string(:too_many_files), do: "You have selected too many files"
<div :for={err <- upload_errors(@uploads.avatar)} class="alert alert-danger">
  {upload_error_to_string(err)}
</div>

  



  
    
      
    
    
      upload_errors(conf, entry)



        
          
        

    

  


  

Returns errors for the upload entry.
For errors that apply to the upload as a whole, use upload_errors/1.
The output is a list. The following errors may be returned:
	:too_large - The entry exceeds the :max_file_size constraint
	:not_accepted - The entry does not match the :accept MIME types
	:external_client_failure - When external upload fails
	{:writer_failure, reason} - When the custom writer fails with reason


  
    
  
  Examples


defp upload_error_to_string(:too_large), do: "The file is too large"
defp upload_error_to_string(:not_accepted), do: "You have selected an unacceptable file type"
defp upload_error_to_string(:external_client_failure), do: "Something went terribly wrong"
<%= for entry <- @uploads.avatar.entries do %>
  <div :for={err <- upload_errors(@uploads.avatar, entry)} class="alert alert-danger">
    {upload_error_to_string(err)}
  </div>
<% end %>

  



  
    
      
    
    
      used_input?(form_field)



        
          
        

    

  


  

Checks if the input field was used by the client.
Used inputs are only those inputs that have been focused, interacted with, or
submitted by the client. For LiveView, this is used to filter errors from the
Phoenix.HTML.FormData implementation to avoid showing "field can't be blank"
in scenarios where the client hasn't yet interacted with specific fields.
Used inputs are tracked internally by the client sending a sibling key
derived from each input name, which indicates the inputs that remain  unused
on the client. For example, a form with email and title fields where only the
title has been modified so far on the client, would send the following payload:
%{
  "title" => "new title",
  "email" => "",
  "_unused_email" => ""
}
The _unused_email key indicates that the email field has not been used by the
client, which is used to filter errors from the UI.
Nested fields are also supported. For example, a form with a nested datetime field
is considered used if any of the nested parameters are used.
%{
  "bday" => %{
    "year" => "",
    "month" => "",
    "day" => "",
    "_unused_day" => ""
  }
}
The _unused_day key indicates that the day field has not been used by the client,
but the year and month fields have been used, meaning the birthday field as a whole
was used.

  
    
  
  Examples


For example, imagine in your template you render a title and email input.
On initial load the end-user begins typing the title field. The client will send
the entire form payload to the server with the typed title and an empty email.
The Phoenix.HTML.FormData implementation will consider an empty email in
this scenario as invalid, but the user shouldn't see the error because they
haven't yet used the email input. To handle this, used_input?/1 can be used to
filter errors from the client by referencing param metadata to distinguish between
used and unused input fields. For non-LiveViews, all inputs are considered used.
<input type="text" name={@form[:title].name} value={@form[:title].value} />

<div :if={used_input?(@form[:title])}>
  <p :for={error <- @form[:title].errors}>{error}</p>
</div>

<input type="text" name={@form[:email].name} value={@form[:email].value} />

<div :if={used_input?(@form[:email])}>
  <p :for={error <- @form[:email].errors}>{error}</p>
</div>

  


        

      


  

    
Phoenix.LiveComponent behaviour
    



      
LiveComponents are a mechanism to compartmentalize state, markup, and
events for sharing across LiveViews.
LiveComponents are defined by using Phoenix.LiveComponent and are used
by calling Phoenix.Component.live_component/1 in a parent LiveView.
They run inside the LiveView process but have their own state and
life-cycle. For this reason, they are also often called "stateful components".
This is a contrast to Phoenix.Component, also known as "function components",
which are stateless and do not have a life-cycle.
The smallest LiveComponent only needs to define a render/1 function:
defmodule HeroComponent do
  # In Phoenix apps, the line is typically: use MyAppWeb, :live_component
  use Phoenix.LiveComponent

  def render(assigns) do
    ~H"""
    <div class="hero">{@content}</div>
    """
  end
end
A LiveComponent is rendered as:
<.live_component module={HeroComponent} id="hero" content={@content} />
You must always pass the module and id attributes. The id will be
available as an assign and it must be used to uniquely identify the
component. All other attributes will be available as assigns inside the
LiveComponent.
Functional components or live components?
Generally speaking, you should prefer function components over live
components, as they are a simpler abstraction, with a smaller surface
area. The use case for live components only arises when there is a need
for encapsulating both event handling and additional state.
Similarly, avoid using LiveComponents for code design purposes, where
their main goal is to organize code, rather than sharing it across
LiveViews. When it comes to code organization and design, Elixir developers
prefer to use functions and data structures.

  
    
  
  Life-cycle



  
    
  
  Mount and update


Live components are identified by the component module and their ID.
We often tie the component ID to some application based ID:
<.live_component module={UserComponent} id={@user.id} user={@user} />
When live_component/1 is called,
mount/1 is called once, when the component is first added to the page.
mount/1 receives a socket as its argument. Note that this is not the
same socket struct from the parent LiveView. It doesn't contain the parent
LiveView's assigns, and updating it won't affect the parent LiveView's
socket.
Then update/2 is invoked with all of the assigns passed to
live_component/1. The assigns
received as the first argument to update/2 will only include those
assigns given to live_component/1,
and not any pre-existing assigns in socket.assigns such as those assigned
by mount/1.
If update/2 is not defined then all assigns given to
live_component/1 will simply be
merged into socket.assigns.
Both mount/1 and update/2 must return a tuple whose first element is
:ok and whose second element is the updated socket.
After the component is updated, render/1 is called with all assigns.
On first render, we get:
mount(socket) -> update(assigns, socket) -> render(assigns)
On further rendering:
update(assigns, socket) -> render(assigns)
Two live components with the same module and ID are treated as the same component,
regardless of where they are in the page. Therefore, if you change the location
of where a component is rendered within its parent LiveView, it won't be remounted.
This means you can use live components to implement cards and other elements that
can be moved around without losing state. A component is only discarded when the
client observes it is removed from the page.
Finally, the given id is not automatically used as the DOM ID. If you want to set
a DOM ID, it is your responsibility to do so when rendering:
defmodule UserComponent do
  # In Phoenix apps, the line is typically: use MyAppWeb, :live_component
  use Phoenix.LiveComponent

  def render(assigns) do
    ~H"""
    <div id={"user-#{@id}"} class="user">
      {@user.name}
    </div>
    """
  end
end

  
    
  
  Events


LiveComponents can also implement the handle_event/3 callback
that works exactly the same as in LiveView. For a client event to
reach a component, the tag must be annotated with a phx-target.
If you want to send the event to yourself, you can simply use the
@myself assign, which is an internal unique reference to the
component instance:
<a href="#" phx-click="say_hello" phx-target={@myself}>
  Say hello!
</a>
Note that @myself is not set for stateless components, as they cannot
receive events.
If you want to target another component, you can also pass an ID
or a class selector to any element inside the targeted component.
For example, if there is a UserComponent with the DOM ID of "user-13",
using a query selector, we can send an event to it with:
<a href="#" phx-click="say_hello" phx-target="#user-13">
  Say hello!
</a>
In both cases, handle_event/3 will be called with the
"say_hello" event. When handle_event/3 is called for a component,
only the diff of the component is sent to the client, making them
extremely efficient.
Any valid query selector for phx-target is supported, provided that the
matched nodes are children of a LiveView or LiveComponent, for example
to send the close event to multiple components:
<a href="#" phx-click="close" phx-target="#modal, #sidebar">
  Dismiss
</a>

  
    
  
  Update many


Live components also support an optional update_many/1 callback
as an alternative to update/2. While update/2 is called for
each component individually, update_many/1 is called with all
LiveComponents of the same module being currently rendered/updated.
The advantage is that you can preload data from the database using
a single query for all components, instead of running one query per
component.
To provide a more complete understanding of why both callbacks are necessary,
let's see an example. Imagine you are implementing a component and the component
needs to load some state from the database. For example:
<.live_component module={UserComponent} id={user_id} />
A possible implementation would be to load the user on the update/2
callback:
def update(assigns, socket) do
  user = Repo.get!(User, assigns.id)
  {:ok, assign(socket, :user, user)}
end
However, the issue with said approach is that, if you are rendering
multiple user components in the same page, you have a N+1 query problem.
By using update_many/1 instead of update/2 , we receive a list
of all assigns and sockets, allowing us to update many at once:
def update_many(assigns_sockets) do
  list_of_ids = Enum.map(assigns_sockets, fn {assigns, _socket} -> assigns.id end)

  users =
    from(u in User, where: u.id in ^list_of_ids, select: {u.id, u})
    |> Repo.all()
    |> Map.new()

  Enum.map(assigns_sockets, fn {assigns, socket} ->
    assign(socket, :user, users[assigns.id])
  end)
end
Now only a single query to the database will be made. In fact, the
update_many/1 algorithm is a breadth-first tree traversal, which means
that even for nested components, the amount of queries are kept to
a minimum.
Finally, note that update_many/1 must return an updated list of
sockets in the same order as they are given. If update_many/1 is
defined, update/2 is not invoked.

  
    
  
  Summary


All of the life-cycle events are summarized in the diagram below.
The bubble events in white are triggers that invoke the component.
In blue you have component callbacks, where the underlined names
represent required callbacks:
flowchart LR

  Managing state


Now that we have learned how to define and use components, as well as
how to use update_many/1 as a data loading optimization, it is important
to talk about how to manage state in components.
Generally speaking, you want to avoid both the parent LiveView and the
LiveComponent working on two different copies of the state. Instead, you
should assume only one of them to be the source of truth. Let's discuss
the two different approaches in detail.
Imagine a scenario where a LiveView represents a board with each card
in it as a separate LiveComponent. Each card has a form to
allow update of the card title directly in the component, as follows:
defmodule CardComponent do
  use Phoenix.LiveComponent

  def render(assigns) do
    ~H"""
    <form phx-submit="..." phx-target={@myself}>
      <input name="title">{@card.title}</input>
      ...
    </form>
    """
  end

  ...
end
We will see how to organize the data flow to keep either the board LiveView or
the card LiveComponents as the source of truth.

  
    
  
  LiveView as the source of truth


If the board LiveView is the source of truth, it will be responsible
for fetching all of the cards in a board. Then it will call
live_component/1
for each card, passing the card struct as argument to CardComponent:
<.live_component
  :for={card <- @cards}
  module={CardComponent}
  card={card}
  id={card.id}
  board_id={@id}
/>
Now, when the user submits the form, CardComponent.handle_event/3
will be triggered. However, if the update succeeds, you must not
change the card struct inside the component. If you do so, the card
struct in the component will get out of sync with the LiveView.  Since
the LiveView is the source of truth, you should instead tell the
LiveView that the card was updated.
Luckily, because the component and the view run in the same process,
sending a message from the LiveComponent to the parent LiveView is as
simple as sending a message to self():
defmodule CardComponent do
  ...
  def handle_event("update_title", %{"title" => title}, socket) do
    send self(), {:updated_card, %{socket.assigns.card | title: title}}
    {:noreply, socket}
  end
end
The LiveView then receives this event using Phoenix.LiveView.handle_info/2:
defmodule BoardView do
  ...
  def handle_info({:updated_card, card}, socket) do
    # update the list of cards in the socket
    {:noreply, updated_socket}
  end
end
Because the list of cards in the parent socket was updated, the parent
LiveView will be re-rendered, sending the updated card to the component.
So in the end, the component does get the updated card, but always
driven from the parent.
Alternatively, instead of having the component send a message directly to the
parent view, the component could broadcast the update using Phoenix.PubSub.
Such as:
defmodule CardComponent do
  ...
  def handle_event("update_title", %{"title" => title}, socket) do
    message = {:updated_card, %{socket.assigns.card | title: title}}
    Phoenix.PubSub.broadcast(MyApp.PubSub, board_topic(socket), message)
    {:noreply, socket}
  end

  defp board_topic(socket) do
    "board:" <> socket.assigns.board_id
  end
end
As long as the parent LiveView subscribes to the board:<ID> topic,
it will receive updates. The advantage of using PubSub is that we get
distributed updates out of the box. Now, if any user connected to the
board changes a card, all other users will see the change.

  
    
  
  LiveComponent as the source of truth


If each card LiveComponent is the source of truth, then the board LiveView
must no longer fetch the card structs from the database. Instead, the board
LiveView must only fetch the card ids, then render each component only by
passing an ID:
<.live_component
  :for={card_id <- @card_ids}
  module={CardComponent}
  id={card_id}
  board_id={@id}
/>
Now, each CardComponent will load its own card. Of course, doing so
per card could be expensive and lead to N queries, where N is the
number of cards, so we can use the update_many/1 callback to make it
efficient.
Once the card components are started, they can each manage their own
card, without concerning themselves with the parent LiveView.
However, note that components do not have a Phoenix.LiveView.handle_info/2
callback. Therefore, if you want to track distributed changes on a card,
you must have the parent LiveView receive those events and redirect them
to the appropriate card. For example, assuming card updates are sent
to the "board:ID" topic, and that the board LiveView is subscribed to
said topic, one could do:
def handle_info({:updated_card, card}, socket) do
  send_update CardComponent, id: card.id, board_id: socket.assigns.id
  {:noreply, socket}
end
With Phoenix.LiveView.send_update/3, the CardComponent given by id
will be invoked, triggering the update or update_many callback, which will
load the most up to date data from the database.

  
    
  
  Unifying LiveView and LiveComponent communication


In the examples above, we have used send/2 to communicate with LiveView
and send_update/2 to communicate with components. This introduces a problem:
what if you have a component that may be mounted both inside a LiveView
or another component? Given each uses a different API for exchanging data,
this may seem tricky at first, but an elegant solution is to use anonymous
functions as callbacks. Let's see an example.
In the sections above, we wrote the following code in our CardComponent:
def handle_event("update_title", %{"title" => title}, socket) do
  send self(), {:updated_card, %{socket.assigns.card | title: title}}
  {:noreply, socket}
end
The issue with this code is that, if CardComponent is mounted inside another
component, it will still message the LiveView. Not only that, this code may
be hard to maintain because the message sent by the component is defined far
away from the LiveView that will receive it.
Instead let's define a callback that will be invoked by CardComponent:
def handle_event("update_title", %{"title" => title}, socket) do
  socket.assigns.on_card_update.(%{socket.assigns.card | title: title})
  {:noreply, socket}
end
And now when initializing the CardComponent from a LiveView, we may write:
<.live_component
  module={CardComponent}
  card={card}
  id={card.id}
  board_id={@id}
  on_card_update={fn card -> send(self(), {:updated_card, card}) end} />
If initializing it inside another component, one may write:
<.live_component
  module={CardComponent}
  card={card}
  id={card.id}
  board_id={@id}
  on_card_update={fn card -> send_update(@myself, card: card) end} />
The major benefit in both cases is that the parent has explicit control
over the messages it will receive.

  
    
  
  Slots


LiveComponent can also receive slots, in the same way as a Phoenix.Component:
<.live_component module={MyComponent} id={@data.id} >
  <div>Inner content here</div>
</.live_component>
If the LiveComponent defines an update/2, be sure that the socket it returns
includes the :inner_block assign it received.
See the docs for Phoenix.Component for more information.

  
    
  
  Live patches and live redirects


A template rendered inside a component can use <.link patch={...}> and
<.link navigate={...}>. Patches are always handled by the parent LiveView,
as components do not provide handle_params.

  
    
  
  Cost of live components


The internal infrastructure LiveView uses to keep track of live
components is very lightweight. However, be aware that in order to
provide change tracking and to send diffs over the wire, all of the
components' assigns are kept in memory - exactly as it is done in
LiveViews themselves.
Therefore it is your responsibility to keep only the assigns necessary
in each component. For example, avoid passing all of LiveView's assigns
when rendering a component:
<.live_component module={MyComponent} {assigns} />
Instead pass only the keys that you need:
<.live_component module={MyComponent} user={@user} org={@org} />
Luckily, because LiveViews and LiveComponents are in the same process,
they share the data structure representations in memory. For example,
in the code above, the view and the component will share the same copies
of the @user and @org assigns.
You should also avoid using live components to provide abstract DOM
components. As a guideline, a good LiveComponent encapsulates
application concerns and not DOM functionality. For example, if you
have a page that shows products for sale, you can encapsulate the
rendering of each of those products in a component. This component
may have many buttons and events within it. On the opposite side,
do not write a component that is simply encapsulating generic DOM
components. For instance, do not do this:
defmodule MyButton do
  use Phoenix.LiveComponent

  def render(assigns) do
    ~H"""
    <button class="css-framework-class" phx-click="click">
      {@text}
    </button>
    """
  end

  def handle_event("click", _, socket) do
    _ = socket.assigns.on_click.()
    {:noreply, socket}
  end
end
Instead, it is much simpler to create a function component:
def my_button(%{text: _, click: _} = assigns) do
  ~H"""
  <button class="css-framework-class" phx-click={@click}>
    {@text}
  </button>
  """
end
If you keep components mostly as an application concern with
only the necessary assigns, it is unlikely you will run into
issues related to live components.

  
    
  
  Limitations


Live Components require a single HTML tag at the root. It is not possible
to have components that render only text or multiple tags.

        mount(socket)

      


    


    
      
        render(assigns)

      


    


    
      
        update(assigns, socket)

      


    


    
      
        update_many(list)

      


    





  
    Functions
  


    
      
        __using__(opts \\ [])

      


        Uses LiveComponent in the current module.



    





      


      
        Callbacks

        


  
    
      
    
    
      handle_async(name, async_fun_result, socket)


        (optional)


        
          
        

    

  


  

      

          @callback handle_async(
  name :: term(),
  async_fun_result :: {:ok, term()} | {:exit, term()},
  socket :: Phoenix.LiveView.Socket.t()
) :: {:noreply, Phoenix.LiveView.Socket.t()}


      



  



  
    
      
    
    
      handle_event(event, unsigned_params, socket)


        (optional)


        
          
        

    

  


  

      

          @callback handle_event(
  event :: binary(),
  unsigned_params :: Phoenix.LiveView.unsigned_params(),
  socket :: Phoenix.LiveView.Socket.t()
) ::
  {:noreply, Phoenix.LiveView.Socket.t()}
  | {:reply, map(), Phoenix.LiveView.Socket.t()}


      



  



  
    
      
    
    
      mount(socket)


        (optional)


        
          
        

    

  


  

      

          @callback mount(socket :: Phoenix.LiveView.Socket.t()) ::
  {:ok, Phoenix.LiveView.Socket.t()}
  | {:ok, Phoenix.LiveView.Socket.t(), keyword()}


      



  



  
    
      
    
    
      render(assigns)


        (optional)


        
          
        

    

  


  

      

          @callback render(assigns :: Phoenix.LiveView.Socket.assigns()) ::
  Phoenix.LiveView.Rendered.t()


      



  



  
    
      
    
    
      update(assigns, socket)


        (optional)


        
          
        

    

  


  

      

          @callback update(
  assigns :: Phoenix.LiveView.Socket.assigns(),
  socket :: Phoenix.LiveView.Socket.t()
) ::
  {:ok, Phoenix.LiveView.Socket.t()}


      



  



  
    
      
    
    
      update_many(list)


        (optional)


        
          
        

    

  


  

      

          @callback update_many([{Phoenix.LiveView.Socket.assigns(), Phoenix.LiveView.Socket.t()}]) ::
  [
    Phoenix.LiveView.Socket.t()
  ]


      



  


        

      

      
        Functions

        


    

  
    
      
    
    
      __using__(opts \\ [])


        (macro)


        
          
        

    

  


  

Uses LiveComponent in the current module.
use Phoenix.LiveComponent

  
    
  
  Options


	:global_prefixes - the global prefixes to use for components. See
Global Attributes in Phoenix.Component for more information.


  


        

      


  

    
Phoenix.LiveView behaviour
    



      
A LiveView is a process that receives events, updates
its state, and renders updates to a page as diffs.
To get started, see the Welcome guide.
This module provides advanced documentation and features
about using LiveView.

  
    
  
  Life-cycle


A LiveView begins as a regular HTTP request and HTML response,
and then upgrades to a stateful view on client connect,
guaranteeing a regular HTML page even if JavaScript is disabled.
Any time a stateful view changes or updates its socket assigns, it is
automatically re-rendered and the updates are pushed to the client.
Socket assigns are stateful values kept on the server side in
Phoenix.LiveView.Socket. This is different from the common stateless
HTTP pattern of sending the connection state to the client in the form
of a token or cookie and rebuilding the state on the server to service
every request.
You begin by rendering a LiveView typically from your router.
When LiveView is first rendered, the mount/3 callback is invoked
with the current params, the current session and the LiveView socket.
As in a regular request, params contains public data that can be
modified by the user. The session always contains private data set
by the application itself. The mount/3 callback wires up socket
assigns necessary for rendering the view. After mounting, handle_params/3
is invoked so uri and query params are handled. Finally, render/1
is invoked and the HTML is sent as a regular HTML response to the
client.
After rendering the static page, LiveView connects from the client
to the server where stateful views are spawned to push rendered updates
to the browser, and receive client events via phx- bindings. Just like
the first rendering, mount/3, is invoked  with params, session,
and socket state. However in the connected client case, a LiveView process
is spawned on the server, runs handle_params/3 again and then pushes
the result of render/1 to the client and continues on for the duration
of the connection. If at any point during the stateful life-cycle a crash
is encountered, or the client connection drops, the client gracefully
reconnects to the server, calling mount/3 and handle_params/3 again.
LiveView also allows attaching hooks to specific life-cycle stages with
attach_hook/4.

  
    
  
  Template collocation


There are two possible ways of rendering content in a LiveView. The first
one is by explicitly defining a render function, which receives assigns
and returns a HEEx template defined with the ~H sigil.
defmodule MyAppWeb.DemoLive do
  # In a typical Phoenix app, the following line would usually be `use MyAppWeb, :live_view`
  use Phoenix.LiveView

  def render(assigns) do
    ~H"""
    Hello world!
    """
  end
end
For larger templates, you can place them in a file in the same directory
and same name as the LiveView. For example, if the file above is placed
at lib/my_app_web/live/demo_live.ex, you can also remove the
render/1 function altogether and put the template code at
lib/my_app_web/live/demo_live.html.heex.

  
    
  
  Async Operations


Performing asynchronous work is common in LiveViews and LiveComponents.
It allows the user to get a working UI quickly while the system fetches some
data in the background or talks to an external service, without blocking the
render or event handling. For async work, you also typically need to handle
the different states of the async operation, such as loading, error, and the
successful result. You also want to catch any errors or exits and translate it
to a meaningful update in the UI rather than crashing the user experience.

  
    
  
  Async assigns


The assign_async/3 function takes the socket, a key or list of keys which will be assigned
asynchronously, and a function. This function will be wrapped in a task by
assign_async, making it easy for you to return the result. This function must
return an {:ok, assigns} or {:error, reason} tuple, where assigns is a map
of the keys passed to assign_async.
If the function returns anything else, an error is raised.
The task is only started when the socket is connected.
For example, let's say we want to async fetch a user's organization from the database,
as well as their profile and rank:
def mount(%{"slug" => slug}, _, socket) do
  {:ok,
   socket
   |> assign(:foo, "bar")
   |> assign_async(:org, fn -> {:ok, %{org: fetch_org!(slug)}} end)
   |> assign_async([:profile, :rank], fn -> {:ok, %{profile: ..., rank: ...}} end)}
end

  
    
  
  Warning


When using async operations it is important to not pass the socket into the function
as it will copy the whole socket struct to the Task process, which can be very expensive.
Instead of:
assign_async(:org, fn -> {:ok, %{org: fetch_org(socket.assigns.slug)}} end)
We should do:
slug = socket.assigns.slug
assign_async(:org, fn -> {:ok, %{org: fetch_org(slug)}} end)
See: https://hexdocs.pm/elixir/process-anti-patterns.html#sending-unnecessary-data
The state of the async operation is stored in the socket assigns within an
Phoenix.LiveView.AsyncResult. It carries the loading and failed states, as
well as the result. For example, if we wanted to show the loading states in
the UI for the :org, our template could conditionally render the states:
<div :if={@org.loading}>Loading organization...</div>
<div :if={org = @org.ok? && @org.result}>{org.name} loaded!</div>
The Phoenix.Component.async_result/1 function component can also be used to
declaratively render the different states using slots:
<.async_result :let={org} assign={@org}>
  <:loading>Loading organization...</:loading>
  <:failed :let={_failure}>there was an error loading the organization</:failed>
  {org.name}
</.async_result>

  
    
  
  Arbitrary async operations


Sometimes you need lower level control of asynchronous operations, while
still receiving process isolation and error handling. For this, you can use
start_async/3 and the Phoenix.LiveView.AsyncResult module directly:
def mount(%{"id" => id}, _, socket) do
  {:ok,
   socket
   |> assign(:org, AsyncResult.loading())
   |> start_async(:my_task, fn -> fetch_org!(id) end)}
end

def handle_async(:my_task, {:ok, fetched_org}, socket) do
  %{org: org} = socket.assigns
  {:noreply, assign(socket, :org, AsyncResult.ok(org, fetched_org))}
end

def handle_async(:my_task, {:exit, reason}, socket) do
  %{org: org} = socket.assigns
  {:noreply, assign(socket, :org, AsyncResult.failed(org, {:exit, reason}))}
end
start_async/3 is used to fetch the organization asynchronously. The
handle_async/3 callback is called when the task completes or exits,
with the results wrapped in either {:ok, result} or {:exit, reason}.
The AsyncResult module provides functions to update the state of the
async operation, but you can also assign any value directly to the socket
if you want to handle the state yourself.

  
    
  
  Endpoint configuration


LiveView accepts the following configuration in your endpoint under
the :live_view key:
	:signing_salt (required) - the salt used to sign data sent
to the client

	:hibernate_after (optional) - the idle time in milliseconds allowed in
the LiveView before compressing its own memory and state.
Defaults to 15000ms (15 seconds)



      


      
        Summary


  
    Types
  


    
      
        unsigned_params()

      


    





  
    Callbacks
  


    
      
        handle_async(name, async_fun_result, socket)

      


        Invoked when the result of an start_async/3 operation is available.



    


    
      
        handle_call(msg, {}, socket)

      


        Invoked to handle calls from other Elixir processes.



    


    
      
        handle_cast(msg, socket)

      


        Invoked to handle casts from other Elixir processes.



    


    
      
        handle_event(event, unsigned_params, socket)

      


        Invoked to handle events sent by the client.



    


    
      
        handle_info(msg, socket)

      


        Invoked to handle messages from other Elixir processes.



    


    
      
        handle_params(unsigned_params, uri, socket)

      


        Invoked after mount and whenever there is a live patch event.



    


    
      
        mount(params, session, socket)

      


        The LiveView entry-point.



    


    
      
        render(assigns)

      


        Renders a template.



    


    
      
        terminate(reason, socket)

      


        Invoked when the LiveView is terminating.



    





  
    Functions
  


    
      
        __live__(opts \\ [])

      


        Defines metadata for a LiveView.



    


    
      
        __using__(opts)

      


        Uses LiveView in the current module to mark it a LiveView.



    


    
      
        allow_upload(socket, name, options)

      


        Allows an upload for the provided name.



    


    
      
        assign_async(socket, key_or_keys, func, opts \\ [])

      


        Assigns keys asynchronously.



    


    
      
        attach_hook(socket, name, stage, fun)

      


        Attaches the given fun by name for the lifecycle stage into socket.



    


    
      
        cancel_async(socket, async_or_keys, reason \\ {:shutdown, :cancel})

      


        Cancels an async operation if one exists.



    


    
      
        cancel_upload(socket, name, entry_ref)

      


        Cancels an upload for the given entry.



    


    
      
        clear_flash(socket)

      


        Clears the flash.



    


    
      
        clear_flash(socket, key)

      


        Clears a key from the flash.



    


    
      
        connected?(socket)

      


        Returns true if the socket is connected.



    


    
      
        consume_uploaded_entries(socket, name, func)

      


        Consumes the uploaded entries.



    


    
      
        consume_uploaded_entry(socket, entry, func)

      


        Consumes an individual uploaded entry.



    


    
      
        detach_hook(socket, name, stage)

      


        Detaches a hook with the given name from the lifecycle stage.



    


    
      
        disallow_upload(socket, name)

      


        Revokes a previously allowed upload from allow_upload/3.



    


    
      
        get_connect_info(socket, key)

      


        Accesses a given connect info key from the socket.



    


    
      
        get_connect_params(socket)

      


        Accesses the connect params sent by the client for use on connected mount.



    


    
      
        on_mount(mod_or_mod_arg)

      


        Declares a module callback to be invoked on the LiveView's mount.



    


    
      
        push_event(socket, event, payload)

      


        Pushes an event to the client.



    


    
      
        push_navigate(socket, opts)

      


        Annotates the socket for navigation to another LiveView in the same live_session.



    


    
      
        push_patch(socket, opts)

      


        Annotates the socket for navigation within the current LiveView.



    


    
      
        put_flash(socket, kind, msg)

      


        Adds a flash message to the socket to be displayed.



    


    
      
        put_private(socket, key, value)

      


        Puts a new private key and value in the socket.



    


    
      
        redirect(socket, opts \\ [])

      


        Annotates the socket for redirect to a destination path.



    


    
      
        render_with(socket, component)

      


        Configures which function to use to render a LiveView/LiveComponent.



    


    
      
        send_update(pid \\ self(), module_or_cid, assigns)

      


        Asynchronously updates a Phoenix.LiveComponent with new assigns.



    


    
      
        send_update_after(pid \\ self(), module_or_cid, assigns, time_in_milliseconds)

      


        Similar to send_update/3 but the update will be delayed according to the given time_in_milliseconds.



    


    
      
        start_async(socket, name, func, opts \\ [])

      


        Wraps your function in an asynchronous task and invokes a callback name to
handle the result.



    


    
      
        static_changed?(socket)

      


        Returns true if the socket is connected and the tracked static assets have changed.



    


    
      
        stream(socket, name, items, opts \\ [])

      


        Assigns a new stream to the socket or inserts items into an existing stream.
Returns an updated socket.



    


    
      
        stream_configure(socket, name, opts)

      


        Configures a stream.



    


    
      
        stream_delete(socket, name, item)

      


        Deletes an item from the stream.



    


    
      
        stream_delete_by_dom_id(socket, name, id)

      


        Deletes an item from the stream given its computed DOM id.



    


    
      
        stream_insert(socket, name, item, opts \\ [])

      


        Inserts a new item or updates an existing item in the stream.



    


    
      
        transport_pid(socket)

      


        Returns the transport pid of the socket.



    


    
      
        uploaded_entries(socket, name)

      


        Returns the completed and in progress entries for the upload.



    





      


      
        Types

        


  
    
      
    
    
      unsigned_params()



        
          
        

    

  


  

      

          @type unsigned_params() :: map()


      



  


        

      

      
        Callbacks

        


  
    
      
    
    
      handle_async(name, async_fun_result, socket)


        (optional)


        
          
        

    

  


  

      

          @callback handle_async(
  name :: term(),
  async_fun_result :: {:ok, term()} | {:exit, term()},
  socket :: Phoenix.LiveView.Socket.t()
) :: {:noreply, Phoenix.LiveView.Socket.t()}


      


Invoked when the result of an start_async/3 operation is available.
For a deeper understanding of using this callback,
refer to the "Arbitrary async operations" section.

  



  
    
      
    
    
      handle_call(msg, {}, socket)


        (optional)


        
          
        

    

  


  

      

          @callback handle_call(
  msg :: term(),
  {pid(), reference()},
  socket :: Phoenix.LiveView.Socket.t()
) ::
  {:noreply, Phoenix.LiveView.Socket.t()}
  | {:reply, term(), Phoenix.LiveView.Socket.t()}


      


Invoked to handle calls from other Elixir processes.
See GenServer.call/3 and GenServer.handle_call/3
for more information.

  



  
    
      
    
    
      handle_cast(msg, socket)


        (optional)


        
          
        

    

  


  

      

          @callback handle_cast(msg :: term(), socket :: Phoenix.LiveView.Socket.t()) ::
  {:noreply, Phoenix.LiveView.Socket.t()}


      


Invoked to handle casts from other Elixir processes.
See GenServer.cast/2 and GenServer.handle_cast/2
for more information. It must always return {:noreply, socket},
where :noreply means no additional information is sent
to the process which cast the message.

  



  
    
      
    
    
      handle_event(event, unsigned_params, socket)


        (optional)


        
          
        

    

  


  

      

          @callback handle_event(
  event :: binary(),
  unsigned_params(),
  socket :: Phoenix.LiveView.Socket.t()
) ::
  {:noreply, Phoenix.LiveView.Socket.t()}
  | {:reply, map(), Phoenix.LiveView.Socket.t()}


      


Invoked to handle events sent by the client.
It receives the event name, the event payload as a map,
and the socket.
It must return {:noreply, socket}, where :noreply means
no additional information is sent to the client, or
{:reply, map(), socket}, where the given map() is encoded
and sent as a reply to the client.

  



  
    
      
    
    
      handle_info(msg, socket)


        (optional)


        
          
        

    

  


  

      

          @callback handle_info(msg :: term(), socket :: Phoenix.LiveView.Socket.t()) ::
  {:noreply, Phoenix.LiveView.Socket.t()}


      


Invoked to handle messages from other Elixir processes.
See Kernel.send/2 and GenServer.handle_info/2
for more information. It must always return {:noreply, socket},
where :noreply means no additional information is sent
to the process which sent the message.

  



  
    
      
    
    
      handle_params(unsigned_params, uri, socket)


        (optional)


        
          
        

    

  


  

      

          @callback handle_params(
  unsigned_params(),
  uri :: String.t(),
  socket :: Phoenix.LiveView.Socket.t()
) ::
  {:noreply, Phoenix.LiveView.Socket.t()}


      


Invoked after mount and whenever there is a live patch event.
It receives the current params, including parameters from
the router, the current uri from the client and the socket.
It is invoked after mount or whenever there is a live navigation
event caused by push_patch/2 or <.link patch={...}>.
It must always return {:noreply, socket}, where :noreply
means no additional information is sent to the client.
Note
handle_params is only allowed on LiveViews mounted at the router,
as it takes the current url of the page as the second parameter.

  



  
    
      
    
    
      mount(params, session, socket)


        (optional)


        
          
        

    

  


  

      

          @callback mount(
  params :: unsigned_params() | :not_mounted_at_router,
  session :: map(),
  socket :: Phoenix.LiveView.Socket.t()
) ::
  {:ok, Phoenix.LiveView.Socket.t()}
  | {:ok, Phoenix.LiveView.Socket.t(), keyword()}


      


The LiveView entry-point.
For each LiveView in the root of a template, mount/3 is invoked twice:
once to do the initial page load and again to establish the live socket.
It expects three arguments:
	params - a map of string keys which contain public information that
can be set by the user. The map contains the query params as well as any
router path parameter. If the LiveView was not mounted at the router,
this argument is the atom :not_mounted_at_router
	session - the connection session
	socket - the LiveView socket

It must return either {:ok, socket} or {:ok, socket, options}, where
options is one of:
	:temporary_assigns - a keyword list of assigns that are temporary
and must be reset to their value after every render. Note that once
the value is reset, it won't be re-rendered again until it is explicitly
assigned

	:layout - the optional layout to be used by the LiveView. Setting
this option will override any layout previously set via
Phoenix.LiveView.Router.live_session/2 or on use Phoenix.LiveView



  



  
    
      
    
    
      render(assigns)


        (optional)


        
          
        

    

  


  

      

          @callback render(assigns :: Phoenix.LiveView.Socket.assigns()) ::
  Phoenix.LiveView.Rendered.t()


      


Renders a template.
This callback is invoked whenever LiveView detects
new content must be rendered and sent to the client.
If you define this function, it must return a template
defined via the Phoenix.Component.sigil_H/2.
If you don't define this function, LiveView will attempt
to render a template in the same directory as your LiveView.
For example, if you have a LiveView named MyApp.MyCustomView
inside lib/my_app/live_views/my_custom_view.ex, Phoenix
will look for a template at lib/my_app/live_views/my_custom_view.html.heex.

  



  
    
      
    
    
      terminate(reason, socket)


        (optional)


        
          
        

    

  


  

      

          @callback terminate(reason, socket :: Phoenix.LiveView.Socket.t()) :: term()
when reason: :normal | :shutdown | {:shutdown, :left | :closed | term()}


      


Invoked when the LiveView is terminating.
In case of errors, this callback is only invoked if the LiveView
is trapping exits. See GenServer.terminate/2 for more info.

  


        

      

      
        Functions

        


    

  
    
      
    
    
      __live__(opts \\ [])



        
          
        

    

  


  

Defines metadata for a LiveView.
This must be returned from the __live__ callback.
It accepts:
	:container - an optional tuple for the HTML tag and DOM attributes to
be used for the LiveView container. For example: {:li, style: "color: blue;"}.

	:layout - configures the layout the LiveView will be rendered in.
This layout can be overridden by on mount/3 or via the :layout
option in Phoenix.LiveView.Router.live_session/2

	:log - configures the log level for the LiveView, either false
or a log level

	:on_mount - a list of tuples with module names and argument to be invoked
as on_mount hooks



  



  
    
      
    
    
      __using__(opts)


        (macro)


        
          
        

    

  


  

Uses LiveView in the current module to mark it a LiveView.
use Phoenix.LiveView,
  container: {:tr, class: "colorized"},
  layout: {MyAppWeb.Layouts, :app},
  log: :info

  
    
  
  Options


	:container - an optional tuple for the HTML tag and DOM attributes to
be used for the LiveView container. For example: {:li, style: "color: blue;"}.
See Phoenix.Component.live_render/3 for more information and examples.

	:global_prefixes - the global prefixes to use for components. See
Global Attributes in Phoenix.Component for more information.

	:layout - configures the layout the LiveView will be rendered in.
This layout can be overridden by on mount/3 or via the :layout
option in Phoenix.LiveView.Router.live_session/2

	:log - configures the log level for the LiveView, either false
or a log level



  



  
    
      
    
    
      allow_upload(socket, name, options)



        
          
        

    

  


  

Allows an upload for the provided name.

  
    
  
  Options


	:accept - Required. A list of unique file extensions (such as ".jpeg") or
mime type (such as "image/jpeg" or "image/*"). You may also pass the atom
:any instead of a list to support to allow any kind of file.
For example, [".jpeg"], :any, etc.

	:max_entries - The maximum number of selected files to allow per
file input. Defaults to 1.

	:max_file_size - The maximum file size in bytes to allow to be uploaded.
Defaults 8MB. For example, 12_000_000.

	:chunk_size - The chunk size in bytes to send when uploading.
Defaults 64_000.

	:chunk_timeout - The time in milliseconds to wait before closing the
upload channel when a new chunk has not been received. Defaults to 10_000.

	:external - A 2-arity function for generating metadata for external
client uploaders. This function must return either {:ok, meta, socket}
or {:error, meta, socket} where meta is a map. See the Uploads section
for example usage.

	:progress - An optional 3-arity function for receiving progress events.

	:auto_upload - Instructs the client to upload the file automatically
on file selection instead of waiting for form submits. Defaults to false.

	:writer - A module implementing the Phoenix.LiveView.UploadWriter
behaviour to use for writing the uploaded chunks. Defaults to writing to a
temporary file for consumption. See the Phoenix.LiveView.UploadWriter docs
for custom usage.


Raises when a previously allowed upload under the same name is still active.

  
    
  
  Examples


allow_upload(socket, :avatar, accept: ~w(.jpg .jpeg), max_entries: 2)
allow_upload(socket, :avatar, accept: :any)
For consuming files automatically as they are uploaded, you can pair auto_upload: true with
a custom progress function to consume the entries as they are completed. For example:
allow_upload(socket, :avatar, accept: :any, progress: &handle_progress/3, auto_upload: true)

defp handle_progress(:avatar, entry, socket) do
  if entry.done? do
    uploaded_file =
      consume_uploaded_entry(socket, entry, fn %{} = meta ->
        {:ok, ...}
      end)

    {:noreply, put_flash(socket, :info, "file #{uploaded_file.name} uploaded")}
  else
    {:noreply, socket}
  end
end

  



    

  
    
      
    
    
      assign_async(socket, key_or_keys, func, opts \\ [])


        (macro)


        
          
        

    

  


  

Assigns keys asynchronously.
Wraps your function in a task linked to the caller, errors are wrapped.
Each key passed to assign_async/3 will be assigned to
an Phoenix.LiveView.AsyncResult struct holding the status of the operation
and the result when the function completes.
The task is only started when the socket is connected.

  
    
  
  Options


	:supervisor - allows you to specify a Task.Supervisor to supervise the task.
	:reset - remove previous results during async operation when true. Possible values are
true, false, or a list of keys to reset. Defaults to false.


  
    
  
  Examples


def mount(%{"slug" => slug}, _, socket) do
  {:ok,
    socket
    |> assign(:foo, "bar")
    |> assign_async(:org, fn -> {:ok, %{org: fetch_org!(slug)}} end)
    |> assign_async([:profile, :rank], fn -> {:ok, %{profile: ..., rank: ...}} end)}
end
See Async Operations for more information.

  
    
  
  assign_async/3 and send_update/3


Since the code inside assign_async/3 runs in a separate process,
send_update(Component, data) does not work inside assign_async/3,
since send_update/2 assumes it is running inside the LiveView process.
The solution is to explicitly send the update to the LiveView:
parent = self()
assign_async(socket, :org, fn ->
  # ...
  send_update(parent, Component, data)
end)

  
    
  
  Testing async operations


When testing LiveViews and LiveComponents with async assigns, use
Phoenix.LiveViewTest.render_async/2 to ensure the test waits until the async operations
are complete before proceeding with assertions or before ending the test. For example:
{:ok, view, _html} = live(conn, "/my_live_view")
html = render_async(view)
assert html =~ "My assertion"
Not calling render_async/2 to ensure all async assigns have finished might result in errors in
cases where your process has side effects:
[error] MyXQL.Connection (#PID<0.308.0>) disconnected: ** (DBConnection.ConnectionError) client #PID<0.794.0>

  



  
    
      
    
    
      attach_hook(socket, name, stage, fun)



        
          
        

    

  


  

Attaches the given fun by name for the lifecycle stage into socket.
Note: This function is for server-side lifecycle callbacks.
For client-side hooks, see the
JS Interop guide.

Hooks provide a mechanism to tap into key stages of the LiveView
lifecycle in order to bind/update assigns, intercept events,
patches, and regular messages when necessary, and to inject
common functionality. Use attach_hook/4 on any of the following
lifecycle stages: :handle_params, :handle_event, :handle_info, :handle_async, and
:after_render. To attach a hook to the :mount stage, use on_mount/1.
Note: only :after_render and :handle_event hooks are currently supported in
LiveComponents.


  
    
  
  Return Values


Lifecycle hooks take place immediately before a given lifecycle
callback is invoked on the LiveView. With the exception of :after_render,
a hook may return {:halt, socket} to halt the reduction, otherwise
it must return {:cont, socket} so the operation may continue until
all hooks have been invoked for the current stage.
For :after_render hooks, the socket itself must be returned.
Any updates to the socket assigns will not trigger a new render
or diff calculation to the client.

  
    
  
  Halting the lifecycle


Note that halting from a hook will halt the entire lifecycle stage.
This means that when a hook returns {:halt, socket} then the
LiveView callback will not be invoked. This has some
implications.

  
    
  
  Implications for plugin authors


When defining a plugin that matches on specific callbacks, you must
define a catch-all clause, as your hook will be invoked even for events
you may not be interested in.

  
    
  
  Implications for end-users


Allowing a hook to halt the invocation of the callback means that you can
attach hooks to intercept specific events before detaching themselves,
while allowing other events to continue normally.

  
    
  
  Replying to events


Hooks attached to the :handle_event stage are able to reply to client events
by returning {:halt, reply, socket}. This is useful especially for JavaScript
interoperability because a client hook
can push an event and receive a reply.

  
    
  
  Sharing event handling logic


Lifecycle hooks are an excellent way to extract related events out of the parent LiveView and
into separate modules without resorting unnecessarily to LiveComponents for organization.
defmodule DemoLive do
  use Phoenix.LiveView

  def render(assigns) do
    ~H"""
    <div>
      <div>
        Counter: {@counter}
        <button phx-click="inc">+</button>
      </div>

      <MySortComponent.display lists={[first_list: @first_list, second_list: @second_list]} />
    </div>
    """
  end

  def mount(_params, _session, socket) do
    first_list = for(i <- 1..9, do: "First List #{i}") |> Enum.shuffle()
    second_list = for(i <- 1..9, do: "Second List #{i}") |> Enum.shuffle()

    socket =
      socket
      |> assign(:counter, 0)
      |> assign(first_list: first_list)
      |> assign(second_list: second_list)
      |> attach_hook(:sort, :handle_event, &MySortComponent.hooked_event/3)  # 2) Delegated events
    {:ok, socket}
  end

  # 1) Normal event
  def handle_event("inc", _params, socket) do
    {:noreply, update(socket, :counter, &(&1 + 1))}
  end
end

defmodule MySortComponent do
  use Phoenix.Component

  def display(assigns) do
    ~H"""
    <div :for={{key, list} <- @lists}>
      <ul><li :for={item <- list}>{item}</li></ul>
      <button phx-click="shuffle" phx-value-list={key}>Shuffle</button>
      <button phx-click="sort" phx-value-list={key}>Sort</button>
    </div>
    """
  end

  def hooked_event("shuffle", %{"list" => key}, socket) do
    key = String.to_existing_atom(key)
    shuffled = Enum.shuffle(socket.assigns[key])

    {:halt, assign(socket, key, shuffled)}
  end

  def hooked_event("sort", %{"list" => key}, socket) do
    key = String.to_existing_atom(key)
    sorted = Enum.sort(socket.assigns[key])

    {:halt, assign(socket, key, sorted)}
  end

  def hooked_event(_event, _params, socket), do: {:cont, socket}
end

  
    
  
  Other examples


Attaching and detaching a hook:
def mount(_params, _session, socket) do
  socket =
    attach_hook(socket, :my_hook, :handle_event, fn
      "very-special-event", _params, socket ->
        # Handle the very special event and then detach the hook
        {:halt, detach_hook(socket, :my_hook, :handle_event)}

      _event, _params, socket ->
        {:cont, socket}
    end)

  {:ok, socket}
end
Replying to a client event:
/**
 * @type {import("phoenix_live_view").HooksOption}
 */
let Hooks = {}
Hooks.ClientHook = {
  mounted() {
    this.pushEvent("ClientHook:mounted", {hello: "world"}, (reply) => {
      console.log("received reply:", reply)
    })
  }
}
let liveSocket = new LiveSocket("/live", Socket, {hooks: Hooks, ...})
def render(assigns) do
  ~H"""
  <div id="my-client-hook" phx-hook="ClientHook"></div>
  """
end

def mount(_params, _session, socket) do
  socket =
    attach_hook(socket, :reply_on_client_hook_mounted, :handle_event, fn
      "ClientHook:mounted", params, socket ->
        {:halt, params, socket}

      _, _, socket ->
        {:cont, socket}
    end)

  {:ok, socket}
end

  



    

  
    
      
    
    
      cancel_async(socket, async_or_keys, reason \\ {:shutdown, :cancel})



        
          
        

    

  


  

Cancels an async operation if one exists.
Accepts either the %AsyncResult{} when using assign_async/3 or
the key passed to start_async/3.
The underlying process will be killed with the provided reason, or
with {:shutdown, :cancel} if no reason is passed. For assign_async/3
operations, the :failed field will be set to {:exit, reason}.
For start_async/3, the handle_async/3 callback will receive
{:exit, reason} as the result.
Returns the %Phoenix.LiveView.Socket{}.

  
    
  
  Examples


cancel_async(socket, :preview)
cancel_async(socket, :preview, :my_reason)
cancel_async(socket, socket.assigns.preview)

  



  
    
      
    
    
      cancel_upload(socket, name, entry_ref)



        
          
        

    

  


  

Cancels an upload for the given entry.

  
    
  
  Examples


<%= for entry <- @uploads.avatar.entries do %>
  ...
  <button phx-click="cancel-upload" phx-value-ref={entry.ref}>cancel</button>
<% end %>
def handle_event("cancel-upload", %{"ref" => ref}, socket) do
  {:noreply, cancel_upload(socket, :avatar, ref)}
end

  



  
    
      
    
    
      clear_flash(socket)



        
          
        

    

  


  

Clears the flash.

  
    
  
  Examples


iex> clear_flash(socket)
Clearing the flash can also be triggered on the client and natively handled by LiveView using the lv:clear-flash event.
For example:
<p class="alert" phx-click="lv:clear-flash">
  {Phoenix.Flash.get(@flash, :info)}
</p>

  



  
    
      
    
    
      clear_flash(socket, key)



        
          
        

    

  


  

Clears a key from the flash.

  
    
  
  Examples


iex> clear_flash(socket, :info)
Clearing the flash can also be triggered on the client and natively handled by LiveView using the lv:clear-flash event.
For example:
<p class="alert" phx-click="lv:clear-flash" phx-value-key="info">
  {Phoenix.Flash.get(@flash, :info)}
</p>

  



  
    
      
    
    
      connected?(socket)



        
          
        

    

  


  

Returns true if the socket is connected.
Useful for checking the connectivity status when mounting the view.
For example, on initial page render, the view is mounted statically,
rendered, and the HTML is sent to the client. Once the client
connects to the server, a LiveView is then spawned and mounted
statefully within a process. Use connected?/1 to conditionally
perform stateful work, such as subscribing to pubsub topics,
sending messages, etc.

  
    
  
  Examples


defmodule DemoWeb.ClockLive do
  use Phoenix.LiveView
  ...
  def mount(_params, _session, socket) do
    if connected?(socket), do: :timer.send_interval(1000, self(), :tick)

    {:ok, assign(socket, date: :calendar.local_time())}
  end

  def handle_info(:tick, socket) do
    {:noreply, assign(socket, date: :calendar.local_time())}
  end
end

  



  
    
      
    
    
      consume_uploaded_entries(socket, name, func)



        
          
        

    

  


  

Consumes the uploaded entries.
Raises when there are still entries in progress.
Typically called when submitting a form to handle the
uploaded entries alongside the form data. For form submissions,
it is guaranteed that all entries have completed before the submit event
is invoked. Once entries are consumed, they are removed from the upload.
The function passed to consume may return a tagged tuple of the form
{:ok, my_result} to collect results about the consumed entries, or
{:postpone, my_result} to collect results, but postpone the file
consumption to be performed later.
A list of all my_result values produced by the passed function is
returned, regardless of whether they were consumed or postponed.

  
    
  
  Examples


def handle_event("save", _params, socket) do
  uploaded_files =
    consume_uploaded_entries(socket, :avatar, fn %{path: path}, _entry ->
      dest = Path.join("priv/static/uploads", Path.basename(path))
      File.cp!(path, dest)
      {:ok, ~p"/uploads/#{Path.basename(dest)}"}
    end)
  {:noreply, update(socket, :uploaded_files, &(&1 ++ uploaded_files))}
end

  



  
    
      
    
    
      consume_uploaded_entry(socket, entry, func)



        
          
        

    

  


  

Consumes an individual uploaded entry.
Raises when the entry is still in progress.
Typically called when submitting a form to handle the
uploaded entries alongside the form data. Once entries are consumed,
they are removed from the upload.
This is a lower-level feature than consume_uploaded_entries/3 and useful
for scenarios where you want to consume entries as they are individually completed.
Like consume_uploaded_entries/3, the function passed to consume may return
a tagged tuple of the form {:ok, my_result} to collect results about the
consumed entries, or {:postpone, my_result} to collect results,
but postpone the file consumption to be performed later.

  
    
  
  Examples


def handle_event("save", _params, socket) do
  case uploaded_entries(socket, :avatar) do
    {[_|_] = entries, []} ->
      uploaded_files = for entry <- entries do
        consume_uploaded_entry(socket, entry, fn %{path: path} ->
          dest = Path.join("priv/static/uploads", Path.basename(path))
          File.cp!(path, dest)
          {:ok, ~p"/uploads/#{Path.basename(dest)}"}
        end)
      end
      {:noreply, update(socket, :uploaded_files, &(&1 ++ uploaded_files))}

    _ ->
      {:noreply, socket}
  end
end

  



  
    
      
    
    
      detach_hook(socket, name, stage)



        
          
        

    

  


  

Detaches a hook with the given name from the lifecycle stage.
Note: This function is for server-side lifecycle callbacks.
For client-side hooks, see the
JS Interop guide.

If no hook is found, this function is a no-op.

  
    
  
  Examples


def handle_event(_, socket) do
  {:noreply, detach_hook(socket, :hook_that_was_attached, :handle_event)}
end

  



  
    
      
    
    
      disallow_upload(socket, name)



        
          
        

    

  


  

Revokes a previously allowed upload from allow_upload/3.

  
    
  
  Examples


disallow_upload(socket, :avatar)

  



  
    
      
    
    
      get_connect_info(socket, key)



        
          
        

    

  


  

Accesses a given connect info key from the socket.
The following keys are supported: :peer_data, :trace_context_headers,
:x_headers, :uri, and :user_agent.
The connect information is available only during mount. During disconnected
render, all keys are available. On connected render, only the keys explicitly
declared in your socket are available. See Phoenix.Endpoint.socket/3 for
a complete description of the keys.

  
    
  
  Examples


The first step is to declare the connect_info you want to receive.
Typically, it includes at least the session, but you must include all
other keys you want to access on connected mount, such as :peer_data:
socket "/live", Phoenix.LiveView.Socket,
  websocket: [connect_info: [:peer_data, session: @session_options]]
Those values can now be accessed on the connected mount as
get_connect_info/2:
def mount(_params, _session, socket) do
  peer_data = get_connect_info(socket, :peer_data)
  {:ok, assign(socket, ip: peer_data.address)}
end
If the key is not available, usually because it was not specified
in connect_info, it returns nil.

  



  
    
      
    
    
      get_connect_params(socket)



        
          
        

    

  


  

Accesses the connect params sent by the client for use on connected mount.
Connect params are sent from the client on every connection and reconnection.
The parameters in the client can be computed dynamically, allowing you to pass
client state to the server. For example, you could use it to compute and pass
the user time zone from a JavaScript client:
let liveSocket = new LiveSocket("/live", Socket, {
  longPollFallbackMs: 2500,
  params: (_liveViewName) => {
    return {
      _csrf_token: csrfToken,
      time_zone: Intl.DateTimeFormat().resolvedOptions().timeZone
    }
  }
})
By computing the parameters with a function, reconnections will reevalute
the code, allowing you to fetch the latest data.
On the LiveView, you will use get_connect_params/1 to read the data,
which only remains available during mount. nil is returned when called
in a disconnected state and a RuntimeError is raised if called after
mount.

  
    
  
  Reserved params


The following params have special meaning in LiveView:
	"_csrf_token" - the CSRF Token which must be explicitly set by the user
when connecting
	"_mounts" - the number of times the current LiveView is mounted.
It is 0 on first mount, then increases on each reconnect. It resets
when navigating away from the current LiveView or on errors
	"_track_static" - set automatically with a list of all href/src from
tags with the phx-track-static annotation in them. If there are no
such tags, nothing is sent
	"_live_referer" - sent by the client as the referer URL when a
live navigation has occurred from push_navigate or client link navigate.


  
    
  
  Examples


def mount(_params, _session, socket) do
  {:ok, assign(socket, width: get_connect_params(socket)["width"] || @width)}
end

  



  
    
      
    
    
      on_mount(mod_or_mod_arg)


        (macro)


        
          
        

    

  


  

Declares a module callback to be invoked on the LiveView's mount.
The function within the given module, which must be named on_mount,
will be invoked before both disconnected and connected mounts. The hook
has the option to either halt or continue the mounting process as usual.
If you wish to redirect the LiveView, you must halt, otherwise an error
will be raised.
Tip: if you need to define multiple on_mount callbacks, avoid defining
multiple modules. Instead, pass a tuple and use pattern matching to handle
different cases:
def on_mount(:admin, _params, _session, socket) do
  {:cont, socket}
end

def on_mount(:user, _params, _session, socket) do
  {:cont, socket}
end
And then invoke it as:
on_mount {MyAppWeb.SomeHook, :admin}
on_mount {MyAppWeb.SomeHook, :user}
Registering on_mount hooks can be useful to perform authentication
as well as add custom behaviour to other callbacks via attach_hook/4.
The on_mount callback can return a keyword list of options as a third
element in the return tuple. These options are identical to what can
optionally be returned in mount/3.

  
    
  
  Examples


The following is an example of attaching a hook via
Phoenix.LiveView.Router.live_session/3:
# lib/my_app_web/live/init_assigns.ex
defmodule MyAppWeb.InitAssigns do
  @moduledoc """
  Ensures common `assigns` are applied to all LiveViews attaching this hook.
  """
  import Phoenix.LiveView
  import Phoenix.Component

  def on_mount(:default, _params, _session, socket) do
    {:cont, assign(socket, :page_title, "DemoWeb")}
  end

  def on_mount(:user, params, session, socket) do
    # code
  end

  def on_mount(:admin, _params, _session, socket) do
    {:cont, socket, layout: {DemoWeb.Layouts, :admin}}
  end
end

# lib/my_app_web/router.ex
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  # pipelines, plugs, etc.

  live_session :default, on_mount: MyAppWeb.InitAssigns do
    scope "/", MyAppWeb do
      pipe_through :browser
      live "/", PageLive, :index
    end
  end

  live_session :authenticated, on_mount: {MyAppWeb.InitAssigns, :user} do
    scope "/", MyAppWeb do
      pipe_through [:browser, :require_user]
      live "/profile", UserLive.Profile, :index
    end
  end

  live_session :admins, on_mount: {MyAppWeb.InitAssigns, :admin} do
    scope "/admin", MyAppWeb.Admin do
      pipe_through [:browser, :require_user, :require_admin]
      live "/", AdminLive.Index, :index
    end
  end
end

  



  
    
      
    
    
      push_event(socket, event, payload)



        
          
        

    

  


  

Pushes an event to the client.
Events can be handled in two ways:
	They can be handled on window via addEventListener.
A "phx:" prefix will be added to the event name.

	They can be handled inside a hook via handleEvent.


Events are dispatched to all active hooks on the client who are
handling the given event. If you need to scope events, then
this must be done by namespacing them.
Events pushed during push_navigate are currently discarded,
as the LiveView is immediately dismounted.

  
    
  
  Hook example


If you push a "scores" event from your LiveView:
{:noreply, push_event(socket, "scores", %{points: 100, user: "josé"})}
A hook declared via phx-hook can handle it via handleEvent:
this.handleEvent("scores", data => ...)

  
    
  
  window example


All events are also dispatched on the window. This means you can handle
them by adding listeners. For example, if you want to remove an element
from the page, you can do this:
{:noreply, push_event(socket, "remove-el", %{id: "foo-bar"})}
And now in your app.js you can register and handle it:
window.addEventListener(
  "phx:remove-el",
  e => document.getElementById(e.detail.id).remove()
)

  



  
    
      
    
    
      push_navigate(socket, opts)



        
          
        

    

  


  

Annotates the socket for navigation to another LiveView in the same live_session.
The current LiveView will be shutdown and a new one will be mounted
in its place, without reloading the whole page. This can
also be used to remount the same LiveView, in case you want to start
fresh. If you want to navigate to the same LiveView without remounting
it, use push_patch/2 instead.

  
    
  
  Options


	:to - the required path to link to. It must always be a local path
	:replace - the flag to replace the current history or push a new state.
Defaults false.


  
    
  
  Examples


{:noreply, push_navigate(socket, to: "/")}
{:noreply, push_navigate(socket, to: "/", replace: true)}

  



  
    
      
    
    
      push_patch(socket, opts)



        
          
        

    

  


  

Annotates the socket for navigation within the current LiveView.
When navigating to the current LiveView, handle_params/3 is
immediately invoked to handle the change of params and URL state.
Then the new state is pushed to the client, without reloading the
whole page while also maintaining the current scroll position.
For live navigation to another LiveView in the same live_session,
use push_navigate/2. Otherwise, use redirect/2.

  
    
  
  Options


	:to - the required path to link to. It must always be a local path
	:replace - the flag to replace the current history or push a new state.
Defaults false.


  
    
  
  Examples


{:noreply, push_patch(socket, to: "/")}
{:noreply, push_patch(socket, to: "/", replace: true)}

  



  
    
      
    
    
      put_flash(socket, kind, msg)



        
          
        

    

  


  

Adds a flash message to the socket to be displayed.
The flash message will stick around until it is read.
If you perform a redirect or a navigation event, the message will be
signed and temporarily stored in the client. Therefore it is important
to use flash messages only for user-facing notifications. Do not store
sensitive information in flash messages.
In a typical LiveView application, the message will be rendered by the
CoreComponents’ flash/1 component. It is up to this function to determine
what kind of messages it supports. By default, the :info and :error
kinds are handled.
Note: You must also place the Phoenix.LiveView.Router.fetch_live_flash/2
plug in your browser's pipeline in place of fetch_flash for LiveView flash
messages be supported, for example:
import Phoenix.LiveView.Router

pipeline :browser do
  ...
  plug :fetch_live_flash
end

  
    
  
  Examples


iex> put_flash(socket, :info, "It worked!")
iex> put_flash(socket, :error, "You can't access that page")

  
    
  
  Inside components


You can use put_flash/3 inside a Phoenix.LiveComponent and
components have their own @flash assigns. The @flash assign
in a component is only copied to its parent LiveView if the component
calls push_navigate/2 or push_patch/2.

  



  
    
      
    
    
      put_private(socket, key, value)



        
          
        

    

  


  

Puts a new private key and value in the socket.
Privates are not change tracked. This storage is meant to be used by
users and libraries to hold state that doesn't require
change tracking. The keys should be prefixed with the app/library name.

  
    
  
  Examples


Key values can be placed in private:
put_private(socket, :myapp_meta, %{foo: "bar"})
And then retrieved:
socket.private[:myapp_meta]

  



    

  
    
      
    
    
      redirect(socket, opts \\ [])



        
          
        

    

  


  

Annotates the socket for redirect to a destination path.
Note: LiveView redirects rely on instructing client
to perform a window.location update on the provided
redirect location. The whole page will be reloaded and
all state will be discarded.

  
    
  
  Options


	:to - the path to redirect to. It must always be a local path
	:status - the HTTP status code to use for the redirect. Defaults to 302.
	:external - an external path to redirect to. Either a string
or {scheme, url} to redirect to a custom scheme


  
    
  
  Examples


{:noreply, redirect(socket, to: "/")}
{:noreply, redirect(socket, to: "/", status: 301)}
{:noreply, redirect(socket, external: "https://example.com")}

  



  
    
      
    
    
      render_with(socket, component)



        
          
        

    

  


  

Configures which function to use to render a LiveView/LiveComponent.
By default, LiveView invokes the render/1 function in the same module
the LiveView/LiveComponent is defined, passing assigns as its sole
argument. This function allows you to set a different rendering function.
One possible use case for this function is to set a different template
on disconnected render. When the user first accesses a LiveView, we will
perform a disconnected render to send to the browser. This is useful for
several reasons, such as reducing the time to first paint and for search
engine indexing.
However, when LiveView is gated behind an authentication page, it may be
useful to render a placeholder on disconnected render and perform the
full render once the WebSocket connects. This can be achieved with
render_with/2 and is particularly useful on complex pages (such as
dashboards and reports).
To do so, you must simply invoke render_with(socket, &some_function_component/1),
configuring your socket with a new rendering function.

  



    

  
    
      
    
    
      send_update(pid \\ self(), module_or_cid, assigns)



        
          
        

    

  


  

Asynchronously updates a Phoenix.LiveComponent with new assigns.
The pid argument is optional and it defaults to the current process,
which means the update instruction will be sent to a component running
on the same LiveView. If the current process is not a LiveView or you
want to send updates to a live component running on another LiveView,
you should explicitly pass the LiveView's pid instead.
The second argument can be either the value of the @myself or the module of
the live component. If you pass the module, then the :id that identifies
the component must be passed as part of the assigns.
When the component receives the update,
update_many/1 will be invoked if
it is defined, otherwise update/2 is
invoked with the new assigns.  If
update/2 is not defined all assigns
are simply merged into the socket. The assigns received as the first argument
of the update/2 callback will only
include the new assigns passed from this function.  Pre-existing assigns may
be found in socket.assigns.
While a component may always be updated from the parent by updating some
parent assigns which will re-render the child, thus invoking
update/2 on the child component,
send_update/3 is useful for updating a component that entirely manages its
own state, as well as messaging between components mounted in the same
LiveView.

  
    
  
  Examples


def handle_event("cancel-order", _, socket) do
  ...
  send_update(Cart, id: "cart", status: "cancelled")
  {:noreply, socket}
end

def handle_event("cancel-order-asynchronously", _, socket) do
  ...
  pid = self()

  Task.Supervisor.start_child(MyTaskSup, fn ->
    # Do something asynchronously
    send_update(pid, Cart, id: "cart", status: "cancelled")
  end)

  {:noreply, socket}
end

def render(assigns) do
  ~H"""
  <.some_component on_complete={&send_update(@myself, completed: &1)} />
  """
end

  



    

  
    
      
    
    
      send_update_after(pid \\ self(), module_or_cid, assigns, time_in_milliseconds)



        
          
        

    

  


  

Similar to send_update/3 but the update will be delayed according to the given time_in_milliseconds.
It returns a reference which can be cancelled with Process.cancel_timer/1.

  
    
  
  Examples


def handle_event("cancel-order", _, socket) do
  ...
  send_update_after(Cart, [id: "cart", status: "cancelled"], 3000)
  {:noreply, socket}
end

def handle_event("cancel-order-asynchronously", _, socket) do
  ...
  pid = self()

  Task.start(fn ->
    # Do something asynchronously
    send_update_after(pid, Cart, [id: "cart", status: "cancelled"], 3000)
  end)

  {:noreply, socket}
end

  



    

  
    
      
    
    
      start_async(socket, name, func, opts \\ [])


        (macro)


        
          
        

    

  


  

Wraps your function in an asynchronous task and invokes a callback name to
handle the result.
The task is linked to the caller and errors/exits are wrapped.
The result of the task is sent to the handle_async/3 callback
of the caller LiveView or LiveComponent.
If there is an in-flight task with the same name, the later start_async wins and the previous task’s result is ignored.
If you wish to replace an existing task, you can use cancel_async/3 before start_async/3.
You are not restricted to just atoms for name, it can be any term such as a tuple.
The task is only started when the socket is connected.

  
    
  
  Options


	:supervisor - allows you to specify a Task.Supervisor to supervise the task.


  
    
  
  Examples


def mount(%{"id" => id}, _, socket) do
  {:ok,
   socket
   |> assign(:org, AsyncResult.loading())
   |> start_async(:my_task, fn -> fetch_org!(id) end)}
end

def handle_async(:my_task, {:ok, fetched_org}, socket) do
  %{org: org} = socket.assigns
  {:noreply, assign(socket, :org, AsyncResult.ok(org, fetched_org))}
end

def handle_async(:my_task, {:exit, reason}, socket) do
  %{org: org} = socket.assigns
  {:noreply, assign(socket, :org, AsyncResult.failed(org, {:exit, reason}))}
end
See the moduledoc for more information.

  



  
    
      
    
    
      static_changed?(socket)



        
          
        

    

  


  

Returns true if the socket is connected and the tracked static assets have changed.
This function is useful to detect if the client is running on an outdated
version of the marked static files. It works by comparing the static paths
sent by the client with the one on the server.
Note: this functionality requires Phoenix v1.5.2 or later.
To use this functionality, the first step is to annotate which static files
you want to be tracked by LiveView, with the phx-track-static. For example:
<link phx-track-static rel="stylesheet" href={~p"/assets/app.css"} />
<script defer phx-track-static type="text/javascript" src={~p"/assets/app.js"}></script>
Now, whenever LiveView connects to the server, it will send a copy src
or href attributes of all tracked statics and compare those values with
the latest entries computed by mix phx.digest in the server.
The tracked statics on the client will match the ones on the server the
huge majority of times. However, if there is a new deployment, those values
may differ. You can use this function to detect those cases and show a
banner to the user, asking them to reload the page. To do so, first set the
assign on mount:
def mount(params, session, socket) do
  {:ok, assign(socket, static_changed?: static_changed?(socket))}
end
And then in your views:
<div :if={@static_changed?} id="reload-static">
  The app has been updated. Click here to <a href="#" onclick="window.location.reload()">reload</a>.
</div>
If you prefer, you can also send a JavaScript script that immediately
reloads the page.
Note: only set phx-track-static on your own assets. For example, do
not set it in external JavaScript files:
<script defer phx-track-static type="text/javascript" src="https://ajax.googleapis.com/ajax/libs/jquery/3.4.1/jquery.min.js"></script>
Because you don't actually serve the file above, LiveView will interpret
the static above as missing, and this function will return true.

  



    

  
    
      
    
    
      stream(socket, name, items, opts \\ [])



        
          
        

    

  


  

      

          @spec stream(
  socket :: Phoenix.LiveView.Socket.t(),
  name :: atom() | String.t(),
  items :: Enumerable.t(),
  opts :: Keyword.t()
) :: Phoenix.LiveView.Socket.t()


      


Assigns a new stream to the socket or inserts items into an existing stream.
Returns an updated socket.
Streams are a mechanism for managing large collections on the client without
keeping the resources on the server.
	name - A string or atom name of the key to place under the
@streams assign.
	items - An enumerable of items to insert.

The following options are supported:
	:at - The index to insert or update the items in the
collection on the client. By default -1 is used, which appends the items
to the parent DOM container. A value of 0 prepends the items.
Note that this operation is equal to inserting the items one by one, each at
the given index. Therefore, when inserting multiple items at an index other than -1,
the UI will display the items in reverse order:
stream(socket, :songs, [song1, song2, song3], at: 0)
In this case the UI will prepend song1, then song2 and then song3, so it will show
song3, song2, song1 and then any previously inserted items.
To insert in the order of the list, use Enum.reverse/1:
stream(socket, :songs, Enum.reverse([song1, song2, song3]), at: 0)

	:reset - A boolean to reset the stream on the client or not. Defaults
to false.

	:limit - An optional positive or negative number of results to limit
on the UI on the client. As new items are streamed, the UI will remove existing
items to maintain the limit. For example, to limit the stream to the last 10 items
in the UI while appending new items, pass a negative value:
stream(socket, :songs, songs, at: -1, limit: -10)
Likewise, to limit the stream to the first 10 items, while prepending new items,
pass a positive value:
stream(socket, :songs, songs, at: 0, limit: 10)


Once a stream is defined, a new @streams assign is available containing
the name of the defined streams. For example, in the above definition, the
stream may be referenced as @streams.songs in your template. Stream items
are temporary and freed from socket state immediately after the render/1
function is invoked (or a template is rendered from disk).
By default, calling stream/4 on an existing stream will bulk insert the new items
on the client while leaving the existing items in place. Streams may also be reset
when calling stream/4, which we discuss below.

  
    
  
  Resetting a stream


To empty a stream container on the client, you can pass :reset with an empty list:
stream(socket, :songs, [], reset: true)
Or you can replace the entire stream on the client with a new collection:
stream(socket, :songs, new_songs, reset: true)

  
    
  
  Limiting a stream


It is often useful to limit the number of items in the UI while allowing the
server to stream new items in a fire-and-forget fashion. This prevents
the server from overwhelming the client with new results while also opening up
powerful features like virtualized infinite scrolling. See a complete
bidirectional infinite scrolling example with stream limits in the
scroll events guide
When a stream exceeds the limit on the client, the existing items will be pruned
based on the number of items in the stream container and the limit direction. A
positive limit will prune items from the end of the container, while a negative
limit will prune items from the beginning of the container.
Note that the limit is not enforced on the first mount/3 render (when no websocket
connection was established yet), as it means more data than necessary has been
loaded. In such cases, you should only load and pass the desired amount of items
to the stream.
When inserting single items using stream_insert/4, the limit needs to be passed
as an option for it to be enforced on the client:
stream_insert(socket, :songs, song, limit: -10)

  
    
  
  Required DOM attributes


For stream items to be trackable on the client, the following requirements
must be met:
	The parent DOM container must include a phx-update="stream" attribute,
along with a unique DOM id.
	Each stream item must include its DOM id on the item's element.

Note
Failing to place phx-update="stream" on the immediate parent for
each stream will result in broken behavior.
Also, do not alter the generated DOM ids, e.g., by prefixing them. Doing so will
result in broken behavior.
When consuming a stream in a template, the DOM id and item is passed as a tuple,
allowing convenient inclusion of the DOM id for each item. For example:
<table>
  <tbody id="songs" phx-update="stream">
    <tr
      :for={{dom_id, song} <- @streams.songs}
      id={dom_id}
    >
      <td>{song.title}</td>
      <td>{song.duration}</td>
    </tr>
  </tbody>
</table>
We consume the stream in a for comprehension by referencing the
@streams.songs assign. We used the computed DOM id to populate
the <tr> id, then we render the table row as usual.
Now stream_insert/3 and stream_delete/3 may be issued and new rows will
be inserted or deleted from the client.

  
    
  
  Handling the empty case


When rendering a list of items, it is common to show a message for the empty case.
But when using streams, we cannot rely on Enum.empty?/1 or similar approaches to
check if the list is empty. Instead we can use the CSS :only-child selector
and show the message client side:
<table>
  <tbody id="songs" phx-update="stream">
    <tr id="songs-empty" class="only:table-row hidden">
      <td colspan="2">No songs found</td>
    </tr>
    <tr
      :for={{dom_id, song} <- @streams.songs}
      id={dom_id}
    >
      <td>{song.title}</td>
      <td>{song.duration}</td>
    </tr>
  </tbody>
</table>
It is important to set a unique ID on the empty row, otherwise it cannot be tracked
in the stream container and subsequent patches will duplicate the node.

  
    
  
  Non-stream items in stream containers


In the section on handling the empty case, we showed how to render a message when
the stream is empty by rendering a non-stream item inside the stream container.
Note that for non-stream items inside a phx-update="stream" container, the following
needs to be considered:
	Non-stream items must have a unique DOM id.

	Items can be added and updated, but not removed, even if the stream is reset.
This means that if you try to conditionally render a non-stream item inside a stream container,
it won't be removed if it was rendered once.

	Items are affected by the :at option.
For example, when you render a non-stream item at the beginning of the stream container and then
prepend items (with at: 0) to the stream, the non-stream item will be pushed down.



  



  
    
      
    
    
      stream_configure(socket, name, opts)



        
          
        

    

  


  

      

          @spec stream_configure(
  socket :: Phoenix.LiveView.Socket.t(),
  name :: atom() | String.t(),
  opts :: Keyword.t()
) :: Phoenix.LiveView.Socket.t()


      


Configures a stream.
The following options are supported:
	:dom_id - An optional function to generate each stream item's DOM id.
The function accepts each stream item and converts the item to a string id.
By default, the :id field of a map or struct will be used if the item has
such a field, and will be prefixed by the name hyphenated with the id.
For example, the following examples are equivalent:
stream(socket, :songs, songs)

socket
|> stream_configure(:songs, dom_id: &("songs-#{&1.id}"))
|> stream(:songs, songs)


A stream must be configured before items are inserted, and once configured,
a stream may not be re-configured. To ensure a stream is only configured a
single time in a LiveComponent, use the mount/1 callback. For example:
def mount(socket) do
  {:ok, stream_configure(socket, :songs, dom_id: &("songs-#{&1.id}"))}
end

def update(assigns, socket) do
  {:ok, stream(socket, :songs, ...)}
end
Returns an updated socket.

  



  
    
      
    
    
      stream_delete(socket, name, item)



        
          
        

    

  


  

      

          @spec stream_delete(
  socket :: Phoenix.LiveView.Socket.t(),
  name :: atom() | String.t(),
  item :: any()
) ::
  Phoenix.LiveView.Socket.t()


      


Deletes an item from the stream.
The item's DOM is computed from the :dom_id provided in the stream/3 definition.
Delete information for this DOM id is sent to the client and the item's element
is removed from the DOM, following the same behavior of element removal, such as
invoking phx-remove commands and executing client hook destroyed() callbacks.

  
    
  
  Examples


def handle_event("delete", %{"id" => id}, socket) do
  song = get_song!(id)
  {:noreply, stream_delete(socket, :songs, song)}
end
See stream_delete_by_dom_id/3 to remove an item without requiring the
original data structure.
Returns an updated socket.

  



  
    
      
    
    
      stream_delete_by_dom_id(socket, name, id)



        
          
        

    

  


  

      

          @spec stream_delete_by_dom_id(
  socket :: Phoenix.LiveView.Socket.t(),
  name :: atom() | String.t(),
  id :: String.t()
) :: Phoenix.LiveView.Socket.t()


      


Deletes an item from the stream given its computed DOM id.
Returns an updated socket.
Behaves just like stream_delete/3, but accept the precomputed DOM id,
which allows deleting from a stream without fetching or building the original
stream data structure.

  
    
  
  Examples


def render(assigns) do
  ~H"""
  <table>
    <tbody id="songs" phx-update="stream">
      <tr
        :for={{dom_id, song} <- @streams.songs}
        id={dom_id}
      >
        <td>{song.title}</td>
        <td><button phx-click={JS.push("delete", value: %{id: dom_id})}>delete</button></td>
      </tr>
    </tbody>
  </table>
  """
end

def handle_event("delete", %{"id" => dom_id}, socket) do
  {:noreply, stream_delete_by_dom_id(socket, :songs, dom_id)}
end

  



    

  
    
      
    
    
      stream_insert(socket, name, item, opts \\ [])



        
          
        

    

  


  

      

          @spec stream_insert(
  socket :: Phoenix.LiveView.Socket.t(),
  name :: atom() | String.t(),
  item :: any(),
  opts :: Keyword.t()
) :: Phoenix.LiveView.Socket.t()


      


Inserts a new item or updates an existing item in the stream.
Returns an updated socket.
See stream/4 for inserting multiple items at once.
The following options are supported:
	:at - The index to insert or update the item in the collection on the client.
By default, the item is appended to the parent DOM container. This is the same as
passing a value of -1.
If the item already exists in the parent DOM container then it will be
updated in place.

	:limit - A limit of items to maintain in the UI. A limit passed to stream/4 does
not affect subsequent calls to stream_insert/4, therefore the limit must be passed
here as well in order to be enforced. See stream/4 for more information on
limiting streams.

	:update_only - A boolean to only update the item in the stream. If the item does not
exist on the client, it will not be inserted. Defaults to false.



  
    
  
  Examples


Imagine you define a stream on mount with a single item:
stream(socket, :songs, [%Song{id: 1, title: "Song 1"}])
Then, in a callback such as handle_info or handle_event, you
can append a new song:
stream_insert(socket, :songs, %Song{id: 2, title: "Song 2"})
Or prepend a new song with at: 0:
stream_insert(socket, :songs, %Song{id: 2, title: "Song 2"}, at: 0)
Or update an existing song (in this case the :at option has no effect):
stream_insert(socket, :songs, %Song{id: 1, title: "Song 1 updated"}, at: 0)
Or append a new song while limiting the stream to the last 10 items:
stream_insert(socket, :songs, %Song{id: 2, title: "Song 2"}, limit: -10)

  
    
  
  Updating Items


As shown, an existing item on the client can be updated by issuing a stream_insert
for the existing item. When the client updates an existing item, the item will remain
in the same location as it was previously, and will not be moved to the end of the
parent children. To both update an existing item and move it to another position,
issue a stream_delete, followed by a stream_insert. For example:
song = get_song!(id)

socket
|> stream_delete(:songs, song)
|> stream_insert(:songs, song, at: -1)
See stream_delete/3 for more information on deleting items.

  



  
    
      
    
    
      transport_pid(socket)



        
          
        

    

  


  

Returns the transport pid of the socket.
Raises ArgumentError if the socket is not connected.

  
    
  
  Examples


iex> transport_pid(socket)
#PID<0.107.0>

  



  
    
      
    
    
      uploaded_entries(socket, name)



        
          
        

    

  


  

Returns the completed and in progress entries for the upload.

  
    
  
  Examples


case uploaded_entries(socket, :photos) do
  {[_ | _] = completed, []} ->
    # all entries are completed

  {[], [_ | _] = in_progress} ->
    # all entries are still in progress
end

  


        

      


  

    
Phoenix.LiveView.AsyncResult 
    



      
Provides a data structure for tracking the state of an async assign.
See the Async Operations section of the Phoenix.LiveView docs for more information.

  
    
  
  Fields


	:ok? - When true, indicates the :result has been set successfully at least once.
	:loading - The current loading state
	:failed - The current failed state
	:result - The successful result of the async task


      


      
        Summary


  
    Functions
  


    
      
        failed(result, reason)

      


        Updates the failed state.



    


    
      
        loading()

      


        Creates an async result in loading state.



    


    
      
        loading(result)

      


        Updates the loading state.



    


    
      
        loading(result, loading_state)

      


        Updates the loading state of an existing async_result.



    


    
      
        ok(value)

      


        Creates a successful result.



    


    
      
        ok(result, value)

      


        Updates the successful result.



    





      


      
        Functions

        


  
    
      
    
    
      failed(result, reason)



        
          
        

    

  


  

Updates the failed state.
When failed, the loading state will be reset to nil.
If the result was previously ok?, both result and
failed will be set.

  
    
  
  Examples


iex> result = AsyncResult.loading()
iex> result = AsyncResult.failed(result, {:exit, :boom})
iex> result.failed
{:exit, :boom}
iex> result.loading
nil

  



  
    
      
    
    
      loading()



        
          
        

    

  


  

Creates an async result in loading state.

  
    
  
  Examples


iex> result = AsyncResult.loading()
iex> result.loading
true
iex> result.ok?
false

  



  
    
      
    
    
      loading(result)



        
          
        

    

  


  

Updates the loading state.
When loading, the failed state will be reset to nil.

  
    
  
  Examples


iex> result = AsyncResult.loading(%{my: :loading_state})
iex> result.loading
%{my: :loading_state}
iex> result = AsyncResult.loading(result)
iex> result.loading
true

  



  
    
      
    
    
      loading(result, loading_state)



        
          
        

    

  


  

Updates the loading state of an existing async_result.
When loading, the failed state will be reset to nil.
If the result was previously ok?, both result and
loading will be set.

  
    
  
  Examples


iex> result = AsyncResult.loading()
iex> result = AsyncResult.loading(result, %{my: :other_state})
iex> result.loading
%{my: :other_state}

  



  
    
      
    
    
      ok(value)



        
          
        

    

  


  

Creates a successful result.
The :ok? field will also be set to true to indicate this result has
completed successfully at least once, regardless of future state changes.

  
    
  
  Examples


iex> result = AsyncResult.ok("initial result")
iex> result.ok?
true
iex> result.result
"initial result"

  



  
    
      
    
    
      ok(result, value)



        
          
        

    

  


  

Updates the successful result.
The :ok? field will also be set to true to indicate this result has
completed successfully at least once, regardless of future state changes.
When ok'd, the loading and failed state will be reset to nil.

  
    
  
  Examples


iex> result = AsyncResult.loading()
iex> result = AsyncResult.ok(result, "completed")
iex> result.ok?
true
iex> result.result
"completed"
iex> result.loading
nil

  


        

      


  

    
Phoenix.LiveView.ColocatedHook 
    



      
A special HEEx :type that extracts hooks
from a co-located <script> tag at compile time.

  
    
  
  Introduction


Colocated hooks are defined as with :type={Phoenix.LiveView.ColocatedHook}:
defmodule MyAppWeb.DemoLive do
  use MyAppWeb, :live_view

  def mount(_params, _session, socket) do
    {:ok, socket}
  end

  def render(assigns) do
    ~H"""
    <input type="text" name="user[phone_number]" id="user-phone-number" phx-hook=".PhoneNumber" />
    <script :type={Phoenix.LiveView.ColocatedHook} name=".PhoneNumber">
      export default {
        mounted() {
          this.el.addEventListener("input", e => {
            let match = this.el.value.replace(/\D/g, "").match(/^(\d{3})(\d{3})(\d{4})$/)
            if(match) {
              this.el.value = `${match[1]}-${match[2]}-${match[3]}`
            }
          })
        }
      }
    </script>
    """
  end
end
You can read more about the internals of colocated hooks in the documentation for colocated JS.
A brief summary: at compile time, the hook's code is extracted into a special folder, typically in your _build directory.
Each hook is also imported into a special manifest file. The manifest file provides
a named export
which allows it to be imported by any JavaScript bundler that supports ES modules:
import {hooks} from "phoenix-colocated/my_app"

console.log(hooks);
/*
{
  "MyAppWeb.DemoLive.PhoneNumber": {...},
  ...
}
*/

  
    
  
  Options


Colocated hooks are configured through the attributes of the <script> tag.
The supported attributes are:
	name - The name of the hook. This is required and must start with a dot,
for example: name=".myhook". The same name must be used when referring to this
hook in the phx-hook attribute of another HTML element.

	runtime - If present, the hook is not extracted, but instead registered at runtime.
You should only use this option if you know that you need it. It comes with some limitations:
	The content is not processed by any bundler, therefore it must only use features
supported by the targeted browsers.
	You need to take special care about any Content Security Policies
that may be in place. See the section on runtime hooks below for more details.




  
    
  
  Runtime hooks


Runtime hooks are a special kind of colocated hook that are not removed from the DOM
when rendering the component. Instead, the hook's code is executed directly in the
browser with no bundler involved.
One example where this can be useful is when you are creating a custom page for a library
like Phoenix.LiveDashboard. The live dashboard already bundles its hooks, therefore there
is no way to add new hooks to the bundle when the live dashboard is used inside your application.
Because of this, runtime hooks must also use a slightly different syntax. While in normal
colocated hooks you'd write an export default statement, runtime hooks must evaluate to the
hook itself:
<script :type={Phoenix.LiveView.ColocatedHook} name=".MyHook" runtime>
  {
    mounted() {
      ...
    }
  }
</script>
This is because the hook's code is wrapped by LiveView into something like this:
window["phx_hook_HASH"] = function() {
  return {
    mounted() {
      ...
    }
  }
}
Still, even for runtime hooks, the hook's name needs to start with a dot and is automatically
prefixed with the module name to avoid conflicts with other hooks.
When using runtime hooks, it is important to think about any limitations that content security
policies may impose. If CSP is involved, the only way to use runtime hooks is by using CSP nonces:
<script :type={Phoenix.LiveView.ColocatedHook} name=".MyHook" runtime nonce={@script_csp_nonce}>
  function() {
    return ...;
  }
</script>
This is assuming that the @script_csp_nonce assign contains the nonce value that is also
sent in the Content-Security-Policy header.

      




  

    
Phoenix.LiveView.ColocatedJS 
    



      
A special HEEx :type that extracts any JavaScript code from a co-located
<script> tag at compile time.
Colocated JavaScript is a more generalized version of Phoenix.LiveView.ColocatedHook.
In fact, colocated hooks are built on top of ColocatedJS.
You can use ColocatedJS to define any JavaScript code (Web Components, global event listeners, etc.)
that do not necessarily need the functionalities of hooks, for example:
<script :type={Phoenix.LiveView.ColocatedJS} name="MyWebComponent">
  export default class MyWebComponent extends HTMLElement {
    connectedCallback() {
      this.innerHTML = "Hello, world!";
    }
  }
</script>
Then, in your app.js file, you could import it like this:
import colocated from "phoenix-colocated/my_app";
customElements.define("my-web-component", colocated.MyWebComponent);
In this example, you don't actually need to have special code for the web component
inside your app.js file, since you could also directly call customElements.define
inside the colocated JavaScript. However, this example shows how you can access the
exported values inside your bundle.
A note on dependencies and umbrella projects
For each application that uses colocated JavaScript, a separate directory is created
inside the phoenix-colocated folder. This allows to have clear separation between
hooks and code of dependencies, but also applications inside umbrella projects.
While dependencies would typically still bundle their own hooks and colocated JavaScript
into a separate file before publishing, simple hooks or code snippets that do not require
access to third-party libraries can also be directly imported into your own bundle.
If a library requires this, it should be stated in its documentation.

  
    
  
  Internals


While compiling the template, colocated JavaScript is extracted into a special folder inside the
Mix.Project.build_path(), called phoenix-colocated. This is customizable, as we'll see below,
but it is important that it is a directory that is not tracked by version control, because the
components are the source of truth for the code. Also, the directory is shared between applications
(this also applies to applications in umbrella projects), so it should typically also be a shared
directory not specific to a single application.
The colocated JS directory follows this structure:
_build/$MIX_ENV/phoenix-colocated/
_build/$MIX_ENV/phoenix-colocated/my_app/
_build/$MIX_ENV/phoenix-colocated/my_app/index.js
_build/$MIX_ENV/phoenix-colocated/my_app/MyAppWeb.DemoLive/line_HASH.js
_build/$MIX_ENV/phoenix-colocated/my_dependency/MyDependency.Module/line_HASH.js
...
Each application has its own folder. Inside, each module also gets its own folder, which allows
us to track and clean up outdated code.
To use colocated JS from your app.js, your bundler needs to be configured to resolve the
phoenix-colocated folder. For new Phoenix applications, this configuration is already included
in the esbuild configuration inside config.exs:
config :esbuild,
  ...
  my_app: [
    args:
      ~w(js/app.js --bundle --target=es2022 --outdir=../priv/static/assets/js --external:/fonts/* --external:/images/* --alias:@=.),
    cd: Path.expand("../assets", __DIR__),
    env: %{
      "NODE_PATH" => [Path.expand("../deps", __DIR__), Mix.Project.build_path()]
    }
  ]
The important part here is the NODE_PATH environment variable, which tells esbuild to also look
for packages inside the deps folder, as well as the Mix.Project.build_path(), which resolves to
_build/$MIX_ENV. If you use a different bundler, you'll need to configure it accordingly. If it is not
possible to configure the NODE_PATH, you can also change the folder to which LiveView writes colocated
JavaScript by setting the :target_directory option in your config.exs:
config :phoenix_live_view, :colocated_js,
  target_directory: Path.expand("../assets/node_modules/phoenix-colocated", __DIR__)
An alternative approach could be to symlink the phoenix-colocated folder into your node_modules
folder.
Tip
If you remove or modify the contents of the :target_directory folder, you can use
mix clean --all and mix compile to regenerate all colocated JavaScript.
Warning!
LiveView assumes full ownership over the configured :target_directory. When
compiling, it will delete any files and folders inside the :target_directory,
that it does not associate with a colocated JavaScript module or manifest.

  
    
  
  Imports in colocated JS


The colocated JS files are fully handled by your bundler. For Phoenix apps, this is typically
esbuild. Because colocated JS is extracted to a folder outside the regular assets folder,
special care is necessary when you need to import other files inside the colocated JS:
import { someFunction } from "some-dependency";
import somethingElse from "@/vendor/vendored-file";
While dependencies from node_modules should work out of the box, you cannot simply refer to your
assets/vendor folder using a relative path. Instead, your bundler needs to be configured to handle
an alias like @ to resolve to your local assets folder. This is configured by default in the
esbuild configuration for new Phoenix 1.8 applications using esbuild's alias option,
as can be seen in the config snippet above (--alias=@=.).

  
    
  
  Options


Colocated JavaScript can be configured through the attributes of the <script> tag.
The supported attributes are:
	name - The name under which the default export of the script is available when importing
the manifest. This is required, even if you don't plan to access the exported values or the
script does not actually export anything.

	key - A custom key to use for the export. This is used by Phoenix.LiveView.ColocatedHook to
export all hooks under the named hooks export (export { ... as hooks }).
For example, you could set this to web_components for each colocated script that defines
a web component and then import all of them as import { web_components } from "phoenix-colocated/my_app".
Defaults to :default, which means the export will be available under the manifest's default export.
This needs to be a valid JavaScript identifier.

	extension - a custom extension to use when writing the extracted file. The default is js.

	manifest - a custom manifest file to use instead of the default index.js. For example,
web_components.ts. If you change the manifest, you will need to change the
path of your JavaScript imports accordingly.



      




  

    
Phoenix.LiveView.Controller 
    



      
Helpers for rendering LiveViews from a controller.

      


      
        Summary


  
    Functions
  


    
      
        live_render(conn, view, opts \\ [])

      


        Renders a live view from a Plug request and sends an HTML response
from within a controller.



    





      


      
        Functions

        


    

  
    
      
    
    
      live_render(conn, view, opts \\ [])



        
          
        

    

  


  

Renders a live view from a Plug request and sends an HTML response
from within a controller.
It also automatically sets the @live_module assign with the value
of the LiveView to be rendered.

  
    
  
  Options


See Phoenix.Component.live_render/3 for all supported options.

  
    
  
  Examples


defmodule ThermostatController do
  use MyAppWeb, :controller

  # "use MyAppWeb, :controller" should import Phoenix.LiveView.Controller.
  # If it does not, you can either import it there or uncomment the line below:
  # import Phoenix.LiveView.Controller

  def show(conn, %{"id" => thermostat_id}) do
    live_render(conn, ThermostatLive, session: %{
      "thermostat_id" => thermostat_id,
      "current_user_id" => get_session(conn, :user_id)
    })
  end
end

  


        

      


  

    
Phoenix.LiveView.Debug 
    



      
Functions for runtime introspection and debugging of LiveViews.
This module provides utilities for inspecting and debugging LiveView processes
at runtime. It allows you to:
	List all currently connected LiveViews
	Check if a process is a LiveView
	Get the socket of a LiveView process
	Inspect LiveComponents rendered in a LiveView


  
    
  
  Examples


# List all LiveViews
iex> Phoenix.LiveView.Debug.list_liveviews()
[%{pid: #PID<0.123.0>, view: MyAppWeb.PostLive.Index, topic: "lv:12345678", transport_pid: #PID<0.122.0>}]

# Check if a process is a LiveView
iex> Phoenix.LiveView.Debug.liveview_process?(pid(0,123,0))
true

# Get the socket of a LiveView process
iex> Phoenix.LiveView.Debug.socket(pid(0,123,0))
{:ok, %Phoenix.LiveView.Socket{...}}

# Get information about LiveComponents
iex> Phoenix.LiveView.Debug.live_components(pid(0,123,0))
{:ok, [%{id: "component-1", module: MyAppWeb.PostLive.Index.Component1, ...}]}

      


      
        Summary


  
    Functions
  


    
      
        list_liveviews()

      


        Returns a list of all currently connected LiveView processes (on the current node).



    


    
      
        live_components(liveview_pid)

      


        Returns a list with information about all LiveComponents rendered in the LiveView.



    


    
      
        liveview_process?(pid)

      


        Checks if the given pid is a LiveView process.



    


    
      
        socket(liveview_pid)

      


        Returns the socket of the LiveView process.



    





      


      
        Functions

        


  
    
      
    
    
      list_liveviews()



        
          
        

    

  


  

Returns a list of all currently connected LiveView processes (on the current node).
Each entry is a map with the following keys:
	pid: The PID of the LiveView process.
	view: The module of the LiveView.
	topic: The topic of the LiveView's channel.
	transport_pid: The PID of the transport process.

The transport_pid can be used to group LiveViews on the same page.

  
    
  
  Examples


iex> list_liveviews()
[%{pid: #PID<0.123.0>, view: MyAppWeb.PostLive.Index, topic: "lv:12345678", transport_pid: #PID<0.122.0>}]

  



  
    
      
    
    
      live_components(liveview_pid)



        
          
        

    

  


  

Returns a list with information about all LiveComponents rendered in the LiveView.

  
    
  
  Examples


iex> live_components(pid)
{:ok,
 [
   %{
     id: "component-1",
     module: MyAppWeb.PostLive.Index.Component1,
     cid: 1,
     assigns: %{
       id: "component-1",
       __changed__: %{},
       flash: %{},
       myself: %Phoenix.LiveComponent.CID{cid: 1},
       ...
     }
   }
 ]}

  



  
    
      
    
    
      liveview_process?(pid)



        
          
        

    

  


  

Checks if the given pid is a LiveView process.

  
    
  
  Examples


iex> list_liveviews() |> Enum.at(0) |> Map.fetch!(:pid) |> liveview_process?()
true

iex> liveview_process?(pid(0,456,0))
false

  



  
    
      
    
    
      socket(liveview_pid)



        
          
        

    

  


  

Returns the socket of the LiveView process.

  
    
  
  Examples


iex> list_liveviews() |> Enum.at(0) |> Map.fetch!(:pid) |> socket()
{:ok, %Phoenix.LiveView.Socket{...}}

iex> socket(pid(0,123,0))
{:error, :not_alive_or_not_a_liveview}

  


        

      


  

    
Phoenix.LiveView.JS 
    



      
Provides commands for executing JavaScript utility operations on the client.
JS commands support a variety of utility operations for common client-side
needs, such as adding or removing CSS classes, setting or removing tag attributes,
showing or hiding content, and transitioning in and out with animations.
While these operations can be accomplished via client-side hooks,
JS commands are DOM-patch aware, so operations applied
by the JS APIs will stick to elements across patches from the server.
In addition to purely client-side utilities, the JS commands include a
rich push API, for extending the default phx- binding pushes with
options to customize targets, loading states, and additional payload values.
If you need to trigger these commands via JavaScript, see JavaScript interoperability.

  
    
  
  Client Utility Commands


The following utilities are included:
	add_class - Add classes to elements, with optional transitions
	remove_class - Remove classes from elements, with optional transitions
	toggle_class - Sets or removes classes from elements, with optional transitions
	set_attribute - Set an attribute on elements
	remove_attribute - Remove an attribute from elements
	toggle_attribute - Sets or removes element attribute based on attribute presence.
	ignore_attributes - Marks attributes as ignored, skipping them when patching the DOM.
	show - Show elements, with optional transitions
	hide - Hide elements, with optional transitions
	toggle - Shows or hides elements based on visibility, with optional transitions
	transition - Apply a temporary transition to elements for animations
	dispatch - Dispatch a DOM event to elements

For example, the following modal component can be shown or hidden on the
client without a trip to the server:
alias Phoenix.LiveView.JS

def hide_modal(js \\ %JS{}) do
  js
  |> JS.hide(transition: "fade-out", to: "#modal")
  |> JS.hide(transition: "fade-out-scale", to: "#modal-content")
end

def modal(assigns) do
  ~H"""
  <div id="modal" class="phx-modal" phx-remove={hide_modal()}>
    <div
      id="modal-content"
      class="phx-modal-content"
      phx-click-away={hide_modal()}
      phx-window-keydown={hide_modal()}
      phx-key="escape"
    >
      <button class="phx-modal-close" phx-click={hide_modal()}>✖</button>
      <p>{@text}</p>
    </div>
  </div>
  """
end

  
    
  
  Enhanced push events


The push/1 command allows you to extend the built-in pushed event handling
when a phx- event is pushed to the server. For example, you may wish to
target a specific component, specify additional payload values to include
with the event, apply loading states to external elements, etc. For example,
given this basic phx-click event:
<button phx-click="inc">+</button>
Imagine you need to target your current component, and apply a loading state
to the parent container while the client awaits the server acknowledgement:
alias Phoenix.LiveView.JS

~H"""
<button phx-click={JS.push("inc", loading: ".thermo", target: @myself)}>+</button>
"""
Push commands also compose with all other utilities. For example,
to add a class when pushing:
<button phx-click={
  JS.push("inc", loading: ".thermo", target: @myself)
  |> JS.add_class("warmer", to: ".thermo")
}>+</button>
Any phx-value-* attributes will also be included in the payload, their
values will be overwritten by values given directly to push/1. Any
phx-target attribute will also be used, and overwritten.
<button
  phx-click={JS.push("inc", value: %{limit: 40})}
  phx-value-room="bedroom"
  phx-value-limit="this value will be 40"
  phx-target={@myself}
>+</button>

  
    
  
  DOM Selectors


The client utility commands in this module all take an optional DOM selector
using the :to option.
This can be a string for a regular DOM selector such as:
JS.add_class("warmer", to: ".thermo")
JS.hide(to: "#modal")
JS.show(to: "body a:nth-child(2)")
It is also possible to provide scopes to the DOM selector. The following scopes
are available:
	{:inner, "selector"} To target an element within the interacted element.
	{:closest, "selector"} To target the closest element from the interacted
element upwards.

 For example, if building a dropdown component, the button could use the :inner
 scope:
 <div phx-click={JS.show(to: {:inner, ".menu"})}>
   <div>Open me</div>
   <div class="menu hidden" phx-click-away={JS.hide()}>
     I'm in the dropdown menu
   </div>
 </div>

  
    
  
  Custom JS events with JS.dispatch/1 and window.addEventListener


dispatch/1 can be used to dispatch custom JavaScript events to
elements. For example, you can use JS.dispatch("click", to: "#foo"),
to dispatch a click event to an element.
This also means you can augment your elements with custom events,
by using JavaScript's window.addEventListener and invoking them
with dispatch/1. For example, imagine you want to provide
a copy-to-clipboard functionality in your application. You can
add a custom event for it:
window.addEventListener("my_app:clipcopy", (event) => {
  if ("clipboard" in navigator) {
    const text = event.target.textContent;
    navigator.clipboard.writeText(text);
  } else {
    alert("Sorry, your browser does not support clipboard copy.");
  }
});
Now you can have a button like this:
<button phx-click={JS.dispatch("my_app:clipcopy", to: "#element-with-text-to-copy")}>
  Copy content
</button>
The combination of dispatch/1 with window.addEventListener is
a powerful mechanism to increase the amount of actions you can trigger
client-side from your LiveView code.
You can also use window.addEventListener to listen to events pushed
from the server. You can learn more in our JS interoperability guide.

      


      
        Summary


  
    Types
  


    
      
        internal()

      


    


    
      
        t()

      


    





  
    Functions
  


    
      
        add_class(names)

      


        Adds classes to elements.



    


    
      
        add_class(js, names)

      


        See add_class/1.



    


    
      
        add_class(js, names, opts)

      


        See add_class/1.



    


    
      
        concat(js1, js2)

      


        Combines two JS commands, appending the second to the first.



    


    
      
        dispatch(js \\ %JS{}, event)

      


        Dispatches an event to the DOM.



    


    
      
        dispatch(js, event, opts)

      


        See dispatch/2.



    


    
      
        exec(attr)

      


        Executes JS commands located in an element's attribute.



    


    
      
        exec(attr, opts)

      


        See exec/1.



    


    
      
        exec(js, attr, opts)

      


        See exec/1.



    


    
      
        focus(opts \\ [])

      


        Sends focus to a selector.



    


    
      
        focus(js, opts)

      


        See focus/1.



    


    
      
        focus_first(opts \\ [])

      


        Sends focus to the first focusable child in selector.



    


    
      
        focus_first(js, opts)

      


        See focus_first/1.



    


    
      
        hide(opts \\ [])

      


        Hides elements.



    


    
      
        hide(js, opts)

      


        See hide/1.



    


    
      
        ignore_attributes(attrs)

      


        Mark attributes as ignored, skipping them when patching the DOM.



    


    
      
        ignore_attributes(attrs, opts)

      


    


    
      
        ignore_attributes(js, attrs, opts)

      


    


    
      
        navigate(href)

      


        Sends a navigation event to the server and updates the browser's pushState history.



    


    
      
        navigate(href, opts)

      


        See navigate/1.



    


    
      
        navigate(js, href, opts)

      


        See navigate/1.



    


    
      
        patch(href)

      


        Sends a patch event to the server and updates the browser's pushState history.



    


    
      
        patch(href, opts)

      


        See patch/1.



    


    
      
        patch(js, href, opts)

      


        See patch/1.



    


    
      
        pop_focus(js \\ %JS{})

      


        Focuses the last pushed element.



    


    
      
        push(event)

      


        Pushes an event to the server.



    


    
      
        push(event, opts)

      


        See push/1.



    


    
      
        push(js, event, opts)

      


        See push/1.



    


    
      
        push_focus(opts \\ [])

      


        Pushes focus from the source element to be later popped.



    


    
      
        push_focus(js, opts)

      


        See push_focus/1.



    


    
      
        remove_attribute(attr)

      


        Removes an attribute from elements.



    


    
      
        remove_attribute(attr, opts)

      


        See remove_attribute/1.



    


    
      
        remove_attribute(js, attr, opts)

      


        See remove_attribute/1.



    


    
      
        remove_class(names)

      


        Removes classes from elements.



    


    
      
        remove_class(js, names)

      


        See remove_class/1.



    


    
      
        remove_class(js, names, opts)

      


        See remove_class/1.



    


    
      
        set_attribute(arg)

      


        Sets an attribute on elements.



    


    
      
        set_attribute(js, opts)

      


        See set_attribute/1.



    


    
      
        set_attribute(js, arg, opts)

      


        See set_attribute/1.



    


    
      
        show(opts \\ [])

      


        Shows elements.



    


    
      
        show(js, opts)

      


        See show/1.



    


    
      
        toggle(opts \\ [])

      


        Toggles element visibility.



    


    
      
        toggle(js, opts)

      


        See toggle/1.



    


    
      
        toggle_attribute(arg)

      


        Sets or removes element attribute based on attribute presence.



    


    
      
        toggle_attribute(js, opts)

      


        See toggle_attribute/1.



    


    
      
        toggle_attribute(js, arg, opts)

      


        See toggle_attribute/1.



    


    
      
        toggle_class(names)

      


        Adds or removes element classes based on presence.



    


    
      
        toggle_class(js, names)

      


    


    
      
        toggle_class(js, names, opts)

      


    


    
      
        transition(transition)

      


        Transitions elements.



    


    
      
        transition(transition, opts)

      


        See transition/1.



    


    
      
        transition(js, transition, opts)

      


        See transition/1.



    





      


      
        Types

        


  
    
      
    
    
      internal()



        
          
        

    

  


  

      

          @opaque internal()


      



  



  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %Phoenix.LiveView.JS{ops: internal()}


      



  


        

      

      
        Functions

        


  
    
      
    
    
      add_class(names)



        
          
        

    

  


  

Adds classes to elements.
	names - A string with one or more class names to add.


  
    
  
  Options


	:to - An optional DOM selector to add classes to.
Defaults to the interacted element. See the DOM selectors
section for details.
	:transition - A string of classes to apply before adding classes or
a 3-tuple containing the transition class, the class to apply
to start the transition, and the ending transition class, such as:
{"ease-out duration-300", "opacity-0", "opacity-100"}
	:time - The time in milliseconds to apply the transition from :transition.
Defaults to 200.
	:blocking - A boolean flag to block the UI during the transition. Defaults true.


  
    
  
  Examples


<div id="item">My Item</div>
<button phx-click={JS.add_class("highlight underline", to: "#item")}>
  highlight!
</button>

  



  
    
      
    
    
      add_class(js, names)



        
          
        

    

  


  

See add_class/1.

  



  
    
      
    
    
      add_class(js, names, opts)



        
          
        

    

  


  

See add_class/1.

  



  
    
      
    
    
      concat(js1, js2)



        
          
        

    

  


  

Combines two JS commands, appending the second to the first.

  



    

  
    
      
    
    
      dispatch(js \\ %JS{}, event)



        
          
        

    

  


  

Dispatches an event to the DOM.
	event - The string event name to dispatch.

Note: All events dispatched are of a type
CustomEvent,
with the exception of "click". For a "click", a
MouseEvent
is dispatched to properly simulate a UI click.
For emitted CustomEvent's, the event detail will contain a dispatcher,
which references the DOM node that dispatched the JS event to the target
element.

  
    
  
  Options


	:to - An optional DOM selector to dispatch the event to.
Defaults to the interacted element. See the DOM selectors
section for details.
	:detail - An optional detail map to dispatch along
with the client event. The details will be available in the
event.detail attribute for event listeners.
	:bubbles – A boolean flag to bubble the event or not. Defaults to true.
	:blocking - A boolean flag to block the UI until the event handler calls event.detail.done().
The done function is injected by LiveView and must be called eventually to unblock the UI.
This is useful to integrate with third party JavaScript based animation libraries.


  
    
  
  Examples


window.addEventListener("click", e => console.log("clicked!", e.detail))
<button phx-click={JS.dispatch("click", to: ".nav")}>Click me!</button>

  



  
    
      
    
    
      dispatch(js, event, opts)



        
          
        

    

  


  

See dispatch/2.

  



  
    
      
    
    
      exec(attr)



        
          
        

    

  


  

Executes JS commands located in an element's attribute.
	attr - The string attribute where the JS command is specified


  
    
  
  Options


	:to - An optional DOM selector to fetch the attribute from.
Defaults to the interacted element. See the DOM selectors
section for details.


  
    
  
  Examples


<div id="modal" phx-remove={JS.hide("#modal")}>...</div>
<button phx-click={JS.exec("phx-remove", to: "#modal")}>close</button>

  



  
    
      
    
    
      exec(attr, opts)



        
          
        

    

  


  

See exec/1.

  



  
    
      
    
    
      exec(js, attr, opts)



        
          
        

    

  


  

See exec/1.

  



    

  
    
      
    
    
      focus(opts \\ [])



        
          
        

    

  


  

Sends focus to a selector.

  
    
  
  Options


	:to - An optional DOM selector to send focus to.
Defaults to the interacted element. See the DOM selectors
section for details.


  
    
  
  Examples


JS.focus(to: "main")

  



  
    
      
    
    
      focus(js, opts)



        
          
        

    

  


  

See focus/1.

  



    

  
    
      
    
    
      focus_first(opts \\ [])



        
          
        

    

  


  

Sends focus to the first focusable child in selector.

  
    
  
  Options


	:to - An optional DOM selector to focus.
Defaults to the interacted element. See the DOM selectors
section for details.


  
    
  
  Examples


JS.focus_first(to: "#modal")

  



  
    
      
    
    
      focus_first(js, opts)



        
          
        

    

  


  

See focus_first/1.

  



    

  
    
      
    
    
      hide(opts \\ [])



        
          
        

    

  


  

Hides elements.
Note: Only targets elements that are visible, meaning they have a height and/or width greater than zero.

  
    
  
  Options


	:to - An optional DOM selector to hide.
Defaults to the interacted element. See the DOM selectors
section for details.
	:transition - A string of classes to apply before hiding or
a 3-tuple containing the transition class, the class to apply
to start the transition, and the ending transition class, such as:
{"ease-out duration-300", "opacity-100", "opacity-0"}
	:time - The time in milliseconds to apply the transition from :transition.
Defaults to 200.
	:blocking - A boolean flag to block the UI during the transition. Defaults true.

During the process, the following events will be dispatched to the hidden elements:
	When the action is triggered on the client, phx:hide-start is dispatched.
	After the time specified by :time, phx:hide-end is dispatched.


  
    
  
  Examples


<div id="item">My Item</div>

<button phx-click={JS.hide(to: "#item")}>
  hide!
</button>

<button phx-click={JS.hide(to: "#item", transition: "fade-out-scale")}>
  hide fancy!
</button>

  



  
    
      
    
    
      hide(js, opts)



        
          
        

    

  


  

See hide/1.

  



  
    
      
    
    
      ignore_attributes(attrs)



        
          
        

    

  


  

Mark attributes as ignored, skipping them when patching the DOM.
Accepts a single attribute name or a list of attribute names.
An asterisk * can be used as a wildcard.
Once set, the given attributes will not be patched across LiveView updates.
This includes attributes that are removed by the server.
If you need to "unmark" an attribute, you need to call ignore_attributes/1 again
with an updated list of attributes.
This is mostly useful in combination with the phx-mounted binding, for example:
<dialog phx-mounted={JS.ignore_attributes("open")}>
  ...
</dialog>

  
    
  
  Options


	:to - An optional DOM selector to select the target element.
Defaults to the interacted element. See the DOM selectors
section for details.


  
    
  
  Examples


JS.ignore_attributes(["open", "data-*"], to: "#my-dialog")

  



  
    
      
    
    
      ignore_attributes(attrs, opts)



        
          
        

    

  


  


  



  
    
      
    
    
      ignore_attributes(js, attrs, opts)



        
          
        

    

  


  


  



  
    
      
    
    
      navigate(href)



        
          
        

    

  


  

Sends a navigation event to the server and updates the browser's pushState history.

  
    
  
  Options


	:replace - Whether to replace the browser's pushState history. Defaults to false.


  
    
  
  Examples


JS.navigate("/my-path")

  



  
    
      
    
    
      navigate(href, opts)



        
          
        

    

  


  

See navigate/1.

  



  
    
      
    
    
      navigate(js, href, opts)



        
          
        

    

  


  

See navigate/1.

  



  
    
      
    
    
      patch(href)



        
          
        

    

  


  

Sends a patch event to the server and updates the browser's pushState history.

  
    
  
  Options


	:replace - Whether to replace the browser's pushState history. Defaults to false.


  
    
  
  Examples


JS.patch("/my-path")

  



  
    
      
    
    
      patch(href, opts)



        
          
        

    

  


  

See patch/1.

  



  
    
      
    
    
      patch(js, href, opts)



        
          
        

    

  


  

See patch/1.

  



    

  
    
      
    
    
      pop_focus(js \\ %JS{})



        
          
        

    

  


  

Focuses the last pushed element.

  
    
  
  Examples


JS.pop_focus()

  



  
    
      
    
    
      push(event)



        
          
        

    

  


  

Pushes an event to the server.
	event - The string event name to push.


  
    
  
  Options


	:target - A selector or component ID to push to. This value will
overwrite any phx-target attribute present on the element.
	:loading - A selector to apply the phx loading classes to,
such as phx-click-loading in case the event was triggered by
phx-click. The element will be locked from server updates
until the push is acknowledged by the server.
	:page_loading - Boolean to trigger the phx:page-loading-start and
phx:page-loading-stop events for this push. Defaults to false.
	:value - A map of values to send to the server. These values will be
merged over any phx-value-* attributes that are present on the element.
All keys will be treated as strings when merging. When used on a form event
like phx-change or phx-submit, the precedence is
JS.push value > phx-value-* > input value.


  
    
  
  Examples


<button phx-click={JS.push("clicked")}>click me!</button>
<button phx-click={JS.push("clicked", value: %{id: @id})}>click me!</button>
<button phx-click={JS.push("clicked", page_loading: true)}>click me!</button>

  



  
    
      
    
    
      push(event, opts)



        
          
        

    

  


  

See push/1.

  



  
    
      
    
    
      push(js, event, opts)



        
          
        

    

  


  

See push/1.

  



    

  
    
      
    
    
      push_focus(opts \\ [])



        
          
        

    

  


  

Pushes focus from the source element to be later popped.

  
    
  
  Options


	:to - An optional DOM selector to push focus to.
Defaults to the interacted element. See the DOM selectors
section for details.


  
    
  
  Examples


JS.push_focus()
JS.push_focus(to: "#my-button")

  



  
    
      
    
    
      push_focus(js, opts)



        
          
        

    

  


  

See push_focus/1.

  



  
    
      
    
    
      remove_attribute(attr)



        
          
        

    

  


  

Removes an attribute from elements.
	attr - The string attribute name to remove.


  
    
  
  Options


	:to - An optional DOM selector to remove attributes from.
Defaults to the interacted element. See the DOM selectors
section for details.


  
    
  
  Examples


<button phx-click={JS.remove_attribute("aria-expanded", to: "#dropdown")}>
  hide
</button>

  



  
    
      
    
    
      remove_attribute(attr, opts)



        
          
        

    

  


  

See remove_attribute/1.

  



  
    
      
    
    
      remove_attribute(js, attr, opts)



        
          
        

    

  


  

See remove_attribute/1.

  



  
    
      
    
    
      remove_class(names)



        
          
        

    

  


  

Removes classes from elements.
	names - A string with one or more class names to remove.


  
    
  
  Options


	:to - An optional DOM selector to remove classes from.
Defaults to the interacted element. See the DOM selectors
section for details.
	:transition - A string of classes to apply before removing classes or
a 3-tuple containing the transition class, the class to apply
to start the transition, and the ending transition class, such as:
{"ease-out duration-300", "opacity-0", "opacity-100"}
	:time - The time in milliseconds to apply the transition from :transition.
Defaults to 200.
	:blocking - A boolean flag to block the UI during the transition. Defaults true.


  
    
  
  Examples


<div id="item">My Item</div>
<button phx-click={JS.remove_class("highlight underline", to: "#item")}>
  remove highlight!
</button>

  



  
    
      
    
    
      remove_class(js, names)



        
          
        

    

  


  

See remove_class/1.

  



  
    
      
    
    
      remove_class(js, names, opts)



        
          
        

    

  


  

See remove_class/1.

  



  
    
      
    
    
      set_attribute(arg)



        
          
        

    

  


  

Sets an attribute on elements.
Accepts a tuple containing the string attribute name/value pair.

  
    
  
  Options


	:to - An optional DOM selector to add attributes to.
Defaults to the interacted element. See the DOM selectors
section for details.


  
    
  
  Examples


<button phx-click={JS.set_attribute({"aria-expanded", "true"}, to: "#dropdown")}>
  show
</button>

  



  
    
      
    
    
      set_attribute(js, opts)



        
          
        

    

  


  

See set_attribute/1.

  



  
    
      
    
    
      set_attribute(js, arg, opts)



        
          
        

    

  


  

See set_attribute/1.

  



    

  
    
      
    
    
      show(opts \\ [])



        
          
        

    

  


  

Shows elements.
Note: Only targets elements that are hidden, meaning they have a height and/or width equal to zero.

  
    
  
  Options


	:to - An optional DOM selector to show.
Defaults to the interacted element. See the DOM selectors
section for details.
	:transition - A string of classes to apply before showing or
a 3-tuple containing the transition class, the class to apply
to start the transition, and the ending transition class, such as:
{"ease-out duration-300", "opacity-0", "opacity-100"}
	:time - The time in milliseconds to apply the transition from :transition.
Defaults to 200.
	:blocking - A boolean flag to block the UI during the transition. Defaults true.
	:display - An optional display value to set when showing. Defaults to "block".

During the process, the following events will be dispatched to the shown elements:
	When the action is triggered on the client, phx:show-start is dispatched.
	After the time specified by :time, phx:show-end is dispatched.


  
    
  
  Examples


<div id="item">My Item</div>

<button phx-click={JS.show(to: "#item")}>
  show!
</button>

<button phx-click={JS.show(to: "#item", transition: "fade-in-scale")}>
  show fancy!
</button>

  



  
    
      
    
    
      show(js, opts)



        
          
        

    

  


  

See show/1.

  



    

  
    
      
    
    
      toggle(opts \\ [])



        
          
        

    

  


  

Toggles element visibility.

  
    
  
  Options


	:to - An optional DOM selector to toggle.
Defaults to the interacted element. See the DOM selectors
section for details.
	:in - A string of classes to apply when toggling in, or
a 3-tuple containing the transition class, the class to apply
to start the transition, and the ending transition class, such as:
{"ease-out duration-300", "opacity-0", "opacity-100"}
	:out - A string of classes to apply when toggling out, or
a 3-tuple containing the transition class, the class to apply
to start the transition, and the ending transition class, such as:
{"ease-out duration-300", "opacity-100", "opacity-0"}
	:time - The time in milliseconds to apply the transition :in and :out classes.
Defaults to 200.
	:display - An optional display value to set when toggling in. Defaults
to "block".
	:blocking - A boolean flag to block the UI during the transition. Defaults true.

When the toggle is complete on the client, a phx:show-start or phx:hide-start, and
phx:show-end or phx:hide-end event will be dispatched to the toggled elements.

  
    
  
  Examples


<div id="item">My Item</div>

<button phx-click={JS.toggle(to: "#item")}>
  toggle item!
</button>

<button phx-click={JS.toggle(to: "#item", in: "fade-in-scale", out: "fade-out-scale")}>
  toggle fancy!
</button>

  



  
    
      
    
    
      toggle(js, opts)



        
          
        

    

  


  

See toggle/1.

  



  
    
      
    
    
      toggle_attribute(arg)



        
          
        

    

  


  

Sets or removes element attribute based on attribute presence.
Accepts a two or three-element tuple:
	{attr, val} - Sets the attribute to the given value or removes it
	{attr, val1, val2} - Toggles the attribute between val1 and val2


  
    
  
  Options


	:to - An optional DOM selector to set or remove attributes from.
Defaults to the interacted element. See the DOM selectors
section for details.


  
    
  
  Examples


<button phx-click={JS.toggle_attribute({"aria-expanded", "true", "false"}, to: "#dropdown")}>
  toggle
</button>

<button phx-click={JS.toggle_attribute({"open", "true"}, to: "#dialog")}>
  toggle
</button>

  



  
    
      
    
    
      toggle_attribute(js, opts)



        
          
        

    

  


  

See toggle_attribute/1.

  



  
    
      
    
    
      toggle_attribute(js, arg, opts)



        
          
        

    

  


  

See toggle_attribute/1.

  



  
    
      
    
    
      toggle_class(names)



        
          
        

    

  


  

Adds or removes element classes based on presence.
	names - A string with one or more class names to toggle.


  
    
  
  Options


	:to - An optional DOM selector to target.
Defaults to the interacted element. See the DOM selectors
section for details.
	:transition - A string of classes to apply before adding classes or
a 3-tuple containing the transition class, the class to apply
to start the transition, and the ending transition class, such as:
{"ease-out duration-300", "opacity-0", "opacity-100"}
	:time - The time in milliseconds to apply the transition from :transition.
Defaults to 200.
	:blocking - A boolean flag to block the UI during the transition. Defaults true.


  
    
  
  Examples


<div id="item">My Item</div>
<button phx-click={JS.toggle_class("active", to: "#item")}>
  toggle active!
</button>

  



  
    
      
    
    
      toggle_class(js, names)



        
          
        

    

  


  


  



  
    
      
    
    
      toggle_class(js, names, opts)



        
          
        

    

  


  


  



  
    
      
    
    
      transition(transition)



        
          
        

    

  


  

Transitions elements.
	transition - A string of classes to apply during the transition or
a 3-tuple containing the transition class, the class to apply
to start the transition, and the ending transition class, such as:
{"ease-out duration-300", "opacity-0", "opacity-100"}

Transitions are useful for temporarily adding an animation class
to elements, such as for highlighting content changes.

  
    
  
  Options


	:to - An optional DOM selector to apply transitions to.
Defaults to the interacted element. See the DOM selectors
section for details.
	:time - The time in milliseconds to apply the transition from :transition.
Defaults to 200.
	:blocking - A boolean flag to block the UI during the transition. Defaults true.


  
    
  
  Examples


<div id="item">My Item</div>
<button phx-click={JS.transition("shake", to: "#item")}>Shake!</button>

<div phx-mounted={JS.transition({"ease-out duration-300", "opacity-0", "opacity-100"}, time: 300)}>
    duration-300 milliseconds matches time: 300 milliseconds
</div>

  



  
    
      
    
    
      transition(transition, opts)



        
          
        

    

  


  

See transition/1.

  



  
    
      
    
    
      transition(js, transition, opts)



        
          
        

    

  


  

See transition/1.

  


        

      


  

    
Phoenix.LiveView.Router 
    



      
Provides LiveView routing for Phoenix routers.

      


      
        Summary


  
    Functions
  


    
      
        fetch_live_flash(conn, opts \\ [])

      


        Fetches the LiveView and merges with the controller flash.



    


    
      
        live(path, live_view, action \\ nil, opts \\ [])

      


        Defines a LiveView route.



    


    
      
        live_session(name, opts \\ [], list)

      


        Defines a live session for live redirects within a group of live routes.



    





      


      
        Functions

        


    

  
    
      
    
    
      fetch_live_flash(conn, opts \\ [])



        
          
        

    

  


  

Fetches the LiveView and merges with the controller flash.
Replaces the default :fetch_flash plug used by Phoenix.Router.

  
    
  
  Examples


defmodule MyAppWeb.Router do
  use LiveGenWeb, :router
  import Phoenix.LiveView.Router

  pipeline :browser do
    ...
    plug :fetch_live_flash
  end
  ...
end

  



    

    

  
    
      
    
    
      live(path, live_view, action \\ nil, opts \\ [])


        (macro)


        
          
        

    

  


  

Defines a LiveView route.
A LiveView can be routed to by using the live macro with a path and
the name of the LiveView:
live "/thermostat", ThermostatLive
To navigate to this route within your app, you can use Phoenix.VerifiedRoutes:
push_navigate(socket, to: ~p"/thermostat")
push_patch(socket, to: ~p"/thermostat?page=#{page}")
HTTP requests
The HTTP request method that a route defined by the live/4 macro
responds to is GET.

  
    
  
  Actions and live navigation


It is common for a LiveView to have multiple states and multiple URLs.
For example, you can have a single LiveView that lists all articles on
your web app. For each article there is an "Edit" button which, when
pressed, opens up a modal on the same page to edit the article. It is a
best practice to use live navigation in those cases, so when you click
edit, the URL changes to "/articles/1/edit", even though you are still
within the same LiveView. Similarly, you may also want to show a "New"
button, which opens up the modal to create new entries, and you want
this to be reflected in the URL as "/articles/new".
In order to make it easier to recognize the current "action" your
LiveView is on, you can pass the action option when defining LiveViews
too:
live "/articles", ArticleLive.Index, :index
live "/articles/new", ArticleLive.Index, :new
live "/articles/:id/edit", ArticleLive.Index, :edit
The current action will always be available inside the LiveView as
the @live_action assign, that can be used to render a LiveComponent:
<.live_component :if={@live_action == :new} module={MyAppWeb.ArticleLive.FormComponent} id="form" />
Or can be used to show or hide parts of the template:
{if @live_action == :edit, do: render("form.html", user: @user)}
Note that @live_action will be nil if no action is given on the route definition.

  
    
  
  Options


	:container - an optional tuple for the HTML tag and DOM attributes to
be used for the LiveView container. For example: {:li, style: "color: blue;"}.
See Phoenix.Component.live_render/3 for more information and examples.

	:as - optionally configures the named helper. Defaults to :live when
using a LiveView without actions or defaults to the LiveView name when using
actions.

	:metadata - a map to optional feed metadata used on telemetry events and route info,
for example: %{route_name: :foo, access: :user}. This data can be retrieved by
calling Phoenix.Router.route_info/4 with the uri from the handle_params
callback. This can be used to customize a LiveView which may be invoked from
different routes.

	:private - an optional map of private data to put in the plug connection,
for example: %{route_name: :foo, access: :user}. The data will be available
inside conn.private in plug functions.



  
    
  
  Examples


defmodule MyApp.Router
  use Phoenix.Router
  import Phoenix.LiveView.Router

  scope "/", MyApp do
    pipe_through [:browser]

    live "/thermostat", ThermostatLive
    live "/clock", ClockLive
    live "/dashboard", DashboardLive, container: {:main, class: "row"}
  end
end

iex> MyApp.Router.Helpers.live_path(MyApp.Endpoint, MyApp.ThermostatLive)
"/thermostat"

  



    

  
    
      
    
    
      live_session(name, opts \\ [], list)


        (macro)


        
          
        

    

  


  

Defines a live session for live redirects within a group of live routes.
live_session/3 allow routes defined with live/4 to support
navigate redirects from the client with navigation purely over the existing
websocket connection. This allows live routes defined in the router to
mount a new root LiveView without additional HTTP requests to the server.
For backwards compatibility reasons, all live routes defined outside
of any live session are considered part of a single unnamed live session.

  
    
  
  Security Considerations


In a regular web application, we perform authentication and authorization
checks on every request. Given LiveViews start as a regular HTTP request,
they share the authentication logic with regular requests through plugs.
Once the user is authenticated, we typically validate the sessions on
the mount callback. Authorization rules generally happen on mount
(for instance, is the user allowed to see this page?) and also on
handle_event (is the user allowed to delete this item?). Performing
authorization on mount is important because navigates do not go
through the plug pipeline.
live_session can be used to draw boundaries between groups of LiveViews.
Redirecting between live_sessions will always force a full page reload
and establish a brand new LiveView connection. This is useful when LiveViews
require different authentication strategies or simply when they use different
root layouts (as the root layout is not updated between live redirects).
Please read our guide on the security model for a
detailed description and general tips on authentication, authorization,
and more.
live_session and forward
live_session does not currently work with forward. LiveView expects
your live routes to always be directly defined within the main router
of your application.
live_session and scope
Aliases set with Phoenix.Router.scope/2 are not expanded in live_session arguments.
You must use the full module name instead.

  
    
  
  Options


	:session - An optional extra session map or MFA tuple to be merged with
the LiveView session. For example, %{"admin" => true} or {MyMod, :session, []}.
For MFA, the function is invoked and the Plug.Conn struct is prepended
to the arguments list.

	:root_layout - An optional root layout tuple for the initial HTTP render to
override any existing root layout set in the router.

	:on_mount - An optional list of hooks to attach to the mount lifecycle of
each LiveView in the session. See Phoenix.LiveView.on_mount/1. Passing a
single value is also accepted.

	:layout - An optional layout the LiveView will be rendered in. Setting
this option overrides the layout via use Phoenix.LiveView. This option
may be overridden inside a LiveView by returning {:ok, socket, layout: ...}
from the mount callback



  
    
  
  Examples


scope "/", MyAppWeb do
  pipe_through :browser

  live_session :default do
    live "/feed", FeedLive, :index
    live "/status", StatusLive, :index
    live "/status/:id", StatusLive, :show
  end

  live_session :admin, on_mount: MyAppWeb.AdminLiveAuth do
    live "/admin", AdminDashboardLive, :index
    live "/admin/posts", AdminPostLive, :index
  end
end
In the example above, we have two live sessions. Live navigation between live views
in the different sessions is not possible and will always require a full page reload.
This is important in the example above because the :admin live session has authentication
requirements, defined by on_mount: MyAppWeb.AdminLiveAuth, that the other LiveViews
do not have.
If you have both regular HTTP routes (via get, post, etc) and live routes, then
you need to perform the same authentication and authorization rules in both.
For example, if you were to add a get "/admin/health" route, then you must create
your own plug that performs the same authentication and authorization rules as
MyAppWeb.AdminLiveAuth, and then pipe through it:
scope "/" do
  # Regular routes
  pipe_through [MyAppWeb.AdminPlugAuth]
  get "/admin/health", AdminHealthController, :index

  # Live routes
  live_session :admin, on_mount: MyAppWeb.AdminLiveAuth do
    live "/admin", AdminDashboardLive, :index
    live "/admin/posts", AdminPostLive, :index
  end
end

  


        

      


  

    
Phoenix.LiveViewTest 
    



      
Conveniences for testing function components as well as
LiveViews and LiveComponents.

  
    
  
  Testing function components


There are two mechanisms for testing function components. Imagine the
following component:
def greet(assigns) do
  ~H"""
  <div>Hello, {@name}!</div>
  """
end
You can test it by using render_component/3, passing the function
reference to the component as first argument:
import Phoenix.LiveViewTest

test "greets" do
  assert render_component(&MyComponents.greet/1, name: "Mary") ==
           "<div>Hello, Mary!</div>"
end
However, for complex components, often the simplest way to test them
is by using the ~H sigil itself:
import Phoenix.Component
import Phoenix.LiveViewTest

test "greets" do
  assigns = %{}
  assert rendered_to_string(~H"""
         <MyComponents.greet name="Mary" />
         """) ==
           "<div>Hello, Mary!</div>"
end
The difference is that we use rendered_to_string/1 to convert the rendered
template to a string for testing.

  
    
  
  Testing LiveViews and LiveComponents


In LiveComponents and LiveView tests, we interact with views
via process communication in substitution of a browser.
Like a browser, our test process receives messages about the
rendered updates from the view which can be asserted against
to test the life-cycle and behavior of LiveViews and their
children.

  
    
  
  Testing LiveViews


The life-cycle of a LiveView as outlined in the Phoenix.LiveView
docs details how a view starts as a stateless HTML render in a disconnected
socket state. Once the browser receives the HTML, it connects to the
server and a new LiveView process is started, remounted in a connected
socket state, and the view continues statefully. The LiveView test functions
support testing both disconnected and connected mounts separately, for example:
import Plug.Conn
import Phoenix.ConnTest
import Phoenix.LiveViewTest
@endpoint MyEndpoint

test "disconnected and connected mount", %{conn: conn} do
  conn = get(conn, "/my-path")
  assert html_response(conn, 200) =~ "<h1>My Disconnected View</h1>"

  {:ok, view, html} = live(conn)
end

test "redirected mount", %{conn: conn} do
  assert {:error, {:redirect, %{to: "/somewhere"}}} = live(conn, "my-path")
end
Here, we start by using the familiar Phoenix.ConnTest function, get/2 to
test the regular HTTP GET request which invokes mount with a disconnected socket.
Next, live/1 is called with our sent connection to mount the view in a connected
state, which starts our stateful LiveView process.
In general, it's often more convenient to test the mounting of a view
in a single step, provided you don't need the result of the stateless HTTP
render. This is done with a single call to live/2, which performs the
get step for us:
test "connected mount", %{conn: conn} do
  {:ok, _view, html} = live(conn, "/my-path")
  assert html =~ "<h1>My Connected View</h1>"
end

  
    
  
  Testing Events


The browser can send a variety of events to a LiveView via phx- bindings,
which are sent to the handle_event/3 callback. To test events sent by the
browser and assert on the rendered side effect of the event, use the
render_* functions:
	render_click/1 - sends a phx-click event and value, returning
the rendered result of the handle_event/3 callback.

	render_focus/2 - sends a phx-focus event and value, returning
the rendered result of the handle_event/3 callback.

	render_blur/1 - sends a phx-blur event and value, returning
the rendered result of the handle_event/3 callback.

	render_submit/1 - sends a form phx-submit event and value, returning
the rendered result of the handle_event/3 callback.

	render_change/1 - sends a form phx-change event and value, returning
the rendered result of the handle_event/3 callback.

	render_keydown/1 - sends a form phx-keydown event and value, returning
the rendered result of the handle_event/3 callback.

	render_keyup/1 - sends a form phx-keyup event and value, returning
the rendered result of the handle_event/3 callback.

	render_hook/3 - sends a hook event and value, returning
the rendered result of the handle_event/3 callback.


For example:
{:ok, view, _html} = live(conn, "/thermo")

assert view
       |> element("button#inc")
       |> render_click() =~ "The temperature is: 31℉"
In the example above, we are looking for a particular element on the page
and triggering its phx-click event. LiveView takes care of making sure the
element has a phx-click and automatically sends its values to the server.
You can also bypass the element lookup and directly trigger the LiveView
event in most functions:
assert render_click(view, :inc, %{}) =~ "The temperature is: 31℉"
The element style is preferred as much as possible, as it helps LiveView
perform validations and ensure the events in the HTML actually matches the
event names on the server.

  
    
  
  Testing regular messages


LiveViews are GenServer's under the hood, and can send and receive messages
just like any other server. To test the side effects of sending or receiving
messages, simply message the view and use the render function to test the
result:
send(view.pid, {:set_temp, 50})
assert render(view) =~ "The temperature is: 50℉"

  
    
  
  Testing LiveComponents


LiveComponents can be tested in two ways. One way is to use the same
render_component/2 function as function components. This will mount
the LiveComponent and render it once, without testing any of its events:
assert render_component(MyComponent, id: 123, user: %User{}) =~
         "some markup in component"
However, if you want to test how components are mounted by a LiveView
and interact with DOM events, you must use the regular live/2 macro
to build the LiveView with the component and then scope events by
passing the view and a DOM selector in a list:
{:ok, view, html} = live(conn, "/users")
html = view |> element("#user-13 a", "Delete") |> render_click()
refute html =~ "user-13"
refute view |> element("#user-13") |> has_element?()
In the example above, LiveView will lookup for an element with
ID=user-13 and retrieve its phx-target. If phx-target points
to a component, that will be the component used, otherwise it will
fallback to the view.

      


      
        Summary


  
    Functions
  


    
      
        assert_patch(view, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))

      


        Asserts a live patch will happen within timeout milliseconds.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).



    


    
      
        assert_patch(view, to, timeout)

      


        Asserts a live patch will happen to a given path within timeout
milliseconds.



    


    
      
        assert_patched(view, to)

      


        Asserts a live patch was performed, and returns the new path.



    


    
      
        assert_push_event(view, event, payload, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))

      


        Asserts an event will be pushed within timeout.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).



    


    
      
        assert_redirect(view, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))

      


        Asserts a redirect will happen within timeout milliseconds.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).



    


    
      
        assert_redirect(view, to, timeout)

      


        Asserts a redirect will happen to a given path within timeout milliseconds.



    


    
      
        assert_redirected(view, to)

      


        Asserts a redirect was performed.



    


    
      
        assert_reply(view, payload, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))

      


        Asserts a hook reply was returned from a handle_event callback.



    


    
      
        element(view, selector, text_filter \\ nil)

      


        Returns an element to scope a function to.



    


    
      
        file_input(view, form_selector, name, entries)

      


        Builds a file input for testing uploads within a form.



    


    
      
        find_live_child(parent, child_id)

      


        Gets the nested LiveView child by child_id from the parent LiveView.



    


    
      
        follow_redirect(reason, conn, to \\ nil)

      


        Follows the redirect from a render_* action or an {:error, redirect}
tuple.



    


    
      
        follow_trigger_action(form, conn)

      


        Receives a form_element and asserts that phx-trigger-action has been
set to true, following up on that request.



    


    
      
        form(view, selector, form_data \\ %{})

      


        Returns a form element to scope a function to.



    


    
      
        has_element?(element)

      


        Checks if the given element exists on the page.



    


    
      
        has_element?(view, selector, text_filter \\ nil)

      


        Checks if the given selector with text_filter is on view.



    


    
      
        live(conn, path \\ nil, opts \\ [])

      


        Spawns a connected LiveView process.



    


    
      
        live_children(parent)

      


        Returns the current list of LiveView children for the parent LiveView.



    


    
      
        live_isolated(conn, live_view, opts \\ [])

      


        Spawns a connected LiveView process mounted in isolation as the sole rendered element.



    


    
      
        live_redirect(view, opts)

      


        Performs a live redirect from one LiveView to another.



    


    
      
        open_browser(view_or_element, open_fun \\ &open_with_system_cmd/1)

      


        Open the default browser to display current HTML of view_or_element.



    


    
      
        page_title(view)

      


        Returns the most recent title that was updated via a page_title assign.



    


    
      
        preflight_upload(upload)

      


        Performs a preflight upload request.



    


    
      
        put_connect_params(conn, params)

      


        Puts connect params to be used on LiveView connections.



    


    
      
        put_submitter(form, element_or_selector)

      


        Puts the submitter element_or_selector on the given form element.



    


    
      
        refute_push_event(view, event, payload, timeout \\ Application.fetch_env!(:ex_unit, :refute_receive_timeout))

      


        Refutes an event will be pushed within timeout.



    


    
      
        refute_redirected(view)

      


        Refutes a redirect to a given path was performed.



    


    
      
        refute_redirected(view, to)

      


    


    
      
        render(view_or_element)

      


        Returns the HTML string of the rendered view or element.



    


    
      
        render_async(view_or_element, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))

      


        Awaits all current assign_async and start_async for a given LiveView or element.



    


    
      
        render_blur(element, value \\ %{})

      


        Sends a blur event given by element and returns the rendered result.



    


    
      
        render_blur(view, event, value)

      


        Sends a blur event to the view and returns the rendered result.



    


    
      
        render_change(element, value \\ %{})

      


        Sends a form change event given by element and returns the rendered result.



    


    
      
        render_change(view, event, value)

      


        Sends a form change event to the view and returns the rendered result.



    


    
      
        render_click(element, value \\ %{})

      


        Sends a click event given by element and returns the rendered result.



    


    
      
        render_click(view, event, value)

      


        Sends a click event to the view with value and returns the rendered result.



    


    
      
        render_component(component, assigns \\ Macro.escape(%{}), opts \\ [])

      


        Renders a component.



    


    
      
        render_focus(element, value \\ %{})

      


        Sends a focus event given by element and returns the rendered result.



    


    
      
        render_focus(view, event, value)

      


        Sends a focus event to the view and returns the rendered result.



    


    
      
        render_hook(view_or_element, event, value \\ %{})

      


        Sends a hook event to the view or an element and returns the rendered result.



    


    
      
        render_keydown(element, value \\ %{})

      


        Sends a keydown event given by element and returns the rendered result.



    


    
      
        render_keydown(view, event, value)

      


        Sends a keydown event to the view and returns the rendered result.



    


    
      
        render_keyup(element, value \\ %{})

      


        Sends a keyup event given by element and returns the rendered result.



    


    
      
        render_keyup(view, event, value)

      


        Sends a keyup event to the view and returns the rendered result.



    


    
      
        render_patch(view, path)

      


        Simulates a push_patch to the given path and returns the rendered result.



    


    
      
        render_submit(element, value \\ %{})

      


        Sends a form submit event given by element and returns the rendered result.



    


    
      
        render_submit(view, event, value)

      


        Sends a form submit event to the view and returns the rendered result.



    


    
      
        render_upload(upload, entry_name, percent \\ 100)

      


        Performs an upload of a file input and renders the result.



    


    
      
        rendered_to_string(rendered)

      


        Converts a rendered template to a string.



    


    
      
        submit_form(form, conn)

      


        Receives a form element and submits the HTTP request through the plug pipeline.



    


    
      
        with_target(view, target)

      


        Sets the target of the view for events.



    





      


      
        Functions

        


    

  
    
      
    
    
      assert_patch(view, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))



        
          
        

    

  


  

Asserts a live patch will happen within timeout milliseconds.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).
It returns the new path.
To assert on the flash message, you can assert on the result of the
rendered LiveView.

  
    
  
  Examples


render_click(view, :event_that_triggers_patch)
assert_patch view

render_click(view, :event_that_triggers_patch)
assert_patch view, 30

render_click(view, :event_that_triggers_patch)
path = assert_patch view
assert path =~ ~r/path/�+/

  



  
    
      
    
    
      assert_patch(view, to, timeout)



        
          
        

    

  


  

Asserts a live patch will happen to a given path within timeout
milliseconds.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).
It returns the new path.
To assert on the flash message, you can assert on the result of the
rendered LiveView.

  
    
  
  Examples


render_click(view, :event_that_triggers_patch)
assert_patch view, "/path"

render_click(view, :event_that_triggers_patch)
assert_patch view, "/path", 30

  



  
    
      
    
    
      assert_patched(view, to)



        
          
        

    

  


  

Asserts a live patch was performed, and returns the new path.
To assert on the flash message, you can assert on the result of
the rendered LiveView.

  
    
  
  Examples


render_click(view, :event_that_triggers_redirect)
assert_patched view, "/path"

  



    

  
    
      
    
    
      assert_push_event(view, event, payload, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))


        (macro)


        
          
        

    

  


  

Asserts an event will be pushed within timeout.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).

  
    
  
  Examples


assert_push_event view, "scores", %{points: 100, user: "josé"}

  



    

  
    
      
    
    
      assert_redirect(view, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))



        
          
        

    

  


  

Asserts a redirect will happen within timeout milliseconds.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).
It returns a tuple containing the new path and the flash messages from said
redirect, if any. Note the flash will contain string keys.

  
    
  
  Examples


render_click(view, :event_that_triggers_redirect)
{path, flash} = assert_redirect view
assert flash["info"] == "Welcome"
assert path =~ ~r/path\/\d+/

render_click(view, :event_that_triggers_redirect)
assert_redirect view, 30

  



  
    
      
    
    
      assert_redirect(view, to, timeout)



        
          
        

    

  


  

Asserts a redirect will happen to a given path within timeout milliseconds.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).
It returns the flash messages from said redirect, if any.
Note the flash will contain string keys.

  
    
  
  Examples


render_click(view, :event_that_triggers_redirect)
flash = assert_redirect view, "/path"
assert flash["info"] == "Welcome"

render_click(view, :event_that_triggers_redirect)
assert_redirect view, "/path", 30

  



  
    
      
    
    
      assert_redirected(view, to)



        
          
        

    

  


  

Asserts a redirect was performed.
It returns the flash messages from said redirect, if any. Note the
flash will contain string keys.

  
    
  
  Examples


render_click(view, :event_that_triggers_redirect)
flash = assert_redirected view, "/path"
assert flash["info"] == "Welcome"

  



    

  
    
      
    
    
      assert_reply(view, payload, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))


        (macro)


        
          
        

    

  


  

Asserts a hook reply was returned from a handle_event callback.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).

  
    
  
  Examples


assert_reply view, %{result: "ok", transaction_id: _}

  



    

  
    
      
    
    
      element(view, selector, text_filter \\ nil)



        
          
        

    

  


  

Returns an element to scope a function to.
It expects the current LiveView, a query selector, and a text filter.
An optional text filter may be given to filter the results by the query
selector. If the text filter is a string or a regex, it will match any
element that contains the string (including as a substring) or matches the
regex.
So a link containing the text "unopened" will match element("a", "opened").
To prevent this, a regex could specify that "opened" appear without the prefix "un".
For example, element("a", ~r{(?<!un)opened}).
But it may be clearer to add an HTML attribute to make the element easier to
select.
After the text filter is applied, only one element must remain, otherwise an
error is raised.
If no text filter is given, then the query selector itself must return
a single element.
assert view
      |> element("#term > :first-child", "Increment")
      |> render() =~ "Increment</a>"
Attribute selectors are also supported, and may be used on special cases
like ids which contain periods:
assert view
       |> element(~s{[href="/foo"][id="foo.bar.baz"]})
       |> render() =~ "Increment</a>"

  



  
    
      
    
    
      file_input(view, form_selector, name, entries)


        (macro)


        
          
        

    

  


  

Builds a file input for testing uploads within a form.
Given the form DOM selector, the upload name, and a list of maps of client metadata
for the upload, the returned file input can be passed to render_upload/2.
Client metadata takes the following form:
	:last_modified - the last modified timestamp
	:name - the name of the file
	:content - the binary content of the file
	:size - the byte size of the content
	:type - the MIME type of the file
	:relative_path - for simulating webkitdirectory metadata
	:meta - optional metadata sent by the client


  
    
  
  Examples


avatar = file_input(lv, "#my-form-id", :avatar, [%{
  last_modified: 1_594_171_879_000,
  name: "myfile.jpeg",
  content: File.read!("myfile.jpg"),
  size: 1_396_009,
  type: "image/jpeg"
}])

assert render_upload(avatar, "myfile.jpeg") =~ "100%"

  



  
    
      
    
    
      find_live_child(parent, child_id)



        
          
        

    

  


  

Gets the nested LiveView child by child_id from the parent LiveView.

  
    
  
  Examples


{:ok, view, _html} = live(conn, "/thermo")
assert clock_view = find_live_child(view, "clock")
assert render_click(clock_view, :snooze) =~ "snoozing"

  



    

  
    
      
    
    
      follow_redirect(reason, conn, to \\ nil)


        (macro)


        
          
        

    

  


  

Follows the redirect from a render_* action or an {:error, redirect}
tuple.
Imagine you have a LiveView that redirects on a render_click
event. You can make sure it immediately redirects after the
render_click action by calling follow_redirect/3:
live_view
|> render_click("redirect")
|> follow_redirect(conn)
Or in the case of an error tuple:
assert {:error, {:redirect, %{to: "/somewhere"}}} = result = live(conn, "my-path")
{:ok, view, html} = follow_redirect(result, conn)
follow_redirect/3 expects a connection as second argument.
This is the connection that will be used to perform the underlying
request.
If the LiveView redirects with a live redirect, this macro returns
{:ok, live_view, disconnected_html} with the content of the new
LiveView, the same as the live/3 macro. If the LiveView redirects
with a regular redirect, this macro returns {:ok, conn} with the
rendered redirected page. In any other case, this macro raises.
Finally, note that you can optionally assert on the path you are
being redirected to by passing a third argument:
live_view
|> render_click("redirect")
|> follow_redirect(conn, "/redirected/page")

  



  
    
      
    
    
      follow_trigger_action(form, conn)


        (macro)


        
          
        

    

  


  

Receives a form_element and asserts that phx-trigger-action has been
set to true, following up on that request.
Imagine you have a LiveView that sends an HTTP form submission. Say that it
sets the phx-trigger-action to true, as a response to a submit event.
You can follow the trigger action like this:
form = form(live_view, selector, %{"form" => "data"})

# First we submit the form. Optionally verify that phx-trigger-action
# is now part of the form.
assert render_submit(form) =~ ~r/phx-trigger-action/

# Now follow the request made by the form
conn = follow_trigger_action(form, conn)
assert conn.method == "POST"
assert conn.params == %{"form" => "data"}

  



    

  
    
      
    
    
      form(view, selector, form_data \\ %{})



        
          
        

    

  


  

Returns a form element to scope a function to.
It expects the current LiveView, a query selector, and the form data.
The query selector must return a single element.
The form data will be validated directly against the form markup and
make sure the data you are changing/submitting actually exists, failing
otherwise.

  
    
  
  Examples


assert view
      |> form("#term", user: %{name: "hello"})
      |> render_submit() =~ "Name updated"
This function is meant to mimic what the user can actually do, so you cannot
 set hidden input values. However, hidden values can be given when calling
 render_submit/2 or render_change/2, see their docs for examples.

  



  
    
      
    
    
      has_element?(element)



        
          
        

    

  


  

Checks if the given element exists on the page.

  
    
  
  Examples


assert view |> element("#some-element") |> has_element?()

  



    

  
    
      
    
    
      has_element?(view, selector, text_filter \\ nil)



        
          
        

    

  


  

Checks if the given selector with text_filter is on view.
See element/3 for more information.

  
    
  
  Examples


assert has_element?(view, "#some-element")

  



    

    

  
    
      
    
    
      live(conn, path \\ nil, opts \\ [])


        (macro)


        
          
        

    

  


  

Spawns a connected LiveView process.
If a path is given, then a regular get(conn, path)
is done and the page is upgraded to a LiveView. If
no path is given, it assumes a previously rendered
%Plug.Conn{} is given, which will be converted to
a LiveView immediately.

  
    
  
  Options


	:on_error - Can be either :raise or :warn to control whether
 detected errors like duplicate IDs or live components fail the test or just log
 a warning. Defaults to :raise.


  
    
  
  Examples


{:ok, view, html} = live(conn, "/path")
assert view.module == MyLive
assert html =~ "the count is 3"

assert {:error, {:redirect, %{to: "/somewhere"}}} = live(conn, "/path")

  



  
    
      
    
    
      live_children(parent)



        
          
        

    

  


  

Returns the current list of LiveView children for the parent LiveView.
Children are returned in the order they appear in the rendered HTML.

  
    
  
  Examples


{:ok, view, _html} = live(conn, "/thermo")
assert [clock_view] = live_children(view)
assert render_click(clock_view, :snooze) =~ "snoozing"

  



    

  
    
      
    
    
      live_isolated(conn, live_view, opts \\ [])


        (macro)


        
          
        

    

  


  

Spawns a connected LiveView process mounted in isolation as the sole rendered element.
Useful for testing LiveViews that are not directly routable, such as those
built as small components to be re-used in multiple parents. Testing routable
LiveViews is still recommended whenever possible since features such as
live navigation require routable LiveViews.

  
    
  
  Options


	:session - the session to be given to the LiveView
	:on_error - Can be either :raise or :warn to control whether
 detected errors like duplicate IDs or live components fail the test or just log
 a warning. Defaults to :raise.

All other options are forwarded to the LiveView for rendering. Refer to
Phoenix.Component.live_render/3 for a list of supported render
options.

  
    
  
  Examples


{:ok, view, html} =
  live_isolated(conn, MyAppWeb.ClockLive, session: %{"tz" => "EST"})
Use put_connect_params/2 to put connect params for a call to
Phoenix.LiveView.get_connect_params/1 in Phoenix.LiveView.mount/3:
{:ok, view, html} =
  conn
  |> put_connect_params(%{"param" => "value"})
  |> live_isolated(AppWeb.ClockLive, session: %{"tz" => "EST"})

  



  
    
      
    
    
      live_redirect(view, opts)



        
          
        

    

  


  

Performs a live redirect from one LiveView to another.
When redirecting between two LiveViews of the same live_session,
mounts the new LiveView and shutsdown the previous one, which
mimics general browser live navigation behaviour.
When attempting to navigate from a LiveView of a different
live_session, an error redirect condition is returned indicating
a failed push_navigate from the client.

  
    
  
  Examples


assert {:ok, page_live, _html} = live(conn, "/page/1")
assert {:ok, page2_live, _html} = live(conn, "/page/2")

assert {:error, {:redirect, _}} = live_redirect(page2_live, to: "/admin")

  



    

  
    
      
    
    
      open_browser(view_or_element, open_fun \\ &open_with_system_cmd/1)



        
          
        

    

  


  

Open the default browser to display current HTML of view_or_element.

  
    
  
  Examples


view
|> element("#term > :first-child", "Increment")
|> open_browser()

assert view
       |> form("#term", user: %{name: "hello"})
       |> open_browser()
       |> render_submit() =~ "Name updated"

  



  
    
      
    
    
      page_title(view)



        
          
        

    

  


  

Returns the most recent title that was updated via a page_title assign.

  
    
  
  Examples


render_click(view, :event_that_triggers_page_title_update)
assert page_title(view) =~ "my title"

  



  
    
      
    
    
      preflight_upload(upload)



        
          
        

    

  


  

Performs a preflight upload request.
Useful for testing external uploaders to retrieve the :external entry metadata.

  
    
  
  Examples


avatar = file_input(lv, "#my-form-id", :avatar, [%{name: ..., ...}, ...])
assert {:ok, %{ref: _ref, config: %{chunk_size: _}}} = preflight_upload(avatar)

  



  
    
      
    
    
      put_connect_params(conn, params)



        
          
        

    

  


  

Puts connect params to be used on LiveView connections.
See Phoenix.LiveView.get_connect_params/1.

  



  
    
      
    
    
      put_submitter(form, element_or_selector)



        
          
        

    

  


  

Puts the submitter element_or_selector on the given form element.
A submitter is an element that initiates the form's submit event on the client. When a submitter
is put on an element created with form/3 and then the form is submitted via render_submit/2,
the name/value pair of the submitter will be included in the submit event payload.
The given element or selector must exist within the form and match one of the following:
	A button or input element with type="submit".

	A button element without a type attribute.



  
    
  
  Examples


form = view |> form("#my-form")

assert form
       |> put_submitter("button[name=example]")
       |> render_submit() =~ "Submitted example"

  



    

  
    
      
    
    
      refute_push_event(view, event, payload, timeout \\ Application.fetch_env!(:ex_unit, :refute_receive_timeout))


        (macro)


        
          
        

    

  


  

Refutes an event will be pushed within timeout.
The default timeout is ExUnit's
refute_receive_timeout (100 ms).

  
    
  
  Examples


refute_push_event view, "scores", %{points: _, user: "josé"}

  



  
    
      
    
    
      refute_redirected(view)



        
          
        

    

  


  

Refutes a redirect to a given path was performed.
It returns :ok if the specified redirect isn't already in the mailbox.
If no path is specified, refutes any redirection on the given view.

  
    
  
  Examples


render_click(view, :event_that_triggers_redirect_to_path)
:ok = refute_redirected view, "/wrong_path"

  



  
    
      
    
    
      refute_redirected(view, to)



        
          
        

    

  


  


  



  
    
      
    
    
      render(view_or_element)



        
          
        

    

  


  

Returns the HTML string of the rendered view or element.
If a view is provided, the entire LiveView is rendered.
If a view after calling with_target/2 or an element
are given, only that particular context is returned.

  
    
  
  Examples


{:ok, view, _html} = live(conn, "/thermo")
assert render(view) =~ ~s|<button id="alarm">Snooze</div>|

assert view
       |> element("#alarm")
       |> render() == "Snooze"

  



    

  
    
      
    
    
      render_async(view_or_element, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))



        
          
        

    

  


  

Awaits all current assign_async and start_async for a given LiveView or element.
It renders the LiveView or Element once complete and returns the result.
The default timeout is ExUnit's
assert_receive_timeout (100 ms).

  
    
  
  Examples


{:ok, lv, html} = live(conn, "/path")
assert html =~ "loading data..."
assert render_async(lv) =~ "data loaded!"

  



    

  
    
      
    
    
      render_blur(element, value \\ %{})



        
          
        

    

  


  

Sends a blur event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-blur attribute in it. The event name
given set on phx-blur is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values can be given
with the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("#inactive")
       |> render_blur() =~ "Tap to wake"

  



  
    
      
    
    
      render_blur(view, event, value)



        
          
        

    

  


  

Sends a blur event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_blur(view, :inactive) =~ "Tap to wake"

  



    

  
    
      
    
    
      render_change(element, value \\ %{})



        
          
        

    

  


  

Sends a form change event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-change attribute in it. The event name
given set on phx-change is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values.
If you need to pass any extra values or metadata, such as the "_target"
parameter, you can do so by giving a map under the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("form")
       |> render_change(%{deg: 123}) =~ "123 exceeds limits"

# Passing metadata
{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("form")
       |> render_change(%{_target: ["deg"], deg: 123}) =~ "123 exceeds limits"
As with render_submit/2, hidden input field values can be provided like so:
refute view
      |> form("#term", user: %{name: "hello"})
      |> render_change(%{user: %{"hidden_field" => "example"}}) =~ "can't be blank"

  



  
    
      
    
    
      render_change(view, event, value)



        
          
        

    

  


  

Sends a form change event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_change(view, :validate, %{deg: 123}) =~ "123 exceeds limits"

  



    

  
    
      
    
    
      render_click(element, value \\ %{})



        
          
        

    

  


  

Sends a click event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-click attribute in it. The event name
given set on phx-click is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values can be given
with the value argument.
If the element does not have a phx-click attribute but it is
a link (the <a> tag), the link will be followed accordingly:
	if the link is a patch, the current view will be patched
	if the link is a navigate, this function will return
{:error, {:live_redirect, %{to: url}}}, which can be followed
with follow_redirect/2
	if the link is a regular link, this function will return
{:error, {:redirect, %{to: url}}}, which can be followed
with follow_redirect/2

It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("button", "Increment")
       |> render_click() =~ "The temperature is: 30℉"

  



  
    
      
    
    
      render_click(view, event, value)



        
          
        

    

  


  

Sends a click event to the view with value and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temperature is: 30℉"
assert render_click(view, :inc) =~ "The temperature is: 31℉"

  



    

    

  
    
      
    
    
      render_component(component, assigns \\ Macro.escape(%{}), opts \\ [])


        (macro)


        
          
        

    

  


  

Renders a component.
The first argument may either be a function component, as an
anonymous function:
assert render_component(&Weather.city/1, name: "Kraków") =~
         "some markup in component"
Or a stateful component as a module. In this case, this function
will mount, update, and render the component. The :id option is
a required argument:
assert render_component(MyComponent, id: 123, user: %User{}) =~
         "some markup in component"
If your component is using the router, you can pass it as argument:
assert render_component(MyComponent, %{id: 123, user: %User{}}, router: SomeRouter) =~
         "some markup in component"

  



    

  
    
      
    
    
      render_focus(element, value \\ %{})



        
          
        

    

  


  

Sends a focus event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-focus attribute in it. The event name
given set on phx-focus is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values can be given
with the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("#inactive")
       |> render_focus() =~ "Tap to wake"

  



  
    
      
    
    
      render_focus(view, event, value)



        
          
        

    

  


  

Sends a focus event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_focus(view, :inactive) =~ "Tap to wake"

  



    

  
    
      
    
    
      render_hook(view_or_element, event, value \\ %{})



        
          
        

    

  


  

Sends a hook event to the view or an element and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_hook(view, :refresh, %{deg: 32}) =~ "The temp is: 32℉"
If you are pushing events from a hook to a component, then you must pass
an element, created with element/3, as first argument and it must point
to a single element on the page with a phx-target attribute in it:
{:ok, view, _html} = live(conn, "/thermo")
assert view
       |> element("#thermo-component")
       |> render_hook(:refresh, %{deg: 32}) =~ "The temp is: 32℉"

  



    

  
    
      
    
    
      render_keydown(element, value \\ %{})



        
          
        

    

  


  

Sends a keydown event given by element and returns the rendered result.
The element is created with element/3 and must point to a single element
on the page with a phx-keydown or phx-window-keydown attribute in it.
The event name given set on phx-keydown is then sent to the appropriate
LiveView (or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values can be given with
the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert view |> element("#inc") |> render_keydown() =~ "The temp is: 31℉"

  



  
    
      
    
    
      render_keydown(view, event, value)



        
          
        

    

  


  

Sends a keydown event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_keydown(view, :inc) =~ "The temp is: 31℉"

  



    

  
    
      
    
    
      render_keyup(element, value \\ %{})



        
          
        

    

  


  

Sends a keyup event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-keyup or phx-window-keyup attribute
in it. The event name given set on phx-keyup is then sent to the
appropriate LiveView (or component if phx-target is set accordingly).
All phx-value-* entries in the element are sent as values. Extra values
can be given with the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert view |> element("#inc") |> render_keyup() =~ "The temp is: 31℉"

  



  
    
      
    
    
      render_keyup(view, event, value)



        
          
        

    

  


  

Sends a keyup event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_keyup(view, :inc) =~ "The temp is: 31℉"

  



  
    
      
    
    
      render_patch(view, path)



        
          
        

    

  


  

Simulates a push_patch to the given path and returns the rendered result.

  



    

  
    
      
    
    
      render_submit(element, value \\ %{})



        
          
        

    

  


  

Sends a form submit event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-submit attribute in it. The event name
given set on phx-submit is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values, including hidden
input fields, can be given with the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("form")
       |> render_submit(%{deg: 123, avatar: upload}) =~ "123 exceeds limits"
To submit a form along with some hidden input values:
assert view
       |> form("#term", user: %{name: "hello"})
       |> render_submit(%{user: %{"hidden_field" => "example"}}) =~ "Name updated"
To submit a form by a specific submit element via put_submitter/2:
assert view
       |> form("#term", user: %{name: "hello"})
       |> put_submitter("button[name=example_action]")
       |> render_submit() =~ "Action taken"

  



  
    
      
    
    
      render_submit(view, event, value)



        
          
        

    

  


  

Sends a form submit event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
    
  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_submit(view, :refresh, %{deg: 32}) =~ "The temp is: 32℉"

  



    

  
    
      
    
    
      render_upload(upload, entry_name, percent \\ 100)



        
          
        

    

  


  

Performs an upload of a file input and renders the result.
See file_input/4 for details on building a file input.

  
    
  
  Examples


Given the following LiveView template:
<%= for entry <- @uploads.avatar.entries do %>
  {entry.name}: {entry.progress}%
<% end %>
Your test case can assert the uploaded content:
avatar = file_input(lv, "#my-form-id", :avatar, [
  %{
    last_modified: 1_594_171_879_000,
    name: "myfile.jpeg",
    content: File.read!("myfile.jpg"),
    size: 1_396_009,
    type: "image/jpeg"
  }
])

assert render_upload(avatar, "myfile.jpeg") =~ "100%"
By default, the entire file is chunked to the server, but an optional
percentage to chunk can be passed to test chunk-by-chunk uploads:
assert render_upload(avatar, "myfile.jpeg", 49) =~ "49%"
assert render_upload(avatar, "myfile.jpeg", 51) =~ "100%"
Before making assertions about the how the upload is consumed server-side,
you will need to call render_submit/1.
In the case where an upload progress callback issues a navigate, patch, or
redirect, the following will be returned:
	for a patch, the current view will be patched
	for a navigate, this function will return
{:error, {:live_redirect, %{to: url}}}, which can be followed
with follow_redirect/2
	for a regular redirect, this function will return
{:error, {:redirect, %{to: url}}}, which can be followed
with follow_redirect/2


  



  
    
      
    
    
      rendered_to_string(rendered)



        
          
        

    

  


  

Converts a rendered template to a string.

  
    
  
  Examples


import Phoenix.Component
import Phoenix.LiveViewTest

test "greets" do
  assigns = %{}
  assert rendered_to_string(~H"""
         <MyComponents.greet name="Mary" />
         """) ==
           "<div>Hello, Mary!</div>"
end

  



  
    
      
    
    
      submit_form(form, conn)


        (macro)


        
          
        

    

  


  

Receives a form element and submits the HTTP request through the plug pipeline.
Imagine you have a LiveView that validates form data, but submits the form to
a controller via the normal form action attribute. This is especially useful
in scenarios where the result of a form submit needs to write to the plug session.
You can follow submit the form with the %Plug.Conn{}, like this:
form = form(live_view, selector, %{"form" => "data"})

# Now submit the LiveView form to the plug pipeline
conn = submit_form(form, conn)
assert conn.method == "POST"
assert conn.params == %{"form" => "data"}

  



  
    
      
    
    
      with_target(view, target)



        
          
        

    

  


  

Sets the target of the view for events.
This emulates phx-target directly in tests, without
having to dispatch the event to a specific element.
This can be useful for invoking events to one or
multiple components at the same time:
view
|> with_target("#user-1,#user-2")
|> render_click("Hide", %{})

  


        

      


  

    
Phoenix.LiveView.HTMLFormatter 
    



      
Format HEEx templates from .heex files or ~H sigils.
This is a mix format plugin.

  
    
  
  Setup


Add it as a plugin to your .formatter.exs file and make sure to put
the heex extension in the inputs option.
[
  plugins: [Phoenix.LiveView.HTMLFormatter],
  inputs: ["*.{heex,ex,exs}", "priv/*/seeds.exs", "{config,lib,test}/**/*.{heex,ex,exs}"],
  # ...
]

  
    
  
  For umbrella projects


In umbrella projects you must also change two files at the umbrella root,
add :phoenix_live_view to your deps in the mix.exs file
and add plugins: [Phoenix.LiveView.HTMLFormatter] in the .formatter.exs file.
This is because the formatter does not attempt to load the dependencies of
all children applications.

  
    
  
  Editor support


Most editors that support mix format integration should automatically format
.heex and ~H templates. Other editors may require custom integration or
even provide additional functionality. Here are some reference posts:
	Formatting HEEx templates in VS Code


  
    
  
  Options


	:line_length - The Elixir formatter defaults to a maximum line length
of 98 characters, which can be overwritten with the :line_length option
in your .formatter.exs file.

	:heex_line_length - change the line length only for the HEEx formatter.
[
  # ...omitted
  heex_line_length: 300
]

	:migrate_eex_to_curly_interpolation - Automatically migrate single expression
<%= ... %> EEx expression to the curly braces one. Defaults to true.

	:attribute_formatters - Specify formatters for certain attributes.
[
  plugins: [Phoenix.LiveView.HTMLFormatter],
  attribute_formatters: %{class: ClassFormatter},
]

	:inline_matcher - a list of regular expressions to determine if a component
should be treated as inline.
Defaults to ["link", "button"], which treats any component with link
or button in its name as inline.
Can be disabled by setting it to an empty list.



  
    
  
  Formatting


This formatter tries to be as consistent as possible with the Elixir formatter.
Given HTML like this:
  <section><h1>   <b>{@user.name}</b></h1></section>
It will be formatted as:
<section>
  <h1><b>{@user.name}</b></h1>
</section>
A block element will go to the next line, while inline elements will be kept in the current line
as long as they fit within the configured line length.
The following links list all block and inline elements.
	https://developer.mozilla.org/en-US/docs/Web/HTML/Block-level_elements#elements
	https://developer.mozilla.org/en-US/docs/Web/HTML/Inline_elements#list_of_inline_elements

It will also keep inline elements in their own lines if you intentionally write them this way:
<section>
  <h1>
    <b>{@user.name}</b>
  </h1>
</section>
This formatter will place all attributes on their own lines when they do not all fit in the
current line. Therefore this:
<section id="user-section-id" class="sm:focus:block flex w-full p-3" phx-click="send-event">
  <p>Hi</p>
</section>
Will be formatted to:
<section
  id="user-section-id"
  class="sm:focus:block flex w-full p-3"
  phx-click="send-event"
>
  <p>Hi</p>
</section>
This formatter does not format Elixir expressions with do...end.
The content within it will be formatted accordingly though. Therefore, the given
input:
<%= live_redirect(
       to: "/my/path",
  class: "my class"
) do %>
        My Link
<% end %>
Will be formatted to
<%= live_redirect(
       to: "/my/path",
  class: "my class"
) do %>
  My Link
<% end %>
Note that only the text My Link has been formatted.

  
    
  
  Intentional new lines


The formatter will keep intentional new lines. However, the formatter will
always keep a maximum of one line break in case you have multiple ones:
<p>
  text


  text
</p>
Will be formatted to:
<p>
  text

  text
</p>

  
    
  
  Inline elements


We don't format inline elements when there is a text without whitespace before
or after the element. Otherwise it would compromise what is rendered adding
an extra whitespace.
This is the list of inline elements:
https://developer.mozilla.org/en-US/docs/Web/HTML/Inline_elements#list_of_inline_elements

  
    
  
  Skip formatting


In case you don't want part of your HTML to be automatically formatted.
You can use the special phx-no-format attribute so that the formatter will
skip the element block. Note that this attribute will not be rendered.
Therefore:
<.textarea phx-no-format>My content</.textarea>
Will be kept as is your code editor, but rendered as:
<textarea>My content</textarea>

      is_tag_open(tag_type)


        (macro)


        
          
        

    

  


  


  


        

      


  

    
Phoenix.LiveView.Logger 
    



      
Instrumenter to handle logging of Phoenix.LiveView and Phoenix.LiveComponent life-cycle events.

  
    
  
  Installation


The logger is installed automatically when Live View starts.
By default, the log level is set to :debug.

  
    
  
  Module configuration


The log level can be overridden for an individual Live View module:
use Phoenix.LiveView, log: :debug
To disable logging for an individual Live View module:
use Phoenix.LiveView, log: false

  
    
  
  Telemetry


The following Phoenix.LiveView and Phoenix.LiveComponent events are logged:
	[:phoenix, :live_view, :mount, :start]
	[:phoenix, :live_view, :mount, :stop]
	[:phoenix, :live_view, :handle_params, :start]
	[:phoenix, :live_view, :handle_params, :stop]
	[:phoenix, :live_view, :handle_event, :start]
	[:phoenix, :live_view, :handle_event, :stop]
	[:phoenix, :live_component, :handle_event, :start]
	[:phoenix, :live_component, :handle_event, :stop]

See the Telemetry guide for more information.

  
    
  
  Parameter filtering


If enabled, Phoenix.LiveView.Logger will filter parameters based on the configuration of Phoenix.Logger.

      




  

    
Phoenix.LiveView.Socket 
    



      
The LiveView socket for Phoenix Endpoints.
This is typically mounted directly in your endpoint.
socket "/live", Phoenix.LiveView.Socket,
  websocket: [connect_info: [session: @session_options]]
To share an underlying transport connection between regular
Phoenix channels and LiveView processes, use Phoenix.LiveView.Socket
from your own MyAppWeb.UserSocket module.
Next, declare your channel definitions and optional connect/3, and
id/1 callbacks to handle your channel specific needs, then mount
your own socket in your endpoint:
socket "/live", MyAppWeb.UserSocket,
  websocket: [connect_info: [session: @session_options]]
If you require session options to be set at runtime, you can use
an MFA tuple. The function it designates must return a keyword list.
socket "/live", MyAppWeb.UserSocket,
  websocket: [connect_info: [session: {__MODULE__, :runtime_opts, []}]]

# ...

def runtime_opts() do
  Keyword.put(@session_options, :domain, host())
end

    Types
  


    
      
        assigns()

      


        The data in a LiveView as stored in the socket.



    


    
      
        assigns_not_in_socket()

      


        Struct returned when assigns is not in the socket.

        Types

        


  
    
      
    
    
      assigns()



        
          
        

    

  


  

      

          @type assigns() :: map() | assigns_not_in_socket()


      


The data in a LiveView as stored in the socket.

  



  
    
      
    
    
      assigns_not_in_socket()



        
          
        

    

  


  

      

          @opaque assigns_not_in_socket()


      


Struct returned when assigns is not in the socket.

          @type t() :: %Phoenix.LiveView.Socket{
  assigns: assigns(),
  endpoint: module(),
  host_uri: URI.t() | :not_mounted_at_router,
  id: binary(),
  parent_pid: nil | pid(),
  private: map(),
  redirected: nil | tuple(),
  root_pid: pid(),
  router: module(),
  sticky?: term(),
  transport_pid: pid() | nil,
  view: module()
}


      



  


        

      


  

    
Phoenix.LiveViewTest.Element 
    



      
The struct returned by Phoenix.LiveViewTest.element/3.
The following public fields represent the element:
	selector - The query selector
	text_filter - The text to further filter the element

See the Phoenix.LiveViewTest documentation for usage.

      




  

    
Phoenix.LiveViewTest.Upload 
    



      
The struct returned by Phoenix.LiveViewTest.file_input/4.
The following public fields represent the element:
	selector - The query selector
	entries - The list of selected file entries

See the Phoenix.LiveViewTest documentation for usage.

      




  

    
Phoenix.LiveViewTest.View 
    



      
The struct for testing LiveViews.
The following public fields represent the LiveView:
	id - The DOM id of the LiveView
	module - The module of the running LiveView
	pid - The Pid of the running LiveView
	endpoint - The endpoint for the LiveView
	target - The target to scope events to

See the Phoenix.LiveViewTest documentation for usage.

      




  

    
Phoenix.LiveView.UploadConfig 
    



      
The struct representing an upload.

          @type t() :: %Phoenix.LiveView.UploadConfig{
  accept: list() | :any,
  acceptable_exts: MapSet.t(),
  acceptable_types: MapSet.t(),
  allowed?: boolean(),
  auto_upload?: boolean(),
  chunk_size: term(),
  chunk_timeout: term(),
  cid: :unregistered | nil | integer(),
  client_key: String.t(),
  entries: list(),
  entry_refs_to_metas: %{required(String.t()) => map()},
  entry_refs_to_pids: %{required(String.t()) => pid() | :unregistered | :done},
  errors: list(),
  external:
    (Phoenix.LiveView.UploadEntry.t(), Phoenix.LiveView.Socket.t() ->
       {:ok | :error, meta :: %{uploader: String.t()},
        Phoenix.LiveView.Socket.t()})
    | false,
  max_entries: pos_integer(),
  max_file_size: pos_integer(),
  name: atom() | String.t(),
  progress_event:
    (name :: atom() | String.t(),
     Phoenix.LiveView.UploadEntry.t(),
     Phoenix.LiveView.Socket.t() ->
       {:noreply, Phoenix.LiveView.Socket.t()})
    | nil,
  ref: String.t(),
  writer: (name :: atom() | String.t(),
           Phoenix.LiveView.UploadEntry.t(),
           Phoenix.LiveView.Socket.t() ->
             {module(), term()})
}


      



  


        

      


  

    
Phoenix.LiveView.UploadEntry 
    



      
The struct representing an upload entry.

          @type t() :: %Phoenix.LiveView.UploadEntry{
  cancelled?: boolean(),
  client_last_modified: integer() | nil,
  client_meta: map() | nil,
  client_name: String.t() | nil,
  client_relative_path: String.t() | nil,
  client_size: integer() | nil,
  client_type: String.t() | nil,
  done?: boolean(),
  preflighted?: term(),
  progress: integer(),
  ref: String.t() | nil,
  upload_config: String.t() | :atom,
  upload_ref: String.t(),
  uuid: String.t() | nil,
  valid?: boolean()
}


      



  


        

      


  

    
Phoenix.LiveView.UploadWriter behaviour
    



      
Provide a behavior for writing uploaded chunks to a final destination.
By default, uploads are written to a temporary file on the server and
consumed by the LiveView by reading the temporary file or copying it to
durable location. Some usecases require custom handling of the uploaded
chunks, such as streaming a user's upload to another server. In these cases,
we don't want the chunks to be written to disk since we only need to forward
them on.
Note: Upload writers run inside the channel uploader process, so
any blocking work will block the channel  errors will crash the channel process.
Custom implementations of Phoenix.LiveView.UploadWriter can be passed to
allow_upload/3. To initialize the writer with options, define a 3-arity function
that returns a tuple of {writer, writer_opts}. For example imagine
an upload writer that logs the chunk sizes and tracks the total bytes sent by the
client:
socket
|> allow_upload(:avatar,
  accept: :any,
  writer: fn _name, _entry, _socket -> {EchoWriter, level: :debug} end
)
And such an EchoWriter could look like this:
defmodule EchoWriter do
  @behaviour Phoenix.LiveView.UploadWriter

  require Logger

  @impl true
  def init(opts) do
    {:ok, %{total: 0, level: Keyword.fetch!(opts, :level)}}
  end

  @impl true
  def meta(state), do: %{level: state.level}

  @impl true
  def write_chunk(data, state) do
    size = byte_size(data)
    Logger.log(state.level, "received chunk of #{size} bytes")
    {:ok, %{state | total: state.total + size}}
  end

  @impl true
  def close(state, reason) do
    Logger.log(state.level, "closing upload after #{state.total} bytes, #{inspect(reason)}")
    {:ok, state}
  end
end
When the LiveView consumes the uploaded entry, it will receive the %{level: ...}
returned from the meta callback. This allows the writer to keep state as it handles
chunks to be later relayed to the LiveView when consumed.

  
    
  
  Close reasons


The close/2 callback is called when the upload is complete or cancelled. The following
values can be passed:
	:done - The client sent all expected chunks and the upload is awaiting consumption
	:cancel - The upload was canceled, either by the server or the client navigating away.
	{:error, reason} - The upload was canceled due to an error returned from write_chunk/2.
For example, if write_chunk/2 returns {:error, :enoent, state}, the upload will be cancelled
and close/2 will be called with the reason {:error, :enoent}.

          @callback close(state :: term(), reason :: :done | :cancel | {:error, term()}) ::
  {:ok, state :: term()} | {:error, term()}


      



  



  
    
      
    
    
      init(opts)



        
          
        

    

  


  

      

          @callback init(opts :: term()) :: {:ok, state :: term()} | {:error, term()}

          @callback meta(state :: term()) :: map()


      



  



  
    
      
    
    
      write_chunk(data, state)



        
          
        

    

  


  

      

          @callback write_chunk(data :: binary(), state :: term()) ::
  {:ok, state :: term()} | {:error, reason :: term(), state :: term()}


      



  


        

      


  

    
Phoenix.LiveComponent.CID 
    



      
The struct representing an internal unique reference to the component instance,
available as the @myself assign in live components.
Read more about the uses of @myself in the Phoenix.LiveComponent docs.

      




  

    
Phoenix.LiveView.Component 
    



      
The struct returned by components in .heex templates.
This component is never meant to be output directly
into the template. It should always be handled by
the diffing algorithm.

          @type t() :: %Phoenix.LiveView.Component{
  assigns: map(),
  component: module(),
  id: binary()
}


      



  


        

      


  

    
Phoenix.LiveView.Comprehension 
    



      
The struct returned by for-comprehensions in .heex templates.
See a description about its fields and use cases
in Phoenix.LiveView.Engine docs.

          @type t() :: %Phoenix.LiveView.Comprehension{
  dynamics: [
    [
      iodata()
      | Phoenix.LiveView.Rendered.t()
      | t()
      | Phoenix.LiveView.Component.t()
    ]
  ],
  fingerprint: integer(),
  static: [String.t()],
  stream: list() | nil
}


      



  


        

      


  

    
Phoenix.LiveView.Engine 
    



      
An EEx template engine that tracks changes.
This is often used by Phoenix.LiveView.TagEngine which also adds
HTML validation. In the documentation below, we will explain how it
works internally. For user-facing documentation, see Phoenix.LiveView.

  
    
  
  Phoenix.LiveView.Rendered


Whenever you render a live template, it returns a
Phoenix.LiveView.Rendered structure. This structure has
three fields: :static, :dynamic and :fingerprint.
The :static field is a list of literal strings. This
allows the Elixir compiler to optimize this list and avoid
allocating its strings on every render.
The :dynamic field contains a function that takes a boolean argument
(see "Tracking changes" below), and returns a list of dynamic content.
Each element in the list is either one of:
	iodata - which is the dynamic content
	nil - the dynamic content did not change
	another Phoenix.LiveView.Rendered struct, see "Nesting and fingerprinting" below
	a Phoenix.LiveView.Comprehension struct, see "Comprehensions" below
	a Phoenix.LiveView.Component struct, see "Component" below

When you render a live template, you can convert the
rendered structure to iodata by alternating the static
and dynamic fields, always starting with a static entry
followed by a dynamic entry. The last entry will always
be static too. So the following structure:
%Phoenix.LiveView.Rendered{
  static: ["foo", "bar", "baz"],
  dynamic: fn track_changes? -> ["left", "right"] end
}
Results in the following content to be sent over the wire
as iodata:
["foo", "left", "bar", "right", "baz"]
This is also what calling Phoenix.HTML.Safe.to_iodata/1
with a Phoenix.LiveView.Rendered structure returns.
Of course, the benefit of live templates is exactly
that you do not need to send both static and dynamic
segments every time. So let's talk about tracking changes.

  
    
  
  Tracking changes


By default, a live template does not track changes.
Change tracking can be enabled by including a changed
map in the assigns with the key __changed__ and passing
true to the dynamic parts. The map should contain
the name of any changed field as key and the boolean
true as value. If a field is not listed in __changed__,
then it is always considered unchanged.
If a field is unchanged and live believes a dynamic
expression no longer needs to be computed, its value
in the dynamic list will be nil. This information
can be leveraged to avoid sending data to the client.

  
    
  
  Nesting and fingerprinting


Phoenix.LiveView also tracks changes across live
templates. Therefore, if your view has this:
{render("form.html", assigns)}
Phoenix will be able to track what is static and dynamic
across templates, as well as what changed. A rendered
nested live template will appear in the dynamic
list as another Phoenix.LiveView.Rendered structure,
which must be handled recursively.
However, because the rendering of live templates can
be dynamic in itself, it is important to distinguish
which live template was rendered. For example,
imagine this code:
<%= if something?, do: render("one.html", assigns), else: render("other.html", assigns) %>
To solve this, all Phoenix.LiveView.Rendered structs
also contain a fingerprint field that uniquely identifies
it. If the fingerprints are equal, you have the same
template, and therefore it is possible to only transmit
its changes.

  
    
  
  Comprehensions


Another optimization done by live templates is to
track comprehensions. If your code has this:
<%= for point <- @points do %>
  x: {point.x}
  y: {point.y}
<% end %>
Instead of rendering all points with both static and
dynamic parts, it returns a Phoenix.LiveView.Comprehension
struct with the static parts, that are shared across all
points, and a list of dynamics to be interpolated inside
the static parts. If @points is a list with %{x: 1, y: 2}
and %{x: 3, y: 4}, the above expression would return:
%Phoenix.LiveView.Comprehension{
  static: ["\n  x: ", "\n  y: ", "\n"],
  dynamics: [
    ["1", "2"],
    ["3", "4"]
  ]
}
This allows live templates to send the static parts only once,
regardless of the number of items. On the other hand, keep in
mind the collection itself is not "diffed" across renders.
If one entry in the comprehension changes, the whole collection
is sent again. Consider using Phoenix.LiveComponent and
Phoenix.LiveView.stream/4 to optimize those cases, or see the
:key attribute when using :for.
The list of dynamics is always a list of iodatas or components,
as we don't perform change tracking inside the comprehensions
themselves. Similarly, comprehensions do not have fingerprints
because they are only optimized at the root, so conditional
evaluation, as the one seen in rendering, is not possible.
The only possible outcome for a dynamic field that returns a
comprehension is nil.

  
    
  
  Components


Live also supports stateful components defined with
Phoenix.LiveComponent. Since they are stateful, they are always
handled lazily by the diff algorithm.

      




  

    
Phoenix.LiveView.HTMLEngine 
    



      
The HTMLEngine that powers .heex templates and the ~H sigil.
It works by adding a HTML parsing and validation layer on top
of Phoenix.LiveView.TagEngine.

      




  

    
Phoenix.LiveView.Rendered 
    



      
The struct returned by .heex templates.
See a description about its fields and use cases
in Phoenix.LiveView.Engine docs.

        Types

        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %Phoenix.LiveView.Rendered{
  caller:
    :not_available
    | {module(), function :: {atom(), non_neg_integer()}, file :: String.t(),
       line :: pos_integer()},
  dynamic: (boolean() ->
              [
                nil
                | iodata()
                | t()
                | Phoenix.LiveView.Comprehension.t()
                | Phoenix.LiveView.Component.t()
              ]),
  fingerprint: integer(),
  root: nil | true | false,
  static: [String.t()]
}


      



  


        

      


  

    
Phoenix.LiveView.TagEngine behaviour
    



      
An EEx engine that understands tags.
This cannot be directly used by Phoenix applications.
Instead, it is the building block by engines such as
Phoenix.LiveView.HTMLEngine.
It is typically invoked like this:
EEx.compile_string(source,
  engine: Phoenix.LiveView.TagEngine,
  line: 1,
  file: path,
  caller: __CALLER__,
  source: source,
  tag_handler: FooBarEngine
)
Where :tag_handler implements the behaviour defined by this module.

      


      
        Summary


  
    Callbacks
  


    
      
        annotate_body(caller)

      


        Callback invoked to add annotations around the whole body of a template.



    


    
      
        annotate_caller(file, line)

      


        Callback invoked to add caller annotations before a function component is invoked.



    


    
      
        annotate_slot(name, tag_meta, close_tag_meta, caller)

      


        Callback invoked to add annotations around each slot of a template.



    


    
      
        classify_type(name)

      


        Classify the tag type from the given binary.



    


    
      
        handle_attributes(ast, meta)

      


        Implements processing of attributes.



    


    
      
        void?(name)

      


        Returns if the given tag name is void or not.



    





  
    Functions
  


    
      
        component(func, assigns, caller)

      


        Renders a component defined by the given function.



    


    
      
        inner_block(name, list)

      


        Define a inner block, generally used by slots.



    





      


      
        Callbacks

        


  
    
      
    
    
      annotate_body(caller)



        
          
        

    

  


  

      

          @callback annotate_body(caller :: Macro.Env.t()) :: {String.t(), String.t()} | nil


      


Callback invoked to add annotations around the whole body of a template.

  



  
    
      
    
    
      annotate_caller(file, line)



        
          
        

    

  


  

      

          @callback annotate_caller(file :: String.t(), line :: integer()) :: String.t() | nil


      


Callback invoked to add caller annotations before a function component is invoked.

  



  
    
      
    
    
      annotate_slot(name, tag_meta, close_tag_meta, caller)



        
          
        

    

  


  

      

          @callback annotate_slot(
  name :: atom(),
  tag_meta :: %{line: non_neg_integer(), column: non_neg_integer()},
  close_tag_meta :: %{line: non_neg_integer(), column: non_neg_integer()},
  caller :: Macro.Env.t()
) :: {String.t(), String.t()} | nil


      


Callback invoked to add annotations around each slot of a template.
In case the slot is an implicit inner block, the tag meta points to
the component.

  



  
    
      
    
    
      classify_type(name)



        
          
        

    

  


  

      

          @callback classify_type(name :: binary()) :: {type :: atom(), name :: binary()}


      


Classify the tag type from the given binary.
This must return a tuple containing the type of the tag and the name of tag.
For instance, for LiveView which uses HTML as default tag handler this would
return {:tag, 'div'} in case the given binary is identified as HTML tag.
You can also return {:error, "reason"} so that the compiler will display this
error.

  



  
    
      
    
    
      handle_attributes(ast, meta)



        
          
        

    

  


  

      

          @callback handle_attributes(ast :: Macro.t(), meta :: keyword()) ::
  {:attributes, [{binary(), Macro.t()} | Macro.t()]} | {:quoted, Macro.t()}


      


Implements processing of attributes.
It returns a quoted expression or attributes. If attributes are returned,
the second element is a list where each element in the list represents
one attribute. If the list element is a two-element tuple, it is assumed
the key is the name to be statically written in the template. The second
element is the value which is also statically written to the template whenever
possible (such as binaries or binaries inside a list).

  



  
    
      
    
    
      void?(name)



        
          
        

    

  


  

      

          @callback void?(name :: binary()) :: boolean()


      


Returns if the given tag name is void or not.
That's mainly useful for HTML tags and used internally by the compiler. You
can just implement as def void?(_), do: false if you want to ignore this.

  


        

      

      
        Functions

        


  
    
      
    
    
      component(func, assigns, caller)



        
          
        

    

  


  

Renders a component defined by the given function.
This function is rarely invoked directly by users. Instead, it is used by ~H
and other engine implementations to render Phoenix.Components. For example,
the following:
<MyApp.Weather.city name="Kraków" />
Is the same as:
<%= component(
      &MyApp.Weather.city/1,
      [name: "Kraków"],
      {__ENV__.module, __ENV__.function, __ENV__.file, __ENV__.line}
    ) %>

  



  
    
      
    
    
      inner_block(name, list)


        (macro)


        
          
        

    

  


  

Define a inner block, generally used by slots.
This macro is mostly used by custom HTML engines that provide
a slot implementation and rarely called directly. The
name must be the assign name the slot/block will be stored
under.
If you're using HEEx templates, you should use its higher
level <:slot> notation instead. See Phoenix.Component
for more information.

  


        

      


  

    
Phoenix.LiveView.ReloadError exception
    






  

    
mix compile.phoenix_live_view 
    



      
A LiveView compiler for HEEx macro components.
Right now, only Phoenix.LiveView.ColocatedHook and Phoenix.LiveView.ColocatedJS
are handled.
You must add it to your mix.exs as:
compilers: [:phoenix_live_view] ++ Mix.compilers()

